

```

;*****
;*          ИГРОВАЯ      ПРОГРАМММА          *
;*          *                                                    *
;*          " Б О Л Д Е Р " v 1.5          *
;*          *                                                    *
;*          (ВЕРСИЯ ДЛЯ ПЭВМ "ВЕКТОР-06Ц")    *
;*          *                                                    *
;*          (С) ЛЕБЕДЕВ АНДРИАН ЗИНОВЬЕВИЧ    *
;*          ВОЛГОГРАД          *
;*          1991.          *
;*****
;
;-----
; ИНИЦИАЛИЗАЦИЯ РЕЖИМОВ ПЭВМ "ВЕКТОР-06Ц".
; (УСТАНОВКА ВЕКТОРА ПРЕРЫВАНИЯ И ПАЛИТРЫ ЦВЕТОВ.
;-----
        ORG 100H
;
STRINT: DI
        MVI A,0C3H
        STA 0000H
        STA 0038H
        LXI H,STRINT      ;УСТАНОВКА СТАРТА ПО "БЛК/СБРОС".
        SHLD 0001H
        LXI H,INIT      ;УСТАНОВКА ВЕКТОРА ПРЕРЫВАНИЯ.
        SHLD 0039H
        XRA A          ;ВЫКЛ. ОЗУ КВАЗИДИСКА.
        OUT 10H
        JMP START      ;ПЕРЕХОД НА НАЧАЛО.
;
;
SCROLL: DB 0FFH      ;ЯЧЕЙКА СКРОЛЛИНГА ЭКРАНА (0FFH - НОРМАЛЬНОЕ ПОЛОЖЕНИЕ)
KEYKOD: DB 0FFH      ;ЯЧЕЙКА КОДА НАЖАТИЯ НА ИГРОВОЙ КЛАВИАТУРЕ.
STATUS: DB 0FFH      ;ЯЧЕЙКА ПРИЗНАКА НАЖАТИЯ НА ЛЮБУЮ КЛАВИШУ.
;
DOWNK: EQU 7FH      ;КОД КЛАВИШИ "КУРСОР ВНИЗ".
UPK: EQU 0DFH      ;КОД КЛАВИШИ "КУРСОР ВВЕРХ".
LEFTK: EQU 0EFH      ;КОД КЛАВИШИ "КУРСОР ВЛЕВО".
RIGHTK: EQU 0BFH      ;КОД КЛАВИШИ "КУРСОР ВПРАВО".
TAB: EQU 0F6H      ;КОД КЛАВИШИ "ТАБУЛЯЦИЯ".
LF: EQU 0FDH      ;КОД КЛАВИШИ "ПС".
DEL: EQU 0F7H      ;КОД КЛАВИШИ "ЗБ".
CR: EQU 0FBH      ;КОД КЛАВИШИ "ВК".
;
; РАБОЧИЙ МАССИВ ПАЛИТРЫ ЦВЕТОВ (ВЫВОДИМЫЙ НА ЭКРАН).
;
BORDER: DB 00H
COLOR: DB 00H
COLR1: DB 3FH
COLR2: DB 87H
COLR3: DB 17H
COLR4: DB 7BH
COLR5: DB 00H
COLR6: DB 00H
COLR7: DB 00H
COLR8: DB 0FFH
COLR9: DB 00H
COLR10: DB 00H
COLR11: DB 00H
COLR12: DB 0DEH
COLR13: DB 00H
COLR14: DB 00H
COLR15: DB 00H
;
; ПРОГРАММА ОБРАБОТКИ СИСТЕМНОГО ПРЕРЫВАНИЯ (ОПРОС КЛАВИШ, ЗАПИСЬ ПАЛИТРЫ).
;
INIT: PUSH H
        PUSH B
        PUSH D
        PUSH PSW
        LDA WRPAL      ; ПРОВЕРКА РАЗРЕШЕНИЯ ЗАПИСИ ПАЛИТРЫ В

```

```

ORA A          ; ОЗУ ЦВЕТОГЕНЕРАТОРА.
JZ NWRPAL
MVI A,88H
OUT 00H
LXI H,COLR15    ; ЗАПИСЬ ПАЛИТРЫ.
LXI D,100FH
INIT1: MOV A,E
OUT 02
MOV A,M
OUT 0CH
OUT 0CH
OUT 0CH
OUT 0CH
OUT 0CH
DCX H
OUT 0CH
DCR E
OUT 0CH
DCR D
OUT 0CH
JNZ INIT1
NWRPAL: MVI A,8AH    ; ПРОВЕРКА НАЖАТИЯ НА ЛЮБУЮ КЛАВИШУ
OUT 00          ; МАТРИЦЫ КЛАВИАТУРЫ 8*8 (КРОМЕ УС,СС,РУС/ЛАТ) .
XRA A
OUT 03
IN 2
STA STATUS
MVI A,0F0H      ; ПРОВЕРКА НАЖАТИЯ НА КЛАВИШИ ИГРОВОГО РЯДА.
OUT 03
IN 02
STA KEYKOD
MVI A,88H
OUT 00
LDA BORDER      ; УСТАНОВКА ЦВЕТА БОРДЮРА.
ANI 0FH
OUT 02
LDA SCROLL      ; УСТАНОВКА ВЕРТИКАЛЬНОГО ПОЛОЖЕНИЯ ЭКРАНА.
OUT 03

;
; ОБРАБОТКА ВЫСОТЫ И ДЛИТЕЛЬНОСТИ ЗВУКА ПО НОМЕРУ ПАРАМЕТРА ЗВУКА.
;
LDA TON         ; ВЗЯТЬ КОД ИЗ БУФЕРА НОМЕРОВ ПАРАМЕТРОВ ЗВУКА.
ORA A          ; НОМЕР ПАРАМЕТРА УСТАНОВЛЕН?
JZ TSSOUN      ; ЕСЛИ НЕТ - ПЕРЕЙТИ НА ПРОВЕРКУ ОТРАБОТКИ ДЛИТ.ЗВУКА.
DCR A          ; УМЕНШИТЬ НОМЕР НА 1, Т.К. МАССИВ НАЧИН. С НУЛЕВОГО.
ADD A          ; УМНОЖИТЬ НОМЕР ПАРАМЕТРОВ ЗВУКА НА 4, ТАК КАК
ADD A          ; ПАРАМЕТРЫ ОДНОГО ЗВУКА ЗАНИМАЮТ 4 БАЙТА.
MOV E,A        ; ПОДГОТОВИТЬ В "DE" СМЕЩЕНИЕ ОТ НАЧАЛА ПАРАМЕТРОВ.
MVI D,0
LXI H,BAZTON    ; УСТАНОВИТЬ В "HL" АДРЕС НАЧАЛА ПАРАМЕТРОВ ЗВУКА.
DAD D          ; СЛОЖЕНИЕМ С "DE" ПОЛУЧИТЬ В "HL" АДРЕС НУЖНЫХ ПАРАМ.
MVI A,0B6H     ; УСТАНОВИТЬ РЕЖИМ 3 ТАЙМЕРА ДЛЯ КАНАЛА 2, ЗАПИСЬ КОЭФ-
OUT 08         ; ФИЦИЕНТА ДЕЛЕНИЯ В ДВА БАЙТА - СНАЧАЛА МЛ. ПОТОМ СТ.
MOV A,M        ; ВЗЯТЬ МЛАДШИЙ БАЙТ КОЭФФ. ДЕЛЕНИЯ И
OUT 09         ; ЗАПИСАТЬ В КАНАЛ 2 ТАЙМЕРА.
INX H          ; ПЕРЕДВИНУТЬ УКАЗАТЕЛЬ НА СЛЕДУЮЩИЙ БАЙТ ПАРАМЕТРОВ.
MOV A,M        ; ВЗЯТЬ СТАРШИЙ БАЙТ КОЭФФ. ДЕЛЕНИЯ И
OUT 09         ; ЗАПИСАТЬ В КАНАЛ 2 ТАЙМЕРА.
INX H
MOV E,M        ; ВЗЯТЬ В ПАРУ "DE" ДЛИТЕЛЬНОСТЬ ЗВУЧАНИЯ.
INX H
MOV D,M
XCHG          ; И ЗАПИСАТЬ ЕЕ В
SHLD STSOUN    ; СЧЕТЧИК ВРЕМЕНИ ЗВУЧАНИЯ.
XRA A          ; ОБНУЛИТЬ БУФЕР НОМЕРОВ ПАРАМЕТРОВ ЗВУКА (Т.Е. ПАРА-
STA TON        ; МЕТРЫ ВЗЯТЫ И ИХ НЕ НУЖНО СНОВА ОПРЕДЕЛЯТЬ).
JMP EXINIT     ; ВЫЙТИ ИЗ П/П ОБРАБОТКИ СИСТЕМНОГО ПРЕРЫВАНИЯ.

;
TSSOUN: LHLD STSOUN ; ВЗЯТЬ СЧЕТЧИК ВРЕМЕНИ ЗВУЧАНИЯ И ПРОВЕРИТЬ,
MOV A,H        ; РАВЕН ЛИ ОН НУЛЮ (ВРЕМЯ ЗВУЧАНИЯ КОНЧИЛОСЬ)?
ORA L

```

```

JZ OFSOUN      ; ЕСЛИ КОНЧИЛОСЬ - ВЫКЛЮЧИТЬ ЗВУК!
DCX H          ; УМЕНЬШИТЬ СЧЕТЧИК.
SHLD STSOUN
JMP EXINIT     ; И ВЫЙТИ ИЗ П/П ПРЕРЫВАНИЯ.
OFSOUN: MVI A,0B6H ; ВЫКЛЮЧИТЬ КАНАЛ 2 ТАЙМЕРА ПУТЕМ ЗАПИСИ В РЕГ.
OUT 08         ; РЕЖИМА УПР. СЛОВА ДЛЯ ЭТОГО КАНАЛА БЕЗ ПАРАМ.
EXINIT: XRA A
STA WRPAL
POP PSW
POP D
POP B
POP H
EI
RET
;
STSOUN: DW 0000H ; СЧЕТЧИК ДЛИТЕЛЬНОСТИ ЗВУКА.
WRPAL:  DB 01H   ; ПРИЗНАК ЗАПИСИ ЦВЕТОВОЙ ПАЛИТРЫ В РАБ.ОБЛАСТЬ.
;
;-----
; ОПРЕДЕЛЕНИЕ КОДОВ ЭЛЕМЕНТОВ,
; ИСПОЛЬЗУЕМЫХ В ИГРОВОМ МАССИВЕ
; ПРОГРАММЫ "BOULDER".
;-----
;
VACUUM: EQU 00 ; ПУСТОЕ МЕСТО.
WALL:    EQU 01 ; СТЕНКА.
SOIL:    EQU 02 ; СЪЕДАЕМЫЙ УЧАСТОК (ГРУНТ) .
STONE:   EQU 03 ; КАМЕНЬ.
CRYSTL:  EQU 04 ; КРИСТАЛ.
EXIT:    EQU 05 ; ВЫХОД.
EXPLOS:  EQU 06 ; ВЗРЫВ.
BUTTER:  EQU 07 ; БАБОЧКА.
MEN:     EQU 08 ; ЧЕЛОВЕЧЕК.
TRACE:   EQU 80H ; КОД СЛЕДА (ВРЕМЕННЫЙ КОД НА СТАРОМ МЕСТЕ ПОСЛЕ ПЕРЕМЕЩ.) .
;
;
;-----
;   ЗНАЧЕНИЯ ВЕКТОРОВ ЧЕЛОВЕЧКА.
;-----
;
; ВЕКТОРА НАХОДЯТСЯ В БАЙТЕ АТРИБУТОВ "ARTMEN"
;
STOP1:   EQU 00H ; СТОИТ, МОРГАЯ ГЛАЗКАМИ.
STOP2:   EQU 10H ; СТОИТ, ТОПАЯ НОЖКОЙ.
RIGHT:   EQU 20H ; БЕЖИТ ВПРАВО.
LEFT:    EQU 30H ; БЕЖИТ ВЛЕВО.
UP:      EQU 40H ; БЕЖИТ ВВЕРХ.
DOWN:    EQU 50H ; БЕЖИТ ВНИЗ.
;
; ЗНАЧЕНИЕ ВЕКТОРА РАСПОЛОЖЕНО В ТРЕХ МЛАДШИХ БИТАХ СТАРШЕЙ ТЕТРАДЫ И
; ЧТОБЫ ЗАНЕСТИ НОВОЕ ЗНАЧЕНИЕ ВЕКТОРА В АТРИБУТ, НАДО СНАЧАЛА ОБНУЛИТЬ
; ТОЛЬКО ЭТИ БИТЫ И ЗАТЕМ ТОЛЬКО В НИХ ЗАПИСАТЬ НОВЫЙ ВЕКТОР, НЕ ТРОГАЯ
; ПРИ ЭТОМ ДРУГИЕ БИТЫ БАЙТА АТРИБУТОВ.
;   ЧЕЛОВЕЧКУ ПРИНАДЛЕЖАТ ТАКЖЕ ЯЧЕЙКИ "INDMEN", В КОТОРЫХ ВСЕГДА
; ДОЛЖНЫ НАХОДИТСЯ ТЕКУЩИЕ ИНДЕКСЫ СТРОКИ И СТОЛБЦА ДЛЯ ЧЕЛОВЕЧКА В
; ИГРОВОМ МАССИВЕ 40*22.
;
;
;*****
;   НАЧАЛО ИГРЫ ПОСЛЕ ВЫХОДА ИЗ
;   ЗАСТАВКИ
;*****
;
GSTART: LXI H,FONPAL ; УСТАНОВИТЬ НЕВИДИМУЮ ПАЛИТРУ ДЛЯ БЫСТРОГО
CALL MOVRAL ; ИСЧЕЗНОВЕНИЯ СТАРОГО ИЗОБРАЖЕНИЯ.
MVI A,1     ; РАЗРЕШИТЬ ЗАПИСЬ ПАЛИТРЫ.
STA WRPAL
HLT         ; ДОЖДАТЬСЯ ПРЕРЫВАНИЯ ДЛЯ УСТАНОВКИ ПАЛИТРЫ.
CALL CLEAR ; СТЕРЕТЬ ЭКРАН.
MVI A,1     ; УСТАНОВИТЬ ПЕРВЫЙ ИГРОВОЙ УРОВЕНЬ.
STA NUMBL

```

```

LXI H,0E030H ; УСТАНОВИТЬ НАЧАЛО ВЫВОДА ИГРОВОГО (СПРАЙТОВОГО)
SHLD STWIND+1 ; ОКНА.
MVI A,0F7H ; ПЕРЕМЕСТИТЬ ЭКРАН ПО ВЕРТИКАЛИ ДЛЯ БОЛЕЕ ЛУЧШЕГО
STA SCROLL ; ВОСПРИЯТИЯ (ЧТОБЫ НЕ ПЕРЕДЕЛЫВАТЬ ВСЕ МЕТКИ).
MVI A,6 ; УСТАНОВИТЬ ПРИ СТАРТЕ 6 ПОПЫТОК.
STA LIVMEN ; (ОДНА БУДЕТ СРАЗУ УНИЧТОЖЕНА).
LXI H,1CFFH ; УСТАНОВИТЬ КОНЕЦ ОБЛАСТИ БУФЕРА ДЕМОНСТРАЦИИ
SHLD DEMKOD ; (БУФЕР КОДОВ КЛАВИАТУРЫ ДЛЯ РЕЖИМА ДЕМО).
;
; НАЧАЛО ЗАГРУЗКИ (ПЕРЕЗАГРУЗКИ) УРОВНЯ ПРИ ГИБЕЛИ ИЛИ ПЕРЕХОДЕ.
;
NSTART: LXI SP,100H ; УСТАНОВИТЬ СТЕК ПРИ ПЕРЕЗАГРУЗКЕ УРОВНЯ.
LXI H,001DH ; УСТАНОВИТЬ ИНДЕКСЫ ЗНАКОМЕСТА НА ВЫВОД
SHLD PLACE ; СООБЩЕНИЯ (УРОВЕНЬ, АЛМАЗЫ, ВРЕМЯ).
LXI H,TEXT1 ; УСТАНОВИТЬ АДРЕС НАЧАЛА ЭТОГО СООБЩЕНИЯ И
CALL TEXOUT ; ВЫВЕСТИ ЕГО.
LXI H,0004H
SHLD PLACE
LXI H,TEXT2
CALL TEXOUT
CALL AVTOR ; ВЫВЕСТИ ГРАФИЧЕСКУЮ НАДПИСЬ АВТОРА ПРОГРАММЫ.
LXI H,GAMPAL ; ЗАПИСАТЬ ИГРОВУЮ ПАЛИТРУ ЦВЕТОВ.
CALL MOVPAL
MVI A,1
STA WRPAL
LXI H,8000H ; УСТАНОВИТЬ НАЧАЛЬНУЮ ВЫСОТУ ТОНА ПРИ
SHLD VOCSPAN ; ОТРАБОТКЕ РАССОГЛАСОВАНИЯ ОКНА И ЧЕЛОВЕЧКА
XRA A ;
STA PRGMEN ; СБРОСИТЬ ПРИЗНАК ГИБЕЛИ ЧЕЛОВЕЧКА.
STA PREXIT ; СБРОСИТЬ ПРИЗНАК ВЫХОДА ИЗ ЛАБИРИНТА.
STA STSTOP ; СБРОСИТЬ СЧЕТЧИК ВРЕМЕНИ ПЕРЕКЛЮЧЕНИЯ ВЕКТ."СТОП..".
STA TON ; ВЫКЛЮЧИТЬ ЗВУК.
LXI H,0 ; ОБНУЛИТЬ СЧЕТЧИК ДЛИТЕЛЬНОСТИ ЗВУЧАНИЯ.
SHLD STSOUN ;
STA PANVER ; СБРОСИТЬ ВЕКТОР ВЕРТИКАЛЬНОГО ПАНОРАМИРОВАНИЯ.
STA PANHOR ; СБРОСИТЬ ВЕКТОР ГОРИЗОНТАЛЬНОГО ПАНОРАМИРОВАНИЯ.
MVI A,01H ; УСТАНОВИТЬ ОПЕРАНД, ДЛЯ НАЧАЛЬНОГО ОТОБРАЖЕНИЯ
STA OPCRY5 ; СЧЕТЧИКА КРИСТАЛЛОВ.
STA CHLEV ; УСТАНОВИТЬ ПРИЗНАК ЗАГРУЗКИ УРОВНЯ.
STA STRKEY ; ЗАБЛОКИРОВАТЬ КЛАВИАТУРУ.
CALL LDLEV ; ЗАГРУЗИТЬ УРОВЕНЬ.
MVI A,01 ; УСТАНОВИТЬ ЕДИНИЦУ В СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕКЛЮЧЕНИЯ
STA TIMEDL ; ВРЕМЕНИ (ПО СБРОСУ ЭТОГО СЧ. ПРОИСХОДИТ ОТОБР.ВРЕМ).
STA STSIDE ; УСТАНОВИТЬ НАПРАВЛЕНИЕ ПРОВЕРКИ СТАБИЛЬНОСТИ ЭЛЕМ.
LHLD INDVMS ; ВЗЯТЬ ИНДЕКСЫ ОКНА И
CALL INDEX ; ПОЛУЧИТЬ АДРЕС НАЧАЛА ОКНА В МАССИВЕ 40*22 И
SHLD MVIDEO ; ЗАПИСАТЬ ЕГО ДЛЯ "OUTVMS"
LHLD INDMEN ; ВЗЯТЬ ИНДЕКСЫ ЧЕЛОВЕЧКА,
CALL INDEX ; ПОЛУЧИТЬ АДРЕС ЧЕЛОВЕЧКА В МАССИВЕ 40*22 И
MVI M,EXIT ; ЗАПИСАТЬ ТУДА КОД ВХОДА (ВРЕМЕННО, ПЕРЕД ПОЯВЛ.ЧЕЛ.)
LDA ATRMEN ; ВЗЯТЬ АТРИБУТЫ ЧЕЛОВЕЧКА И
ANI 8FH ; УСТАНОВИТЬ ВЕКТОР "СТОЙ".
ORA STOP1
STA ATRMEN
;
; ВЫХОД ИГРОВОГО ОКНА НА ИСХОДНУЮ ПОЗИЦИЮ.
;
BEGIN: CALL CALCUL ; ОТОБРАЗИТЬ СЧЕТЧИКИ.
BEG1: CALL WINDOW ; ОТОБРАЗИТЬ ОКНО.
CALL OUTLIV ; ОТОБРАЗИТЬ ЖИЗНИ ЧЕЛОВЕЧКА.
CALL TSTRAN ; ПРОВЕРИТЬ РАССОГЛАСОВАНИЕ ОКНА И ЧЕЛОВЕЧКА.
CALL PANORM ; ОТРАБОТАТЬ ПАНОРАМИРОВАНИЕ ОКНА.
LHLD VOCSPAN ; ВЗЯТЬ ВЫСОТУ ТОНА ПРИ ПАНОРАМИРОВАНИИ
MOV A,H ; УМЕНЬШИТЬ СТАРШИЙ БАЙТ.
SUI 4
MOV H,A
SHLD VOCSPAN ; ЗАПИСАТЬ В ЯЧЕЙКИ ВЫСОТЫ ТОНА ПАНОРАМИРОВАНИЯ.
MVI A,TONPAN ; ПАРАМЕТРЫ ТОНА ПАНОРАМИРОВАНИЯ (УСТРАН. РАССОГЛАСС.)
STA TON
HLT ; ЗАДЕРЖКА ЦИКЛА, ДЛЯ ЗАМЕДЛЕНИЯ.
MOVTO1: HLT

```

```

      HLT
      CALL TSTRAN ; ЕЩЕ РАЗ УСТАНОВИТЬ ВЕКТОРА ПАНОРАМИРОВАНИЯ.
      LDA PANVER ; ВЗЯТЬ ВЕКТОРА ПАНОРАМИРОВАНИЯ И
      MOV B,A
      LDA PANHOR
      ORA B ; ОПРЕДЕЛИТЬ: РАССОГЛАСОВАНИЕ ОТРАБОТАНО (ЕСТЬ СОВМЕЩ?)
      JNZ BEG1
      MVI B,12 ; ЗАДЕРЖКА ДО ПОЯВЛЕНИЯ ЧЕЛОВЕЧКА.
BEG2:  PUSH B
      CALL WINDOW
      CALL OUTLIV
      HLT
MOVTO2: HLT
      HLT
      POP B
      DCR B
      JNZ BEG2
      LHLD INDMEN ; ВЗЯТЬ ИНДЕКСЫ ПОЯВЛЕНИЯ ЧЕЛОВЕЧКА.
      CALL INDEX ; ОПРЕДЕЛИТЬ АДРЕС В МАССИВЕ
      MVI M,EXPLOS ; ЗАПИСАТЬ КОД ВЗРЫВА ПЕРЕД ПОЯВЛЕНИЕМ ЧЕЛОВЕЧКА.
      MVI A,TIMEXP ; ВРЕМЯ ВЗРЫВА ПЕРЕД ПОЯВЛЕНИЕМ ЧЕЛОВЕЧКА.
      STA STEXPL
      MVI A,TONEXP ; ЗВУК ВЗРЫВА
      STA TON
      MVI A,MEN ; ПОСЛЕ ВЗРЫВА ПОЯВИТСЯ ЧЕЛОВЕЧЕК.
      STA BUFEXP
      XRA A ; РАЗБЛОКИРОВАТЬ КЛАВИАТУРУ.
      STA STRKEY
;
;
; *****
; ОСНОВНОЙ ЦИКЛ ИГРЫ - ЦИКЛИЧЕСКИЙ ВЫЗОВ
; ВЫЗОВ ГЛАВНЫХ МОДУЛЕЙ.
; *****
;
;
LOOP:  CALL WINDOW ; ОТОБРАЗИТЬ ИГРОВОЕ ОКНО (ВИДЕОМАССИВ).
      HLT ; ЖДАТЬ ПРЕР. ПЕРЕД ОПР. КЛАВИШ, ИНАЧЕ НЕЧЕТКАЯ РЕАКЦ.
      CALL KEYWOR ; ОПРОС КЛАВИАТУРЫ И УСТАНОВКА ВЕКТОРОВ ЧЕЛОВЕЧКА.
      CALL MOVMEN ; ПРОВЕРКИ И ПЕРЕМЕЩЕНИЕ ЧЕЛОВЕЧКА В МАССИВЕ.
      CALL MOVELM ; ПРОВЕРКИ И ПЕРЕМЕЩЕНИЕ ЭЛЕМЕНТОВ В МАССИВЕ.
      CALL TSTRAN ; ПРОВЕРКА ПАНОРАМИРОВАНИЯ ОКНА.
      CALL PANORM ; ПАНОРАМИРОВАНИЕ ОКНА (ЕСЛИ УСТАНОВЛЕНЫ ВЕКТОРА).
      CALL CALCUL ; ПРОВЕРКА, ОБРАБОТКА И ВЫВОД СЧЕТЧИКОВ.
      CALL OUTLIV ; ВЫВОД В ГРАФИКЕ КОЛ-ВА ЖИЗНЕЙ ЧЕЛОВЕЧКА.
      JMP LOOP ; ЦИКЛИЧЕСКИ ОПРАШИВАЯ МОДУЛИ ИМИТИРОВАТЬ ПАРАЛЛ. ПРОЦ.
;
; *****
; ОТОБРАЖЕНИЕ ИГРОВОГО ОКНА (ВИДЕОМАССИВА)
; В ОСНОВНОМ ЦИКЛЕ ИГРЫ.
; *****
;
; ДЛЯ ОТОБРАЖЕНИЯ ОКНА ИСПОЛЬЗУЮТСЯ СТАНДАРТНЫЕ МОДУЛИ
; ВИДЕОИНТЕРФЕЙСА. НАДО ЛИШЬ ЗАДАТЬ АБСОЛЮТНЫЙ АДРЕС
; ДЛЯ ОКНА В ОСНОВНОМ МАССИВЕ. (ЛЕВЫЙ НИЖНИЙ УГОЛ УЧАСТКА МАССИВА).
;
WINDOW: LHLD INDVMS ; ВЗЯТЬ ИНДЕКСЫ ОКНА (ВИДЕОМАССИВА)
      CALL INDEX ; ОПРЕДЕЛИТЬ АБСОЛЮТНЫЙ АДРЕС ОКНА.
      SHLD MVIDEO ; ЗАДАТЬ ЕГО ДЛЯ ОТОБРАЖЕНИЯ.
      CALL VMSOUT ; ВЫВЕСТИ ВИДЕОМАССИВ НА ЭКРАН.
      RET
;
;
; *****
; ПОДПРОГРАММА ЗАГРУЗКИ ИГРОВОГО УРОВНЯ
; И ПАРАМЕТРОВ УРОВНЯ.
; *****
;
; КАЖДЫЙ УРОВЕНЬ В ИГРЕ "BOULDER" - ЭТО ДВУХМЕРНЫЙ МАССИВ РАЗМЕРОМ
; 40 НА 22 БАЙТ СОДЕРЖАЩИЙ КОДЫ ЭЛЕМЕНТОВ ОДНОГО ЛАБИРИНТА.
; МАССИВЫ РАСПОЛОЖЕНЫ В ПАМЯТИ ДРУГ ЗА ДРУГОМ БЕЗ РАЗРЫВОВ, В ПО-

```

```

; РЯДКЕ ВОЗРАСТАНИЯ НОМЕРОВ. НАЧАЛО БЛОКА МАССИВОВ ОПРЕДЕЛЯЕТСЯ
; МЕТКОЙ "LEVEL". ДАННАЯ П/П ПО НОМЕРУ УРОВНЯ ОТЫСКИВАЕТ НУЖНЫЙ
; МАССИВ И ПЕРЕСЫЛАЕТ В РАБОЧУЮ ОБЛАСТЬ - ДВУХМЕРНЫЙ МАССИВ 40*22
; С НАЧАЛЬНОЙ МЕТКОЙ "BAZA", ГДЕ И ИДЕТ ПРОЦЕСС ИГРЫ.
; КРОМЕ 2-Х МЕРНОГО МАССИВА 40*22 КАЖДЫЙ УРОВЕНЬ ИМЕЕТ ЕЩЕ 12 БАЙТ
; ПАРАМЕТРОВ УРОВНЯ. ПАРАМЕТРЫ ВСЕХ УРОВНЕЙ ТАКЖЕ СОБРАНЫ В УПОРЯДОЧЕН-
; НЫЙ МАССИВ С НАЧАЛЬНОЙ МЕТКОЙ "PARLEV". ДАННАЯ П/П ПО НОМЕРУ УРОВНЯ
; РАЗЫСКИВАЕТ НУЖНУЮ ГРУППУ ПАРАМЕТРОВ И УСТАНАВЛИВАЕТ ИХ В РАБОЧИЕ
; ЯЧЕЙКИ.
; ПОРЯДОК РАСПОЛОЖЕНИЯ ПАРАМЕТРОВ ОДНОГО УРОВНЯ ТАКОВ:
;
; N БАЙТА    ЗНАЧЕНИЕ ПАРАМЕТРА.
; -----
; 0          ИНДЕКС СТРОКИ ПОЯВЛЕНИЯ ЧЕЛОВЕЧКА ВНАЧАЛЕ УРОВНЯ.
; 1          ИНДЕКС СТОЛБЦА >> >>.
; 2          ИНДЕКС СТРОКИ ОКНА (РАССОГЛАСОВАНИЕ) ВНАЧАЛЕ УРОВНЯ.
; 3          ИНДЕКС СТОЛБЦА >> >>.
; 4          ИНДЕКС СТРОКИ ВЫХОДА ИЗ УР. (ПОЯВИТСЯ ПОСЛЕ СЪЕД.ВСЕХ КРИСТ.).
; 5          ИНДЕКС СТОЛБЦА >> >>.
; 6          КОЛИЧЕСТВО КРИСТАЛЛОВ КОТОРОЕ НУЖНО СЪЕСТЬ, ЧТОБЫ ПОЯВИЛСЯ ВЫХ.
; 7          ВРЕМЯ, В ТЕЧЕНИИ КОТОРОГО НУЖНО УСПЕТЬ СЪЕСТЬ КРИСТАЛЛЫ.
; 8          ЦВЕТ ИЗОБРАЖЕНИЯ УРОВНЯ В МОНОХРОМЕ (ДЛЯ РЕЖИМА "ТУРБО").
; 9          ЦВЕТ 1, ДЛЯ ИЗОБРАЖЕНИЯ УРОВНЯ В РЕЖИМЕ "ЦВЕТ".
; 10         ЦВЕТ 2 >> >>.
; 11         ЦВЕТ 3 >> >>.
;
LDLEV:  LDA CHLEV    ; ВЗЯТЬ ПРИЗНАК ЗАГРУЗКИ (СМЕНЫ) УРОВНЯ И ПРОВЕРИТЬ:
        ORA A        ; ПРИЗНАК УСТАНОВЛЕН?
        RZ           ; ЕСЛИ НЕТ - ВЫЙТИ.
        PUSH H       ; ЕСЛИ ДА - СОХРАНИТЬ РЕГИСТРЫ.
        PUSH B
        PUSH D
        PUSH PSW
        IN 01        ; ПЕРЕКЛЮЧЕНИЕ ЦВЕТ/МОНО/ТУРБО.
        ANI 80H      ; ПО РУС/ЛАТ - ТУРБО
        JNZ TSTRCL
        STA TURBO
        JMP LDL01
TSTRCL: IN 01
        ANI 40H
        JNZ LDL03
        MVI A,02
        STA TURBO
        JMP LDL01
LDL03:  IN 01
        ANI 20H
        JNZ LDL01
        MVI A,1
        STA TURBO
LDL01:  MVI A,01     ; УСТАНОВИТЬ ПРИЗНАК ДЛЯ ОТОБРАЖЕНИЯ
        STA OPLEVE   ; НОМЕРА УРОВНЯ.
        LDA NUMBL    ; ВЗЯТЬ НОМЕР УРОВНЯ.
        LXI H,LEVEL  ; В "HL" АДРЕС НАЧАЛА МАССИВОВ УРОВНЕЙ.
        LXI B,880    ; В "BC" РАЗМЕР МАССИВА ОДНОГО УРОВНЯ.
LDLEV1: ORA A        ; УРОВЕНЬ НУЛЕВОЙ? (ИЛИ АДРЕС НАЙДЕН?).
        JZ LDLEV2    ; ЕСЛИ ДА - ЗАГРУЗИТЬ.
        DAD B        ; ЕСЛИ НЕТ - ПЕРЕЙТИ НА АДРЕС НАЧАЛА СЛЕД.УРОВНЯ.
        DCR A        ; УМЕНЬШИТЬ СЧЕТЧИК НОМЕРА УРОВНЯ.
        JMP LDLEV1   ; ПРОВЕРИТЬ - НАЙДЕН ЛИ УРОВЕНЬ?
LDLEV2: LXI D,BAZA   ; В "DE" - АДРЕС РАБОЧЕГО МАССИВА.
LDLEV3: MOV A,M      ; ПЕРЕСЛАТЬ УРОВЕНЬ В РАБОЧИЙ МАССИВ.
        STAX D
        INX H
        INX D
        DCX B
        MOV A,B
        ORA C
        JNZ LDLEV3
        LDA NUMBL    ; СНОВА ВЗЯТЬ НОМЕР УРОВНЯ ДЛЯ ПОИСКА ПАРАМЕТРОВ.
        LXI H,PARLEV  ; В "HL" АДРЕС НАЧАЛА МАССИВОВ ПАРАМЕТРОВ УРОВНЕЙ.
        LXI B,12     ; В "BC" РАЗМЕР БЛОКА ПАРАМЕТРОВ ДЛЯ ОДНОГО УРОВНЯ.

```

```

LDLEV4: ORA A          ; ПОИСК ПАРАМЕТРОВ АНАЛОГИЧЕН ПОИСКУ МАССИВА.
        JZ LDLEV5      ; ЕСЛИ БЛОК ПАРАМ. НАЙДЕН - ЗАГРУЖАТЬ ПАРАМЕТРЫ.
        DAD B
        DCR A
        JMP LDLEV4
LDLEV5: MOV E,M        ; ВЗЯТЬ ИНДЕКСЫ ПОЯВЛЕНИЯ ЧЕЛОВЕЧКА В УРОВНЕ
        INX H
        MOV D,M
        INX H
        XCHG
        SHLD INDMEN ; И ЗАПИСАТЬ В "INDMEN".
        XCHG
        MOV E,M      ; ВЗЯТЬ ИНДЕКСЫ ОКНА (НАЧАЛЬНОЕ РАССОГЛАСОВАНИЕ).
        INX H
        MOV D,M
        INX H
        XCHG
        SHLD INDVMS ; И ЗАПИСАТЬ В "INDVMS".
        XCHG
        MOV E,M      ; ВЗЯТЬ ИНДЕКСЫ ПОЯВЛЕНИЯ ВЫХОДА, ПОСЛЕ СЪЕДАНИЯ
        INX H        ; НЕОБХОДИМОГО КОЛИЧЕСТВА КРИСТАЛЛОВ.
        MOV D,M
        INX H
        XCHG
        SHLD INDEXT ; И ЗАПИСАТЬ В "INDEXT".
        XCHG
        MOV A,M      ; ВЗЯТЬ ЧИСЛО КРИСТАЛЛОВ, КОТОРОЕ НУЖНО СЪЕСТЬ
        INR A        ; ПРИБАВИТЬ ОДИН, ТАК КАК ОН ПОТЕРЯЕТСЯ ПРИ ВЫВОДЕ.
        STA STCRYS   ; И ЗАПИСАТЬ В СЧЕТЧИК КРИСТАЛЛОВ.
        INX H
        MOV A,M      ; ВЗЯТЬ ВРЕМЯ ПРОХОДА УРОВНЯ
        INR A        ; ПРИБАВИТЬ ОДИН, (КОМПЕНСИРОВАТЬ ПОТЕРЮ ПРИ ВЫВОДЕ)
        STA TIME     ; И ЗАПИСАТЬ В СЧЕТЧИК ВРЕМЕНИ.
        INX H
        LDA TURBO    ; ВЗЯТЬ ПРИЗНАК ТУРБО И ЕСЛИ 00H - ЗАПИСАТЬ 1 ЦВЕТ,
        ORA A        ;
        JZ COLTUR
        CPI 1        ; ЕСЛИ 1 - ЦВЕТНОЙ РЕЖИМ.
        JZ CWET
        MOV A,M      ; ЕСЛИ БОЛЬШЕ 1, ТО РЕЖИМ MONO.
        STA COLR1
        STA COLR2
        STA COLR3
        JMP CWETMN
CWET:   INX H
        MOV A,M      ; ЗАПИСАТЬ ТРИ ЦВЕТА ДЛЯ РЕЖИМА "ЦВЕТ".
        STA COLR1
        INX H
        MOV A,M
        STA COLR2
        INX H
        MOV A,M
        STA COLR3
CWETMN: MVI A,2      ; УСТАНОВИТЬ РЕЖИМ "ЦВЕТ", ВКЛЮЧИВ 2 ПЛОСКОСТИ
        STA STPLOS   ; ДЛЯ ОТОБРАЖЕНИЯ ГРАФИЧЕСКИМ ИНТЕРФЕЙСОМ ИГРЫ.
        STA COLSOI+1 ; ЦВЕТ 2 ДЛЯ СЪЕДАЕМОГО УЧАСТКА
        LXI H,0000H ; ЗАПИСАТЬ КОД КОМАНДЫ "NOR" , ЧТОБЫ УБРАТЬ
        SHLD MOVTO1 ; ЗАДЕРЖКУ ДВИЖЕНИЯ ОКНА К ИСХОДНОЙ ПОЗИЦИИ В
        SHLD MOVTO2 ; НАЧАЛЕ УРОВНЯ ПРИ ЦВЕТНОМ РЕЖИМЕ.
        MVI A,10H   ; ЗАПИСАТЬ ВРЕМЯ ПОКАЗА ЗАПОЛНЕННОГО ЭКРАНА В
        STA TIMSHO+1 ; КОНЦЕ УРОВНЯ В ЦВЕТНОМ РЕЖИМЕ.
        JMP EXLDLV  ; ВЫЙТИ ИЗ П/П ЗАГРУЗКИ УРОВНЯ.
COLTUR: MOV A,M      ; ЗАПИСАТЬ ОДИН ЦВЕТ ДЛЯ РЕЖИМА "ТУРБО".
        STA COLR1
        MVI A,1      ; УСТАНОВИТЬ РЕЖИМ "ТУРБО", ВКЛЮЧИВ ТОЛЬКО ОДНУ
        STA STPLOS   ; ПЛОСКОСТЬ.
        STA COLSOI+1 ; ЦВЕТ 1 ДЛЯ СЪЕДАЕМОГО УЧАСТКА.
        LXI H,7676H ; ЗАПИСАТЬ КОД КОМАНДЫ "HLT" ДЛЯ ЗАДЕРЖКИ ДВИЖЕНИЯ
        SHLD MOVTO1 ; ОКНА К ИСХОДНОЙ ПОЗИЦИИ В НАЧАЛЕ УРОВНЯ.
        SHLD MOVTO2
        MVI A,20H   ; ЗАПИСАТЬ ВРЕМЯ ПОКАЗА ЗАПОЛНЕННОГО ЭКРАНА ПРИ

```

```

    STA TIMSHO+1      ; "ТУРБО".
EXLDELV: MVI A,1
    STA WRPAL
    POP PSW
    POP D
    POP B
    POP H
    RET
;
LEVEL: EQU 4800H      ; НАЧАЛО 2-Х МЕРНЫХ МАССИВОВ УРОВНЕЙ.
PARLEV: EQU 4700H     ; НАЧАЛО ГРУПП ПАРАМЕТРОВ МАССИВОВ.
CHLEV: DB 00H         ; ПРИЗНАК ЗАГРУЗКИ (СМЕНЫ) УРОВНЯ.
NUMBL: DB 00H         ; НОМЕР УРОВНЯ
INDEXT: DW 0000H      ; ИНДЕКСЫ ПОЯВЛЕНИЯ ВЫХОДА.
TURBO: DB 00H         ; ПЕРЕКЛЮЧАТЕЛЬ "ТУРБО"/"ЦВЕТ".
;
; *****
; ПОДПРОГРАММА УСТАНОВКИ ВЕКТОРА (НАПРАВЛЕНИЯ ДВИЖЕНИЯ)
; ЧЕЛОВЕЧКА ПО КОДУ НАЖАТОЙ КЛАВИШИ УПРАВЛЕНИЯ.
; *****
;
; П/П УСТАНОВЛИВАЕТ ВЕКТОР В БАЙТЕ АТРИБУТОВ ЧЕЛОВЕЧКА
; "ATRMEN". ЕСЛИ НИ ОДНА ИЗ КЛАВИШ УПРАВЛЕНИЯ ЧЕЛОВЕЧКОМ
; НЕ НАЖАТА, ПРОИСХОДИТ ПЕРЕКЛЮЧЕНИЕ ВЕКТОРОВ "STOP1" И "STOP2"
; ЧЕРЕЗ ПРОМЕЖУТОК ВРЕМЕНИ, ОПРЕДЕЛЯЕМЫЙ "TIMSTR" (СТОИТ,
; МОРГАЯ ГЛАЗКАМИ, А ЗАТЕМ ТОПАЕТ НОЖКОЙ). ПРИ ВЗРЫВАХ (ГИБЕЛИ
; ЧЕЛОВЕЧКА) КЛАВИАТУРА БЛОКИРУЕТСЯ ПРИЗНАКОМ "STRKEY".
;
KEYBOR: LDA TURBO      ; В РЕЖИМЕ ТУРБО МОЖНО УПРАВЛЯТЬ ЧЕЛОВЕЧКОМ ПРИ
    ORA A              ; ПАНОРАМИРОВАНИИ ЭКРАНА. А В ДРУГИХ РЕЖИМАХ
    JZ KEYBR1          ; УПРАВЛЕНИЕ ВОЗМОЖНО ЛИШЬ ПРИ НУЛЕВЫХ
    LDA PANHOR         ; ЗНАЧЕНИЯХ ВЕКТОРОВ ПАНОРАМИРОВАНИЯ (КОГДА ЕГО НЕТ).
    MOV B,A
    LDA PANVER
    ORA B
    RNZ
KEYBR1: LDA STRKEY      ; ПРОВЕРИТЬ БЛОКИРОВКУ.
    ORA A
    RNZ                ; ЕСЛИ НЕ НОЛЬ - КЛАВИАТУРА ЗАБЛОКИРОВАНА.
    LDA DEMO           ; ПРОВЕРИТЬ ПРИЗНАК РЕЖИМА ДЕМОНСТРАЦИИ.
    ORA A              ; ЕСЛИ НЕ ДЕМО, ТО ПЕРЕЙТИ К УПРАВЛЕНИЮ
    JZ KEYUPR          ; ОТ КЛАВИАТУРЫ.
    LDA STATUS         ; ЕСЛИ ДЕМО, ТО ПРОВЕРИТЬ НАЖАТИЕ НА
    CPI 0FFH          ; ЛЮБУЮ КЛАВИШУ МАТРИЦЫ 8*8 И ЕСЛИ НЕ
    JZ DEMUP1          ; НАЖАТА НИ ОДНА - УПРАВЛЯТЬ ИЗ ФАЙЛА КОДОВ ДЕМО.
    POP PSW            ; ЕСЛИ В ПРОЦЕССЕ ИГРЫ НАЖАТЬ КЛАВИШУ, ТО
    JMP START          ; СРАЗУ ПЕРЕЙТИ К ЗАСТАВКЕ.
DEMUP1: LHLD DEMKOD     ; ВЗЯТЬ УКАЗАТЕЛЬ В ФАЙЛЕ КОДОВ ДЕМО.
    MOV A,M            ; ПОЛУЧИТЬ КОД УПРАВЛЕНИЯ ЧЕЛОВЕЧКОМ,
    DCX H              ; ПЕРЕДВИНУТЬ УКАЗАТЕЛЬ И
    SHLD DEMKOD        ; ЗАПОМНИТЬ.
    JMP DEMUPR          ; ОБОЙТИ ПОЛУЧЕНИЕ КОДА С КЛАВИАТУРЫ.
KEYUPR: LDA KEYKOD      ; ВЗЯТЬ КОД НАЖАТОЙ КЛАВИШИ.
DEMUPR: CPI 0FFH        ; КЛАВИША НАЖАТА?
    JZ SETSTR          ; ЕСЛИ НЕТ - НА УСТАНОВКУ ВЕКТОРОВ "STOP.."
    CPI TAB            ; НАЖАТА ТАБУЛЯЦИЯ ?
    JZ SETEX1          ; САМОУНИЧТОЖЕНИЕ ЧЕЛОВЕЧКА.
    CPI CR             ; НАЖАТА "БК" ?
    JZ START           ; НА НАЧАЛО ИГРЫ.
    PUSH PSW           ; СОХРАНИТЬ КОД НАЖАТОЙ КЛАВИШИ.
    MVI A,STOP1        ; УСТАНОВИТЬ КОД ВЕКТОРА "STOP1" В
    STA BFSTOP         ; БУФЕРЕ ВЕКТОРОВ "STOP.."
    MVI A,TIMSTR        ; УСТАНОВИТЬ ВРЕМЯ ПЕРЕКЛЮЧЕНИЯ ВЕКТОРОВ В
    STA STSTOP         ; СЧЕТЧИК ВРЕМЕНИ ПЕРЕКЛЮЧЕНИЯ.
    POP PSW            ; ВЕРНУТЬ КОД НАЖАТОЙ КЛАВИШИ.
    CPI UPK            ; КЛАВИША "ВВЕРХ"?
    JZ SETUP           ; УСТАНОВИТЬ ВЕКТОР "ВВЕРХ" (UP).
    CPI DOWNK          ; КЛАВИША "ВНИЗ"?
    JZ SETDOW          ;
    CPI RIGHTK         ; КЛАВИША "НАПРАВО"?
    JZ SETRIG          ;

```



```

CPI LEFTK      ; КЛАВИША "НАЛЕВО"?
JZ SETLEF      ;
JMP SETSTP     ; ДРУГИЕ КЛАВИШИ НЕ ПРОВЕРЯТЬ.
SETUP: MVI B,UP      ; УСТАНОВИТЬ В РЕГ. "В" КОД ВЕКТОРА "UP".
JMP SETVEK     ; ПЕРЕЙТИ НА ЗАПИСЬ В АТРИБУТ ЧЕЛОВЕЧКА.
SETDOW: MVI B,DOWN  ;
JMP SETVEK
SETRIG: MVI B,RIGHT
JMP SETVEK
SETLEF: MVI B,LEFT
JMP SETVEK
SETSTP: LDA STSTOP  ; ВЗЯТЬ СЧЕТЧИК ВРЕМЕНИ ПЕРЕКЛЮЧЕНИЯ ВЕКТОРОВ "STOP.."
DCR A          ; УМЕНЬШИТЬ И
STA STSTOP     ; СОХРАНИТЬ ЕГО.
LDA BFSTOP     ; ВЗЯТЬ ЗНАЧЕНИЕ ИЗ БУФЕРА ВЕКТОРОВ "STOP.."
MOV B,A        ; И ПОМЕСТИТЬ ЕГО В "В".
JNZ SETVEK     ; ЕСЛИ ВРЕМЯ ВЕКТОРА НЕ КОНЧИЛОСЬ - УСТАНОВИТЬ ЕГО.
XRI 10H        ; ИЗМЕНИТЬ ВЕКТОР НА ПРОТИВОПОЛОЖНЫЙ.
STA BFSTOP     ; СОХРАНИТЬ ЕГО В БУФЕРЕ И
MOV B,A        ; ПОМЕСТИТЬ В "В".
MVI A,TIMSTR   ; И УСТАНОВИТЬ ДЛЯ НЕГО ВРЕМЯ.
STA STSTOP     ;
SETVEK: LDA ATRMEN ; ВЗЯТЬ БАЙТ АТРИБУТОВ И
ANI 8FH        ; ОБНУЛИТЬ ТРИ БИТА ВЕКТОРА.
ORA B          ; УСТАНОВИТЬ НОВЫЙ ВЕКТОР И
STA ATRMEN     ; ЗАПИСАТЬ БАЙТ АТРИБУТОВ С НОВЫМ ВЕКТОРОМ.
RET           ; ВЫЙТИ ИЗ П/П УСТАНОВКИ ВЕКТОРА.
;
TIMSTR: EQU 12      ; ВРЕМЯ ПЕРЕКЛЮЧЕНИЯ ВЕКТОРОВ "STOP.."
STSTOP: DB 00H      ; СЧЕТЧИК ВРЕМЕНИ ПЕРЕКЛЮЧЕНИЯ ВЕКТОРОВ "STOP.."
BFSTOP: DB 00H      ; БУФЕР ВЕКТОРОВ "STOP.."
STRKEY: DB 00H      ; ПРИЗНАК БЛОКИРОВКИ КЛАВИАТУРЫ.
DEMKOD: DW 0000H    ; УКАЗАТЕЛЬ В МАССИВЕ КОДОВ ДЕМОНСТРАЦИИ
;
;*****
; ПЕРЕМЕЩЕНИЕ И ПРОВЕРКИ ЧЕЛОВЕЧКА В МАССИВЕ
;*****
;
; ПОДПРОГРАММА ОПРЕДЕЛЯЕТ НОВОЕ МЕСТО ЧЕЛОВЕЧКА
; В МАССИВЕ ПО ЕГО ВЕКТОРУ И ПРОВЕРЯЕТ, МОЖНО
; ЛИ ЕГО ТУДА ПЕРЕМЕСТИТЬ. ЕСЛИ МОЖНО, ТО КОД
; ЧЕЛОВЕЧКА УДАЛЯЕТСЯ СО СТАРОГО МЕСТА И ЗАПИСЫ-
; ВАЕТСЯ НА НОВОЕ. ПРИ ЭТОМ ПОДПРОГРАММА ДЕЛАЕТ
; ДОПОЛНИТЕЛЬНЫЕ ПРОВЕРКИ КОДОВ ЭЛЕМЕНТОВ, КОТО-
; РЫЕ БЫЛИ НА ЭТОМ МЕСТЕ И ИЗМЕНЯЕТ СЧЕТЧИКИ,
; ПРИЗНАКИ И ВКЛЮЧАЕТ ЗВУК ПЕРЕМЕЩЕНИЯ ЧЕЛОВЕЧКА
; С ВЫСОТОЙ ТОНА, ЗАВИСЯЩИМ ОТ КОДА "СЪЕДЕННОГО"
; ЭЛЕМЕНТА МАССИВА.
;
MOV MEN: LDA TURBO  ; В РЕЖИМЕ ТУРБО ЧЕЛОВЕЧКА МОЖНО ПЕРЕМЕЩАТЬ ВСЕГДА.
ORA A
JZ MOV M00
LDA PANVER        ; В ДРУГИХ РЕЖИМАХ - ТОЛЬКО ПРИ ОТСУТСТВИИ
MOV B,A           ; ПАНОРАМИРОВАНИЯ ОКНА.
LDA PANHOR
ORA B
RNZ
MOV M00: LHLD INDMEN ; ВЗЯТЬ ИНДЕКСЫ ЧЕЛОВЕЧКА В МАССИВЕ.
LDA ATRMEN        ; ВЗЯТЬ БАЙТ АТРИБУТОВ И
ANI 70H           ; ВЫДЕЛИТЬ ВЕКТОР.
CPI 20H           ; ПРОВЕРИТЬ: ЭТО "STOP1" ИЛИ "STOP2"?
JNC CONTIN        ; ЕСЛИ НЕТ - ПРОДОЛЖАТЬ.
MVI A,TIMPUS      ; ЕСЛИ ДА, УСТАНОВИТЬ ВРЕМЯ ТОЛКАНИЯ КАМНЯ В СЧЕТЧИКЕ
STA STRUSH        ; ВР.ТОЛКАНИЯ, ЧТОБЫ ПОСЛЕ ПЕРЕРЫВА В ТОЛКАНИИ КАМНЯ,
RET               ;ПРИШЛОСЬ ОПЯТЬ ТРАТИТЬ ТАКОЕ ЖЕ ВРЕМЯ ЧТОБЫ ТОЛКНУТЬ.
CONTIN: MOV C,A    ; СОХРАНИТЬ ВЫДЕЛЕННЫЙ ВЕКТОР В РЕГ. "С".
CALL TESTVM       ; ПОЛУЧИТЬ КОД ЭЛ. ПО ВЕКТОРУ И НОВЫЕ ИНД. В "DE".
CPI WALL          ; ЭТО СТЕНА?
RZ                ; ЕСЛИ ДА - ВЫЙТИ ИЗ "MOV MEN".
ORA A             ; ЭТО ПУСТОЕ МЕСТО?
JZ GOVACM        ; ЕСЛИ ДА - ПЕРЕЙТИ НА НЕГО.

```

```

CPI SOIL      ; ЭТО СЪЕДАЕМЫЙ УЧАСТОК (ГРУНТ)?
JZ GOSOIL     ; ПЕРЕЙТИ НА НЕГО.
CPI CRYSTL    ; ЭТО КРИСТАЛЛ?
JZ GOCRYS     ; "СЪЕСТЬ" ЕГО!
CPI EXIT      ; ЭТО ВЫХОД ИЗ ЛАБИРИНТА?
JZ GOEXIT     ; ПЕРЕЙТИ НА ЕГО МЕСТО И УСТАНОВИТЬ ПРИЗНАКИ.
CPI STONE     ; ЭТО КАМЕНЬ?
JZ TSTONE     ; ЕСЛИ ДА - ПРОВЕРИТЬ: МОЖНО ЛИ ЕГО "ТОЛКНУТЬ".
CPI EXPLOS    ; ЭТО ВЗРЫВ?
JZ SETEX1     ; ВЗОРВАТЬ ЧЕЛОВЕЧКА ТОЖЕ.
CPI BUTTER    ; ЭТО БАБОЧКА?
JZ SETEXP     ; ВЗОРВАТЬ ЧЕЛОВЕЧКА И БАБОЧКУ ПРИ ВСТРЕЧЕ С БАБОЧКОЙ.
RET          ; ВЫЙТИ ИЗ "MOV MEN" НЕ ПРОВЕРЯЯ ДРУГИЕ КОДЫ.
;
GOVACM: MVI A,TONVAC      ; УСТАНОВИТЬ ТОН ЗВУКА ПРИ ПЕРЕХ. НА ПУСТ.МЕСТО
      STA TON            ; (ЗАПИСАТЬ В ЯЧЕЙКУ "TON").
      JMP SETMEN         ; ПЕРЕЙТИ НА ПЕРЕСТАНОВКУ ЧЕЛ. В МАССИВЕ.
GOSOIL: MVI A,TONSOL      ; УСТАНОВИТЬ ЗВУК СЪЕДАНИЯ ГРУНТА.
      STA TON
      JMP SETMEN
GOCRYS: MVI A,TONCRS      ; УСТАНОВИТЬ ЗВУК СЪЕДАНИЯ КРИСТАЛЛА.
      STA TON
      MVI A,81H          ; УСТАНОВИТЬ СЛАГАЕМОЕ (-1) ДЛЯ СЧЕТЧ.КРИСТАЛЛОВ.
      STA OPCRYS        ; (ЗАПИСАТЬ МИНУС ОДИН В ОПЕРАНД ДЛЯ СЧЕТЧ.КРИСТ.).
      JMP SETMEN
GOEXIT: MVI A,01          ; УСТАНОВИТЬ ПРИЗНАК ВЫХОДА ИЗ ЛАБИРИНТА.
      STA PREXIT        ; (ЛОГ. "1" В ЯЧЕЙКУ "PREXIT").
      JMP SETMEN
SETEXP: MVI M,EXPLOS      ; ЗАПИСАТЬ НА МЕСТО БАБОЧКИ КОД ВЗРЫВА.
SETEX1: LHLD INDMEN       ; ВЗЯТЬ СТАРЫЕ ИНДЕКСЫ ЧЕЛОВЕЧКА,
      CALL INDEX        ; ПОЛУЧИТЬ АДРЕС СТАРОГО МЕСТА В МАССИВЕ И
      MVI M,EXPLOS      ; ЗАПИСАТЬ КОД ВЗРЫВА.
      MVI A,CRYSTL      ; ПОСЛЕ ВЗРЫВА БАБОЧКИ И ЧЕЛОВЕЧКА НА ИХ МЕСТЕ
      ORI 80H           ; БУДЕТ КОД КРИСТАЛЛА С ПРИЗНАКОМ ПЕРЕСТАНОВКИ.
      STA BUFEHP        ; (ОН БУДЕТ ВЗЯТ ИЗ ЯЧЕЙКИ "BUFEHP").
      MVI A,TONEXP      ; УСТАНОВИТЬ ЗВУК ВЗРЫВА.
      STA TON
      MVI A,TIMEXP      ; УСТАНОВИТЬ ВРЕМЯ ВЗРЫВА В
      STA STEXP        ; СЧЕТЧИК ВРЕМЕНИ ВЗРЫВА.
      MVI A,01          ; УСТАНОВИТЬ ПРИЗНАК ГИБЕЛИ ЧЕЛОВЕЧКА ( ЛОГ "1" В
      STA PRGMEN       ; ЯЧЕЙКУ "PRGMEN").
      RET
TSTONE: MOV A,C          ; ВЗЯТЬ ВЫДЕЛЕННЫЙ КОД ВЕКТОРА ИЗ РЕГ. "C".
      CPI UP            ; ЭТО ВЕКТОР "ВВЕРХ"?
      RZ                ; ЕСЛИ ДА - ВЫЙТИ Т.К. КАМЕНЬ ЧЕЛОВЕЧКУ НЕ ПОДНЯТЬ!
      PUSH H            ; СОХРАНИТЬ ИНДЕКСЫ И АДРЕС.
      PUSH D
      XCHG              ; ПОМЕСТИТЬ ИНДЕКСЫ КАМНЯ В "HL".
      CALL TESTVM       ; ПОЛУЧИТЬ КОД ЭЛЕМЕНТА "ЗА КАМНЕМ" ПО ВЕКТОРУ.
      ORA A             ; ЗА КАМНЕМ ПУСТО?
      JNZ NOPUSH        ; ЕСЛИ НЕТ - ТО КАМЕНЬ НЕ СТОЛКНУТЬ.
      LDA STPUSH        ; ЕСЛИ ДА - ПРОВЕРИТЬ СЧЕТЧИК ВРЕМЕНИ ТОЛКАНИЯ КАМНЯ
      ORA A             ; СЧЕТЧИК РАВЕН НУЛЮ?
      JZ MOVSTN         ; ЕСЛИ ДА - ТОЛКНУТЬ КАМЕНЬ!
      DCR A             ; ЕСЛИ НЕТ - УМЕНЬШИТЬ СЧ.ВРЕМЕНИ ТОЛКАНИЯ И
      STA STPUSH        ; СОХРАНИТЬ ДО СЛЕДУЮЩЕЙ ПРОВЕРКИ.
NOPUSH: POP D            ; ВЕРНУТЬ ИНДЕКСЫ И АДРЕС И
      POP H
      RET              ; ВЫЙТИ ИЗ "MOV MEN"
MOVSTN: MVI A,STONE      ; ВЗЯТЬ КОД КАМНЯ И
      ORI 80H           ; УСТАНОВИТЬ ПРИЗНАК ПЕРЕСТАНОВКИ (СМОТРИ В "MOVELM").
      MOV M,A           ; УСТАНОВИТЬ КАМЕНЬ НА НОВОЕ МЕСТО, Т.Е."ТОЛКНУТЬ" ЕГО.
      MVI A,TONMOV      ; УСТАНОВИТЬ ЗВУК ТОЛЧКА.
      STA TON
      MVI A,TIMPUS      ; УСТАНОВИТЬ ВРЕМЯ ТОЛКАНИЯ КАМНЯ НА СЛУЧАЙ, ЕСЛИ
      STA STPUSH        ; КАМЕНЬ ТОЛКАЕТСЯ ЧЕЛОВЕЧКОМ НЕПРЕРЫВНО.
      POP D             ; ВЕРНУТЬ ИНДЕКСЫ И АДРЕС ДЛЯ ПЕРЕМЕЩЕНИЯ ЧЕЛОВЕЧКА И
      POP H
      JMP SETMEN        ; ПЕРЕЙТИ НА ЕГО ПЕРЕМЕЩЕНИЕ.
;
; П/П ПОЛУЧЕНИЯ КОДА ЭЛЕМЕНТА ПО ВЕКТОРУ.

```

```

;
; ВХОД: В "A" - ВЕКТОР, "HL" - ТЕКУЩИЕ ИНДЕКСЫ.
; ВЫХОД: В "A" - КОД, "HL" - АДРЕС ЭТОГО ЭЛ. В МАССИВЕ, "DE" - ИНД. ЭТОГО ЭЛ.
;
TESTVM: CPI DOWN      ; ВЕКТОР ВНИЗ?
      JZ MOVDOW      ; ЕСЛИ ДА - ИЗМЕНИТЬ ИНДЕКСЫ ПО ВЕКТОРУ.
      CPI UP        ; ВЕКТОР ВВЕРХ?
      JZ MOVUP
      CPI RIGHT     ; ВЕКТОР НАПРАВО?
      JZ MOVRIG
      CPI LEFT      ; ВЕКТОР НАЛЕВО?
      JZ MOVLEF
      POP PSW       ; ЕСЛИ ВЕКТОР НЕОПРЕДЕЛЕН - ВЕРНУТЬ ЗНАЧЕНИЕ "SP" И
      RET          ; ВЫЙТИ ВООБЩЕ ИЗ П/П "MOVME".
;
MOVDOW: DCR L        ; УМЕНЬШИТЬ ИНДЕКС СТРОКИ И
      JMP NEWVEK    ; ПЕРЕЙТИ НА ЗАПОМИНАНИЕ НОВЫХ ИНДЕКСОВ И ОПРЕД.КОДА.
MOVUP:  INR L        ; УВЕЛИЧИТЬ ИНДЕКС СТРОКИ
      JMP NEWVEK
MOVRIG: INR H        ; УВЕЛИЧИТЬ ИНДЕКС СТОЛБЦА
      JMP NEWVEK
MOVLEF: DCR H        ; УМЕНЬШИТЬ ИНДЕКС СТОЛБЦА
NEWVEK: MOV D,H      ; ЗАПИСАТЬ НОВЫЕ ИНДЕКСЫ В "DE".
      MOV E,L
      CALL INDEX    ; ПОЛУЧИТЬ В "HL" АДРЕС ЭЛЕМЕНТА ПО ВЕКТОРУ.
      MOV A,M       ; ВЗЯТЬ КОД ЭТОГО ЭЛЕМЕНТА В "A".
      ANI 0CFH      ; ИСКЛЮЧИТЬ ИЗ КОДА БИТЫ ОПОРЫ БАБОЧКИ.
      RET          ; ВЫЙТИ ИЗ ПОДПРОГРАММЫ "TESTVM".
;
; П/П ПЕРЕМЕЩЕНИЯ ЧЕЛОВЕЧКА В МАССИВЕ СО СТАРОГО МЕСТА НА НОВОЕ.
;
; ВХОД: В "HL" - АДРЕС НОВОГО МЕСТА, В "DE" - ИНДЕКСЫ НОВОГО МЕСТА.
; НА СТАРОЕ МЕСТО БУДЕТ ЗАПИСАН КОД СЛЕДА "TRACE" - ЭТО ВРЕМЕН-
; НЫЙ КОД, НЕ ОТОБРАЖАЕМЫЙ НА ЭКРАНЕ. ОН НУЖЕН ЛИШЬ НА ВРЕМЯ ПРОВЕРКИ
; МАССИВА ПОСЛЕ ПЕРЕМЕЩЕНИЯ ЧЕЛОВЕЧКА, ЧТОБЫ СТАРОЕ МЕСТО ЧЕЛОВЕЧКА НЕ
; БЫЛО МГНОВЕННО ЗАНЯТО. ПОСЛЕ ПРОВЕРКИ МАССИВА НА МЕСТО СЛЕДА ДОЛЖЕН
; БЫТЬ ЗАПИСАН КОД ПУСТОГО МЕСТА "VACUUM".
;
SETMEN: MVI M,MEN    ; ЗАПИСАТЬ НА НОВОЕ МЕСТО КОД ЧЕЛОВЕЧКА.
      LHLD INDMEN ; ВЗЯТЬ СТАРЫЕ ИНДЕКСЫ ЧЕЛОВЕЧКА И
      CALL INDEX ; ПОЛУЧИТЬ АДРЕС СТАРОГО МЕСТА В МАССИВЕ.
      MVI M,TRACE ; ЗАПИСАТЬ СЛЕД НА СТАРОЕ МЕСТО.
      XCHG       ; ПОМЕСТИТЬ НОВЫЕ ИНДЕКСЫ В "HL" И
      SHLD INDMEN ; ЗАПОМНИТЬ ИХ В ЯЧЕЙКЕ "INDMEN".
      RET
;
INDMEN: DW 0000H     ; ИНДЕКСЫ ЧЕЛОВЕЧКА В МАССИВЕ 40*22.
;
OPCRYS: DB 00H      ; СЛАГАЕМОЕ ДЛЯ СЧЕТЧИКА КРИСТАЛЛОВ.
STCRYS: DB 40H      ; СЧЕТЧИК КРИСТАЛЛОВ.
TIMEXP: EQU 5       ; ВРЕМЯ ВЗРЫВА.
STEXPL: DB 00H      ; СЧЕТЧИК ВРЕМЕНИ ВЗРЫВА.
TIMPUS: EQU 01H     ; ВРЕМЯ ТОЛКАНИЯ КАМНЯ.
STRUSH: DB 00H      ; СЧЕТЧИК ВРЕМЕНИ ТОЛКАНИЯ КАМНЯ.
BUFXPR: DB 00H      ; БУФЕР ДЛЯ КОДА, КОТОРЫЙ БУДЕТ ПОСЛЕ ВЗРЫВА.
PREXIT: DB 00H      ; ПРИЗНАК ВЫХОДА ЧЕЛОВЕЧКА ИЗ ЛАБИРИНТА.
PRGMEN: DB 00H      ; ПРИЗНАК ГИБЕЛИ ЧЕЛОВЕЧКА.
TON:     DB 00H      ; ВЫСОТА ТОНА ПРИ ГЕНЕРАЦИИ ЗВУКА.
;
; НОМЕРА ПАРАМЕТРОВ ДЛЯ ГЕНЕРАЦИИ ЗВУКА. ПАРАМЕТРЫ РАСПОЛОЖЕНЫ В МАССИВЕ
; С АДРЕСОМ НАЧАЛА "BAZTON". КАЖДЫЙ ПАРАМЕТР СОДЕРЖИТ ДВА СЛОВА - ПЕРВОЕ
; ЭТО ВЫСОТА ТОНА, ВТОРОЕ ДЛИТЕЛЬНОСТЬ ЗВУЧАНИЯ.
;
TONVAC: EQU 01      ; ДЛЯ ПЕРЕХОДА НА ПУСТОЕ МЕСТО ИСПОЛЬЗ. 1-Й ПАРАМ.
TONSOL: EQU 02      ; N ПАРАМЕТРА ДЛЯ ЗВУКА СЪЕДАНИЯ ГРУНТА.
TONCRS: EQU 03      ; N ПАРАМЕТРА ДЛЯ ЗВУКА СЪЕДАНИЯ РОМБА.
TONMOV: EQU 04      ; N ПАРАМЕТРА ДЛЯ ЗВУКА ТОЛКАНИЯ КАМНЯ.
TONEXP: EQU 05      ; N ПАРАМЕТРА ДЛЯ ЗВУКА ВЗРЫВА.
TONTIM: EQU 06      ; N ПАРАМЕТРА ДЛЯ ЗВУКА ТРЕВОГИ, КОГДА МАЛО ВРЕМЕНИ.
TONPAN: EQU 07      ; N ПАРАМЕТРА ЗВУКА ПРИ НАЧАЛЬНОМ ПАНОРАМИРОВАНИИ.
;

```

```

;
BAZTON: DW 0050      ; ЗВУК ПЕРЕХОДА НА ПУСТОЕ МЕСТО
        DW 0001      ; ВРЕМЯ ЗВУКА ПЕРЕХОДА НА ПУСТОЕ МЕСТО
;
        DW 0000      ; ЗВУК СЪЕДАНИЯ ГРУНТА.
        DW 0001      ; ВРЕМЯ ЗВУКА СЪЕДАНИЯ ГРУНТА.
;
        DW 0500      ; ЗВУК СЪЕДАНИЯ РОМБА.
        DW 0003      ; ВРЕМЯ ЗВУКА СЪЕДАНИЯ РОМБА.
;
        DW 1000      ; ЗВУК ТОЛКАНИЯ КАМНЯ.
        DW 0001      ; ВРЕМЯ ЗВУКА ТОЛКАНИЯ КАМНЯ.
;
        DW 1200      ; ЗВУК ВЗРЫВА.
        DW 0003      ; ВРЕМЯ ЗВУКА ВЗРЫВА.
;
        DW 3000      ; ЗВУК ТРЕВОГИ , КОГДА ОСТАЛОСЬ МАЛО ВРЕМЕНИ.
        DW 0004      ; ВРЕМЯ ЗВУКА ТРЕВОГИ.
;
VOSPAN: DW 0000      ; ЗВУК, УСТАНОВЛИВАЕМЫЙ ПРИ ПАНОРАМИРОВАНИИ.
        DW 0010      ; ДЛИТЕЛЬНОСТЬ.
;
;*****
; ПРОВЕРКИ И ПЕРЕМЕЩЕНИЯ ЭЛЕМЕНТОВ В МАССИВЕ 40*22.
;*****
;
; ДАННАЯ П/П ПРОВЕРЯЕТ ВЕСЬ МАССИВ ЭЛЕМЕНТОВ 40*22.
; ДЛЯ УВЕЛИЧЕНИЯ СКОРОСТИ РАБОТЫ ЗДЕСЬ НЕ ПРИМЕНЯЕТСЯ
; П/П ИНДЕКСИРОВАНИЯ МАССИВА, Т.К. ЕЕ ПРИШЛОСЬ БЫ
; ВЫЗЫВАТЬ ДЛЯ КАЖДОГО ЭЛЕМЕНТА Т.Е. 880 РАЗ!
; ВМЕСТО ЭТОГО, МАССИВ ПРОВЕРЯЕТСЯ КАК ОБЫЧНАЯ ОБЛАСТЬ
; ПАМЯТИ, А ЕСЛИ НЕОБХОДИМО ПЕРЕДВИГАТЬСЯ ПО СТРОКАМ,
; ТО К ТЕКУЩЕМУ АДРЕСУ ПРИБАВЛЯЕТСЯ (ИЛИ ОТНИМАЕТСЯ)
; КОЛИЧЕСТВО ЭЛЕМЕНТОВ В СТРОКЕ - Т.Е. 40.
; КРОМЕ ТОГО, ДЛЯ УВЕЛИЧЕНИЯ БЫСТРОДЕЙСТВИЯ НЕ ПРОВЕ-
; РЯЕТСЯ НИЖНЯЯ И ВЕРХНЯЯ СТРОКИ МАССИВА 40*22, Т.К. ЭТО
; ВСЕ РАВНО КОДЫ СТЕНОК, А ПРОВЕРЯЕМАЯ ОБЛАСТЬ СТАНОВИТСЯ
; КОРОЧЕ НА 80 ЭЛЕМЕНТОВ (ЭТО 10% ОТ ВСЕГО МАССИВА).
; ЕСЛИ В РЕЗУЛЬТАТЕ ПРОВЕРОК ЭЛЕМЕНТ МАССИВА ПЕРЕСТАВЛЯЕТСЯ,
; ТО НА ЕГО СТАРОЕ МЕСТО ЗАПИСЫВАЕТСЯ "СЛЕД", А САМ
; ЭЛЕМЕНТ ЗАПИСЫВАЕТСЯ НА НОВОМ МЕСТЕ С ПРИЗНАКОМ ТОГО,
; ЧТО ОН БЫЛ ПЕРЕСТАВЛЕН ВО ВРЕМЯ ЭТОЙ ПРОВЕРКИ. ( "1" В
; СТАРШЕМ РАЗРЯДЕ КОДА ЭТОГО ЭЛЕМЕНТА). ЭТО НЕОБХОДИМО
; ДЛЯ ТОГО, ЧТОБЫ ОДИН И ТОТ ЖЕ ЭЛЕМЕНТ НЕ БЫЛ ВЫ
; ПЕРЕСТАВЛЕН БОЛЕЕ ОДНОГО РАЗА ЗА ОДНУ ПРОВЕРКУ ВСЕГО
; МАССИВА.
;
MOVELM: LDA TURBO    ; ПЕРЕМЕЩАТЬ ЭЛЕМЕНТЫ В РЕЖИМЕ ТУРБО МОЖНО ВСЕГДА.
        ORA A
        JZ MOVL00
        LDA PANVER    ; В ДРУГИХ РЕЖИМАХ - ТОЛЬКО ПРИ ОТСУТСТВИИ ПАНОРА-
        MOV B,A        ; МИРОВАНИЯ ОКНА.
        LDA PANHOR
        ORA B
        RNZ
MOVL00: LXI H,BAZA+40 ; В "H1" АДРЕС НАЧАЛА ПЕРВОЙ СТРОКИ (0-Я ИСКЛ.)
        LXI B,880-80  ; В "BC" СЧЕТЧИК ЭЛЕМЕНТОВ МАССИВА (МИНУС ДВЕ СТРОКИ) .
MOVEL1: PUSH B        ; СОХРАНИТЬ ТЕКУЩЕЕ ЗНАЧЕНИЕ СЧЕТЧИКА.
        PUSH H        ; СОХРАНИТЬ АДРЕС ЭЛЕМЕНТА.
        MOV A,M        ; ВЗЯТЬ КОД ЭЛЕМЕНТА ИЗ МАССИВА.
        ANI 0CFH       ; ИСКЛЮЧИТЬ ИЗ КОДА БИТЫ ОПОРЫ БАБОЧКИ.
        CPI STONE      ; КОД ЭЛЕМЕНТА МЕНЬШЕ КОДА КАМНЯ?
        JC NEXELM      ; ЕСЛИ ДА - ПЕРЕЙТИ К СЛЕДУЮЩЕМУ ЭЛЕМЕНТУ МАССИВА.
        CPI EXIT       ; ЭТО ВЫХОД?
        JZ NEXELM      ; ЕСЛИ ДА - ТОЖЕ ОБОЙТИ.
        CPI TRACE      ; ЭТО КОД СЛЕДА ИЛИ ПЕРЕСТАВЛЕННОГО ЭЛЕМЕНТА?
        JNC NEXELM     ; ЕСЛИ ДА - СЛЕДУЮЩИЙ.
        CPI EXPLOS     ; ЭТО ВЗРЫВ?
        JZ OREXPRL     ; ЕСЛИ ДА - НА ПРОВЕРКУ ВРЕМЕНИ ВЗРЫВА.
        CPI BUTTER     ; ЭТО БАБОЧКА?
        JZ ORBUTT      ; ЕСЛИ ДА - НА ПРОВЕРКУ ПЕРЕМЕЩЕНИЯ БАБОЧКИ.

```

```

CPI MEN      ; ЭТО ЧЕЛОВЕЧЕК?
JZ NEXELM    ; ОБОЙТИ.
ORI 80H      ; КОДУ КАМНЯ(ИЛИ КРИСТАЛЛА)ПРИСВОИТЬ ПРИЗН.ПЕРЕСТАНОВА
STA BUFKOD   ; И СОХРАНИТЬ В БУФЕРЕ.
LXI B,0040   ; УСТАНОВИТЬ В "BC" ШАГ ДЛЯ ПЕРЕХОДА ЧЕРЕЗ СТРОКУ.
MOV A,L      ; ПЕРЕЙТИ НА СТРОКУ НИЖЕ,Т.Е. "ПОД КАМЕНЬ ИЛИ КРИСТАЛЛ"
SUB C
MOV L,A
MOV A,H
SBB B
MOV H,A
MOV A,M      ; ВЗЯТЬ КОД, НАХОДЯЩИЙСЯ ПОД КАМНЕМ (ИЛИ КРИСТАЛЛОМ).
ANI 0FH      ; ИСКЛЮЧИТЬ ИЗ КОДА БИТЫ ОПОРЫ БАБОЧКИ И ПР.ПЕРЕСТАНОВА
CPI BUTTER   ; НА ВРЕМЯ ПРОВЕРКИ БАБОЧКИ. ЭЛЕМ. СТОИТ НА БАБОЧКЕ?
JZ EXPCRY    ; ЕСЛИ ДА - ТО ВЗОРВАТЬ БАБОЧКУ!
MOV A,M      ; ЕСЛИ НЕТ - ТО СНОВА ВЗЯТЬ КОД, ДЛЯ ПРОВЕР.ДР.ЭЛЕМЕН.
ANI 0CFH     ; ИСКЛЮЧИТЬ БИТЫ ОПОРЫ БАБОЧКИ.
ORA A        ; ЭТО ПУСТОЕ МЕСТО?
JNZ TSTPOZ   ; ЕСЛИ НЕТ-ПРОВЕРИТЬ ПОЗИЦИЮ ЭЛЕМЕНТА НА СТАБИЛЬНОСТЬ.
LDA BUFKOD   ; ЕСЛИ ДА-ВЗЯТЬ КОД ИЗ БУФЕРА ПЕРЕСТАВЛЯЕМОГО ЭЛЕМЕНТА
MOV M,A      ; И ЗАПИСАТЬ ЕГО НА ПУСТОЕ МЕСТО.
MOV D,H      ; СОХРАНИТЬ НОВЫЙ АДРЕС ЭЛЕМЕНТА В "DE".
MOV E,L
DAD B        ; ПЕРЕЙТИ НА СТРОКУ ВЫШЕ, Т.Е. НА СТАРОЕ МЕСТО ЭЛЕМ.
TSBLOW: MVI M,TRACE ; И ЗАПИСАТЬ В НЕГО КОД СЛЕДА.
XCHG        ; ВЕРНУТЬ В "HL" НОВЫЙ АДРЕС И ПЕРЕЙТИ НА СТРОКУ НИЖЕ,
MOV A,L      ; ЧТОБЫ ПРОВЕРИТЬ: НА ЧТО "УПАЛ" ЭЛЕМЕНТ.
SUB C
MOV L,A
MOV A,H
SBB B
MOV H,A
MOV A,M      ; ВЗЯТЬ В "A" КОД, НА КОТОРЫЙ "УПАЛ" ЭЛЕМЕНТ.
ANI 0FH      ; ИСКЛЮЧИТЬ ИЗ КОДА БИТЫ ОПОРЫ БАБОЧКИ И ПР.ПЕРЕСТ.
CPI BUTTER   ; НА ВРЕМЯ ПРОВЕРКИ БАБОЧКИ. УПАЛ НА БАБОЧКУ?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗОРВАТЬ БАБОЧКУ!
MOV A,M      ; СНОВА ВЗЯТЬ КОД ДЛЯ ПРОВ. ДРУГИХ ЭЛЕМЕНТОВ.
ANI 0CFH     ; ИСКЛЮЧИТЬ БИТЫ ОПОРЫ БАБОЧКИ.
ORA A        ; ЭТО ПУСТОЕ МЕСТО?
JZ NEXELM    ; ЕСЛИ ДА - ПРОВЕРЯТЬ СЛЕДУЮЩИЙ ЭЛЕМЕНТ В МАССИВЕ.
CPI TRACE    ; ЭТО КОД СЛЕДА ИЛИ ПЕРЕСТАВЛЕННОГО ЭЛЕМЕНТА?
JNC NEXELM   ; ЕСЛИ ДА - СЛЕДУЮЩИЙ ЭЛ.МАССИВА.
CPI MEN      ; ЭТО ЧЕЛОВЕЧЕК?
JNZ TONBOM   ; ЕСЛИ НЕТ - УСТАНОВИТЬ ТОН УДАРА ОБ ОПОРУ.
MVI A,01     ; ЕСЛИ ДА - УСТАНОВИТЬ ПРИЗНАК ГИБЕЛИ ЧЕЛОВЕЧКА.
STA PRGMEN
MVI M,EXPLOS ; НА МЕСТО ЧЕЛОВЕЧКА - КОД ВЗРЫВА.
MVI A,TIMEXP+8 ; УСТАНОВИТЬ ВРЕМЯ
STA STEXPRL
MVI A,TONEXP ; И ЗВУК ВЗРЫВА.
STA TON
MVI A,TRACE  ; ПОСЛЕ ВЗРЫВА БУДЕТ КОД СЛЕДА.
STA BUFEHP
JMP NEXELM   ; НА ПРОВЕРКУ СЛЕД.ЭЛЕМЕНТА МАССИВА 40*22.
TONBOM: MVI A,TONMOV ; УСТАНОВИТЬ ЗВУК УДАРА ОБ ОПОРУ (ОН ЗДЕСЬ ТАКОЙ ЖЕ,
STA TON      ; КАК И ПРИ ТОЛКАНИИ КАМНЯ).
NEXELM: POP H ; ВЗЯТЬ ИЗ СТЕКА АДРЕС ЭЛЕМЕНТА И
POP B        ; ТЕКУЩЕЕ ЗНАЧЕНИЕ СЧЕТЧИКА.
INX H        ; ПЕРЕДВИНУТЬ АДРЕС НА СЛЕДУЮЩИЙ ЭЛЕМЕНТ МАССИВА 40*22
DCX B        ; УМЕНЬШИТЬ СЧЕТЧИК.
MOV A,B      ; ПРОВЕРИТЬ СЧЕТЧИК ЭЛЕМЕНТОВ.
ORA C
JNZ MOVE11   ; ЕСЛИ НЕ ВСЕ ЭЛ. ПРОВЕРЕНЫ - ПРОДОЛЖАТЬ.
JMP AVTRAC   ; УДАЛИТЬ ИЗ МАССИВА ВСЕ КОДЫ СЛЕДОВ И ПЕРЕСТАНОВОК.
;
OPREXPL: LDA STEXPRL ; ВЗЯТЬ СЧЕТЧИК ВРЕМЕНИ ВЗРЫВА,
ORA A        ; ПРОВЕРИТЬ ЕГО НА НОЛЬ
JNZ NOABEX   ; ЕСЛИ ВРЕМЯ ВЗРЫВА НЕ КОНЧИЛОСЬ - ОСТАВИТЬ КОД ВЗРЫВА.
LDA BUFEHP   ; ЕСЛИ КОНЧИЛОСЬ - ВЗЯТЬ КОД ИЗ БУФЕРА ВЗРЫВА И
MOV M,A      ; ЗАПИСАТЬ НА МЕСТО ВЗРЫВА.
JMP NEXELM   ; ПРОВЕРЯТЬ СЛЕДУЮЩИЙ ЭЛЕМЕНТ МАССИВА.

```

```

NOABEX: DCR A          ; УМЕНЬШИТЬ СЧЕТЧИК ВРЕМЕНИ ВЗРЫВА
        STA STEXPL     ;
        JMP NEXELM     ; ПРОВЕРЯТЬ СЛЕДУЮЩИЙ ЭЛЕМЕНТ.
;
; ПРОВЕРКА СТАБИЛЬНОСТИ ПОЛОЖЕНИЯ КАМНЯ ИЛИ КРИСТАЛЛА НА ОПОРЕ.
; ПРИ НЕСТАБИЛЬНОЙ ПОЗИЦИИ ЭЛЕМЕНТ ДОЛЖЕН "СВАЛИТЬСЯ" С ОПОРЫ ВЛЕВО ИЛИ
; ВПРАВО. ЧТОБЫ ЭЛЕМЕНТЫ НЕ ПАДАЛИ В ОДНУ СТОРОНУ В ПРОГРАММУ ВВЕДЕНО
; ИЗМЕНЕНИЕ НАПРАВЛЕНИЯ ПРОВЕРКИ - Т.Е. ЕСЛИ ПРЕДЫДУЩИЙ ЭЛЕМЕНТ УПАЛ ВЛЕВО,
; ТО ПРОВЕРКА ВОЗМОЖНОСТИ ПАДЕНИЯ СЛЕДУЮЩЕГО ЭЛЕМЕНТА БУДЕТ НАЧИНАТЬСЯ С
; ПРАВОЙ СТОРОНЫ И НАОБОРОТ.
;
TSTPROZ: CPI STONE     ; ЭЛЕМЕНТ (КАМЕНЬ ИЛИ КРИСТАЛЛ) СТОИТ НА КАМНЕ?
        JZ NOSTAB      ; ЕСЛИ ДА - ТО ЭТО НЕСТАБИЛЬНОЕ СОСТОЯНИЕ - ПРОВЕРИТЬ!
        CPI CRYSTL     ; СТОИТ НА КРИСТАЛЛЕ?
        JZ NOSTAB      ; ТОЖЕ НЕСТАБИЛЬНОЕ.
        CPI BUTTER     ; НА БАБОЧКЕ?
        JNZ NEXELM     ; ЕСЛИ НЕТ - ТО ПОЛОЖЕНИЕ СТАБИЛЬНОЕ (ПРОВЕРЯТЬ СЛ.ЭЛ.)
NOSTAB: POP H          ; ВЗЯТЬ АДРЕС ПРОВЕРЯЕМОГО ЭЛЕМЕНТА НЕ МЕНЯЯ "SP".
        PUSH H
        MVI E,2        ; СЧЕТЧИК ПРОВЕРОК СТАБИЛЬНОСТИ ЭЛЕМЕНТА.
        LDA STSIDE     ; ВЗЯТЬ ПЕРЕКЛЮЧАТЕЛЬ НАПРАВЛЕНИЙ ПРОВЕРКИ СТАБИЛЬНОСТИ
        MOV D,A         ; И НА ВРЕМЯ ПРОВЕРКИ ПОМЕСТИТЬ В РЕГ "D".
LSTAB: INR D            ; ИЗМЕНИТЬ НАПРАВЛЕНИЕ ПРОВЕРКИ(СЛЕВА ИЛИ СПРАВА ОТ ЭЛ)
        MOV A,D         ; ОПРЕДЕЛИТЬ
        RRC            ; НОВОЕ НАПРАВЛЕНИЕ ПРОВЕРКИ ПО МЛАДШЕМУ БИТУ:
        JNC TSTLEF     ; ЕСЛИ "0" - НАЧИНАТЬ ПРОВЕРКУ СЛЕВА ОТ ЭЛЕМЕНТА,
        INX H          ; ЕСЛИ "1" - СПРАВА.
        JMP TSTVC1     ; ПЕРЕЙТИ НА ПРОВЕРКУ ПУСТОГО МЕСТА СЛЕВА ИЛИ СПРАВА.
TSTLEF: DCX H          ; УКАЗАТЬ НА АДРЕС СЛЕВА ОТ ЭЛЕМЕНТА.
TSTVC1: MOV A,M        ; ВЗЯТЬ КОД СЛЕВА (ИЛИ СПРАВА) ОТ ЭЛЕМЕНТА И
        ORA A          ; ПРОВЕРИТЬ НА ПУСТОТУ.
        JNZ OTSIDE     ; ЕСЛИ МЕСТО ЗАНЯТО - ИЗМЕНИТЬ НАПРАВЛЕНИЕ ПРОВЕРКИ.
        MOV A,L        ; ПЕРЕЙТИ НА СТРОКУ НИЖЕ ПОД ПУСТЫМ МЕСТОМ.
        SUB C
        MOV L,A
        MOV A,H
        SBB B
        MOV H,A
        MOV A,M        ; ВЗЯТЬ КОД В ЭТОМ МЕСТЕ.
        ORA A          ; ЭТО ПУСТОЕ МЕСТО?
        JNZ OTSIDE     ; ЕСЛИ НЕТ - ИЗМЕНИТЬ НАПРАВЛЕНИЕ ПРОВЕРКИ.
        LDA BUFKOD     ; ЕСЛИ ДА - ЗАПИСАТЬ НА ЭТО МЕСТО КОД ЭЛЕМЕНТА С
        MOV M,A        ; ПРИЗНАКОМ ПЕРЕСТАНОВКИ.
        MOV A,D        ; ЗАПОМНИТЬ НАПРАВЛЕНИЕ ПРОВЕРКИ СТАБИЛЬНОСТИ, ЧТОБЫ
        STA STSIDE     ; В СЛЕДУЮЩЕЙ ПРОВЕРКЕ ПОМЕНИТЬ НА ПРОТИВОПОЛОЖНОЕ.
        XCHG           ; СОХРАНИТЬ АДРЕС НОВОГО МЕСТА В "DE".
        POP H          ; ВЗЯТЬ В "HL" АДРЕС СТАРОГО МЕСТА
        PUSH H         ; НЕ МЕНЯЯ ЗНАЧЕНИЯ УКАЗАТЕЛЯ СТЕКА.
        JMP TSBLOW     ; ПЕРЕЙТИ НА ЗАПИСЬ СЛЕДА И ПРОВЕРКУ УДАРА.
OTSIDE: DCR E          ; УМЕНЬШИТЬ СЧЕТЧИК ПРОВЕРОК.
        JZ NEXELM     ; ЕСЛИ БЫЛО ДВЕ ПРОВЕРКИ - ЭЛЕМЕНТ СТАБИЛЕН.
        POP H         ; ВЗЯТЬ АДРЕС ЭЛЕМЕНТА И
        PUSH H
        JMP LSTAB      ; СНОВА ПРОВЕРИТЬ ЕГО НА СТАБИЛЬНОСТЬ В ДР.НАПРАВЛЕНИИ.
;
; ПРОВЕРКА И ПЕРЕМЕЩЕНИЕ БАБОЧЕК. БАБОЧКА ДОЛЖНА ДВИГАТЬСЯ ПО ПУСТЫМ МЕСТАМ,
; ОГИБАЯ ПРЕПЯТСТВИЯ. АЛГОРИТМ ПРОВЕРОК И ДВИЖЕНИЯ СОСТОИТ В СЛЕДУЮЩЕМ:
; БАБОЧКА ДВИЖЕТСЯ, ПРИЖИМАЯСЬ К ПРЕПЯТСТВИЯМ, И ОБХОДИТ ДОСТУПНОЕ ЕЙ СВОБОДНОЕ
; ПРОСТРАНСТВО ПО ПЕРИМЕТРУ, ПО ЧАСОВОЙ СТРЕЛКЕ. ПРЕПЯТСТВИЕ, ВДОЛЬ КОТОРОГО
; ОНА ДВИЖЕТСЯ, НАЗЫВАЕТСЯ ЗДЕСЬ ОПОРОЙ. ЕСЛИ ДАЛЬНЕЙШЕЕ ПРОДВИЖЕНИЕ ВДОЛЬ
; ДАННОЙ ОПОРЫ НЕВОЗМОЖНО, ПРОВЕРЯЕТСЯ ВОЗМОЖНОСТЬ ДВИЖЕНИЯ ВДОЛЬ ДРУГОЙ ОПОРЫ.
; ДЛЯ СКОРОСТИ И ОДНОЗНАЧНОСТИ ВЫБОРА ОПОРЫ, В КОД БАБОЧКИ ВВЕДЕНО РАСШИРЕНИЕ:
; 4-Й И 5-Й БИТЫ ОПРЕДЕЛЯЮТ, С КАКОЙ СТОРОНЫ НАХОДИТСЯ ОПОРА. (ИМЕННО ПОЭТОМУ
; ПРИ ПРОВЕРКАХ КОДОВ В "MOVMEH" И "MOVELM", А ТАКЖЕ В "OUTSPR" ПРОИЗВОДИТСЯ
; ИСКЛЮЧЕНИЕ ЭТИХ БИТОВ КОМАНДОЙ ANI 0CFH ).
;
ORBUTT: LXI B,0040     ; УСТАНОВИТЬ В "BC" ШАГ ПЕРЕХОДА ЧЕРЕЗ СТРОКУ.
        MOV A,M        ; ВЗЯТЬ ПОЛНЫЙ БАЙТ БАБОЧКИ.
        ANI 30H        ; ВЫДЕЛИТЬ КОД ОПОРЫ, ВДОЛЬ КОТОРОЙ ЛЕТИТ БАБОЧКА.
        CPI 00H        ; ОПОРА СНИЗУ?
        JZ SUPDOW

```

```

CPI 10H      ; ОПОРА СЛЕВА?
JZ SUPLEF
CPI 20H      ; ОПОРА СВЕРХУ?
JZ SUPUP
CPI 30H      ; ОПОРА СПРАВА?
JZ SUPRIG
;
SUPDOW: MOV A,L      ; ПЕРЕЙТИ НА СТРОКУ НИЖЕ БАБОЧКИ.
SUB C
MOV L,A
MOV A,H
SBB B
MOV H,A
MOV A,M      ; ВЗЯТЬ КОД ЭЛЕМЕНТА "ПОД БАБОЧКОЙ"
CPI MEN      ; ЭТО ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗОРВАТЬ ЧЕЛОВЕЧКА И БАБОЧКУ.
POP H        ; ВЗЯТЬ АДРЕС БАБОЧКИ, НЕ МЕНЯЯ "SP".
PUSH H
DCX H        ; УКАЗАТЬ НА МЕСТО СЛЕВА ОТ БАБОЧКИ.
MOV A,M      ; ВЗЯТЬ КОД С ЭТОГО МЕСТА И ПРОВЕРИТЬ:
ORA A        ; ПУСТО?
JZ SUPDW1    ; ЕСЛИ ДА - ПРОВЕРИТЬ ОПОРУ НА НОВОМ МЕСТЕ.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
JMP TSTSPL   ; ВЛЕВО ДВИГАТЬСЯ НЕЛЬЗЯ - ПРОВЕРИТЬ ДВИЖЕНИЕ ВВЕРХ.
SUPDW1: MOV D,H      ; ЗАПОМНИТЬ НОВЫЙ АДРЕС БАБОЧКИ В "DE".
MOV E,L
MOV A,L      ; ПЕРЕЙТИ НА СТРОКУ НИЖЕ (ПРОВЕРИТЬ ОПОРУ).
SUB C
MOV L,A
MOV A,H
SBB B
MOV H,A
MOV A,M      ; ВЗЯТЬ КОД ОПОРЫ НА НОВОМ МЕСТЕ.
ORA A        ; ПУСТО?
JZ SETSPR    ; ЕСЛИ ДА - УСТАНОВИТЬ ОПОРУ СПРАВА.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
XRA A        ; ЕСЛИ НЕТ - УСТАНОВИТЬ ОПОРУ СНИЗУ (КАКАЯ И БЫЛА).
JMP MOV BUT  ; ПЕРЕЙТИ НА ПЕРЕМЕЩЕНИЕ БАБОЧКИ В МАССИВЕ.
SETSPR: MVI A,30H    ; УСТАНОВИТЬ ОПОРУ СПРАВА.
JMP MOV BUT  ; НА ПЕРЕМЕЩЕНИЕ.
;
SUPLEF: DCX H      ; УКАЗАТЬ НА МЕСТО СЛЕВА ОТ БАБОЧКИ.
MOV A,M      ; ВЗЯТЬ КОД.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
TSTSPL: POP H      ; ВЗЯТЬ АДРЕС БАБОЧКИ.
PUSH H
DAD B        ; УКАЗАТЬ МЕСТО НАД БАБОЧКОЙ.
MOV A,M      ; ВЗЯТЬ КОД С ЭТОГО МЕСТА.
ORA A        ; ПУСТО?
JZ SUPLF1    ; ЕСЛИ ДА - ПРОВЕРИТЬ ОПОРУ НА НОВОМ МЕСТЕ.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
JMP TSTSPU   ; ПРОВЕРИТЬ ВОЗМОЖНОСТЬ ДВИЖЕНИЯ ВПРАВО (ОПОРА СВЕРХУ).
SUPLF1: MOV D,H      ; ЗАПОМНИТЬ НОВЫЙ АДРЕС БАБОЧКИ В "DE".
MOV E,L
DCX H        ; АДРЕС ОПОРЫ СЛЕВА.
MOV A,M      ; ВЗЯТЬ КОД ОПОРЫ И ПРОВЕРИТЬ:
ORA A        ; ПУСТО?
JZ SETSPD    ; ЕСЛИ ДА - УСТАНОВИТЬ ОПОРУ СНИЗУ.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
MVI A,10H    ; ЕСЛИ НЕТ - УСТАНОВИТЬ ОПОРУ СЛЕВА.
JMP MOV BUT  ; НА ПЕРЕМЕЩЕНИЕ.
SETSPD: MVI A,00    ; УСТАНОВИТЬ ОПОРУ СНИЗУ.
JMP MOV BUT  ; НА ПЕРЕМЕЩЕНИЕ.
;
SUPUP: DAD B      ; УКАЗАТЬ МЕСТО НАД БАБОЧКОЙ.
MOV A,M      ; ВЗЯТЬ КОД.

```

```

CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
TSTSPU: POP H      ; ВЗЯТЬ АДРЕС БАБОЧКИ.
PUSH H
INX H        ; УКАЗАТЬ МЕСТО СПРАВА ОТ БАБОЧКИ.
MOV A,M      ; ВЗЯТЬ КОД.
ORA A        ; ПУСТО?
JZ SUPUP1    ; ЕСЛИ ДА - ПРОВЕРИТЬ ОПОРУ НА НОВОМ МЕСТЕ.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
JMP TSTRIG   ; ПРОВЕРИТЬ ВОЗМОЖНОСТЬ ДВИЖЕНИЯ ВНИЗ (ОПОРА СПРАВА).
SUPUP1: MOV D,H    ; СОХРАНИТЬ НОВЫЙ АДРЕС БАБОЧКИ В "DE".
MOV E,L
DAD B        ; АДРЕС ОПОРЫ СВЕРХУ.
MOV A,M      ; ВЗЯТЬ КОД ОПОРЫ.
ORA A        ; ПУСТО?
JZ SETSPL    ; ЕСЛИ ДА - УСТАНОВИТЬ ОПОРУ СЛЕВА
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
MVI A,20H    ; ЕСЛИ НЕТ - УСТАНОВИТЬ ОПОРУ СВЕРХУ.
JMP MOV BUT  ; НА ПЕРЕМЕЩЕНИЕ.
SETSPL: MVI A,10H  ; УСТАНОВИТЬ ОПОРУ СЛЕВА.
JMP MOV BUT  ; НА ПЕРЕМЕЩЕНИЕ.
;
SUPRIG: INX H      ; УКАЗАТЬ НА МЕСТО СПРАВА ОТ БАБОЧКИ.
MOV A,M      ; ВЗЯТЬ КОД.
CPI MEN      ; ЭТО ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
TSTRIG: POP H      ; ВЗЯТЬ АДРЕС БАБОЧКИ.
PUSH H
MOV A,L      ; ПЕРЕЙТИ НА СТРОКУ НИЖЕ ("ПОД БАБОЧКУ").
SUB C
MOV L,A
MOV A,H
SBB B
MOV H,A
MOV A,M      ; ВЗЯТЬ КОД.
ORA A        ; ПУСТО?
JZ SUPRG1    ; ЕСЛИ ДА - ПРОВЕРИТЬ ОПОРУ НА НОВОМ МЕСТЕ.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА - ВЗРЫВ.
POP H        ; ЕСЛИ НЕТ - НЕ ДЕЛАТЬ ПЕРЕМЕЩЕНИЯ БАБОЧКИ (НОВЫЙ
PUSH H      ; АДРЕС БУДЕТ
MOV D,H      ; РАВЕН СТАРОМУ.
MOV E,L
MVI A,00H    ; УСТАНОВИТЬ ОПОРУ СНИЗУ И
JMP MOV BUT  ; ПЕРЕЙТИ НА ПЕРЕМЕЩЕНИЕ, НО ЛИШЬ ДЛЯ ЗАПИСИ ОПОРЫ.
SUPRG1: MOV D,H    ; СОХРАНИТЬ НОВЫЙ АДРЕС БАБОЧКИ В "DE".
MOV E,L
INX H        ; АДРЕС ОПОРЫ СПРАВА НА НОВОМ МЕСТЕ.
MOV A,M      ; ВЗЯТЬ КОД ОПОРЫ.
ORA A        ; ПУСТО?
JZ SETSPU    ; ЕСЛИ ДА - УСТАНОВИТЬ ОПОРУ СВЕРХУ.
CPI MEN      ; ЧЕЛОВЕЧЕК?
JZ EXPCRY    ; ЕСЛИ ДА -ВЗРЫВ.
MVI A,30H    ; ЕСЛИ НЕТ - УСТАНОВИТЬ ОПОРУ СПРАВА.
JMP MOV BUT  ; НА ПЕРЕМЕЩЕНИЕ.
SETSPU: MVI A,20H  ; УСТАНОВИТЬ ОПОРУ СВЕРХУ.
JMP MOV BUT  ; НА ПЕРЕМЕЩЕНИЕ.
;
MOV BUT: POP H      ; ВЗЯТЬ АДРЕС СТАРОГО МЕСТА БАБОЧКИ И
PUSH H
MVI M,TRACE  ; ЗАПИСАТЬ НА ЕГО МЕСТО КОД СЛЕДА.
XCHG        ; ВЗЯТЬ В "HL" АДРЕС НОВОГО МЕСТА БАБОЧКИ ИЗ "DE".
ORI BUTTER   ; УСТАНОВИТЬ В "A" КОД БАБОЧКИ, НЕ МЕНЯЯ ОПОРЫ.
ORI 80H      ; УСТАНОВИТЬ ПРИЗНАК ПЕРЕСТАНОВКИ.
MOV M,A      ; ЗАПИСАТЬ НА НОВОЕ МЕСТО ДЛЯ БАБОЧКИ.
JMP NEXELM   ; НА ПРОВЕРКУ СЛЕДУЮЩЕГО ЭЛЕМЕНТА МАССИВА 40*22.
;
EXPCRY: MVI A,CRYSTL ; ПОСЛЕ ВЗРЫВА БУДУТ КРИСТАЛЛЫ
ORI 80H      ; С ПРИЗНАКОМ ПЕРЕСТАНОВКИ.

```


-17-

```

JNZ NEXLEV ; ЕСЛИ УСТАНОВЛЕН - ПЕРЕХОД НА СЛ. УРОВЕНЬ.
LDA OPCRY5 ; ВЗЯТЬ ОПЕРАНД ДЛЯ СЧЕТЧИКА КРИСТАЛЛОВ.
ORA A ; ОПЕРАНД РАВЕН НУЛЮ?
JZ CALC1 ; ЕСЛИ ДА - ПРОВЕРЯТЬ ДРУГИЕ ОПЕРАНДЫ.
XRA A ; СБРОСИТЬ ОПЕРАНД.
STA OPCRY5
LDA STCRY5 ; ВЗЯТЬ СЧЕТЧИК КРИСТАЛЛОВ И
DCR A ; УМЕНЬШИТЬ.
JM CALC1 ; ЕСЛИ МЕНЬШЕ НУЛЯ - НЕ ЗАНОСИТЬ В СЧЕТЧИК.
JZ OPENEX ; ЕСЛИ НОЛЬ (Т.Е. ВСЕ ЧТО НУЖНО СЪЕДЕНО) - ОТКР.ВЫХОД.
CALC7: STA STCRY5
LXI H,131CH ; УСТАНОВИТЬ МЕСТО ДЛЯ ВЫВОДА СЧЕТЧИКА КРИСТАЛЛОВ.
SHLD PLACE
LXI D, 0802H ; УСТАНОВИТЬ ЦВЕТ И РАЗРЯДНОСТЬ.
CALL OUTST ; ВЫВЕСТИ НА ЭКРАН.
CALC1: LDA OPLEVE ; ВЗЯТЬ ОПЕРАНД (ДЛЯ ЭТОГО СЛУЧАЯ ЛИШЬ ПРИЗНАК) УРОВНЯ
ORA A ; И ПРОВЕРИТЬ.
JZ CALC2 ; ЕСЛИ СБРОШЕН, ПРОВЕРЯТЬ ДРУГИЕ ОПЕРАНДЫ.
XRA A ; СБРОСИТЬ ПРИЗНАК ОТОБРАЖЕНИЯ УРОВНЯ.
STA OPLEVE
LDA NUMBL ; ВЗЯТЬ НОМЕР УРОВНЯ
LXI H,091CH ; УСТАНОВИТЬ МЕСТО ВЫВОДА.
SHLD PLACE
LXI D, 0802H ; УСТАНОВИТЬ ЦВЕТ И РАЗРЯДНОСТЬ.
CALL OUTST ; ВЫВЕСТИ НА ЭКРАН.
CALC2: LDA TIMEDL ; ВЗЯТЬ СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕКЛЮЧЕНИЯ ЧАСОВ
DCR A ; И УМЕНЬШИТЬ ЕГО.
STA TIMEDL
JNZ CALC3 ; ЕСЛИ НЕ НОЛЬ - ПРОВЕРЯТЬ СЛЕДУЮЩИЕ ОПЕРАНДЫ.
LDA CONTDL ; ВЗЯТЬ КОНСТАНТУ ВРЕМЕНИ ЗАДЕРЖКИ И
STA TIMEDL ; ЗАПИСАТЬ В СЧЕТЧИК ЗАДЕРЖКИ.
LDA TIME ; ВЗЯТЬ СЧЕТЧИК ВРЕМЕНИ ПРОХОДА УРОВНЯ (ЧАСЫ) И
ORA A ; ПРОВЕРИТЬ НА НОЛЬ.
JZ KILMEN ; ЕСЛИ ВРЕМЯ КОНЧИЛОСЬ - УСТАНОВИТЬ ГИБЕЛЬ ЧЕЛОВЕЧКА.
DCR A ; УМЕНЬШИТЬ СЧЕТЧИК ВРЕМЕНИ.
STA TIME
LXI H,1C1CH ; УСТАНОВИТЬ МЕСТО ВЫВОДА.
SHLD PLACE
LXI D,0803H ; УСТАНОВИТЬ ЦВЕТ И РАЗРЯДНОСТЬ.
CALL OUTST ; ВЫВЕСТИ НА ЭКРАН.
CALC3: LDA TIME
CPI 20 ; ПРОВЕРИТЬ СКОЛЬКО ВРЕМЕНИ ОСТАЛОСЬ И
JNC CALC9 ; ЕСЛИ БОЛЬШЕ ЭТОГО ЗНАЧЕНИЯ - НЕ ДАВАТЬ СИГНАЛ ТРЕВОГИ
LDA FAZTIM ; ВЗЯТЬ ШАБЛОН ЧАСТОТЫ МИГАНИЯ ТРЕВОГИ ПО ВРЕМЕНИ
RRC ; СДВИНУТЬ.
STA FAZTIM
CC FLASHT ; ЕСЛИ ПЕРЕНОС - МИГАТЬ НАДПИСЬЮ "ВРЕМЯ".
CALC9: LDA PRGMEN ; ВЗЯТЬ ПРИЗНАК ГИБЕЛИ ЧЕЛОВЕЧКА.
ORA A ; ПРИЗНАК УСТАНОВЛЕН?
JZ CALC4 ; ЕСЛИ НЕТ - ПРОВЕРЯТЬ СЛ. ОПЕРАНДЫ.
XRA A ; СБРОСИТЬ ПРИЗНАК ГИБЕЛИ.
STA PRGMEN
LDA LIVMEN ; ВЗЯТЬ СЧЕТЧИК ЖИЗНЕЙ ЧЕЛОВЕЧКА.
ORA A ; БОЛЬШЕ НЕТ ЖИЗНЕЙ?
JZ GMOVE1 ; ЕСЛИ ДА - КОНЕЦ ИГРЫ (НО СНАЧАЛА ЗАДЕРЖКА).
DCR A ; УМЕНЬШИТЬ СЧЕТЧИК ЖИЗНЕЙ.
STA LIVMEN
GMOVE1: MVI A,01 ; БЛОКИРОВАТЬ КЛАВИАТУРУ.
STA STRKEY
LDA ATRMEN ; УСТАНОВИТЬ ПРИЗНАК "STOP1".
ANI 8FH
ORI STOP1
STA ATRMEN
MVI A,28H ; УСТАНОВИТЬ ВРЕМЯ ЗАДЕРЖКИ ПЕРЕД ПЕРЕЗАГРУЗКОЙ
STA TIMLDR ; УРОВНЯ.
MVI A,TIMEXR+10 ; УСТАНОВИТЬ ВРЕМЯ ДЛЯ ОТОБРАЖЕНИЯ ГИБЕЛИ ПОПЫТКИ
STA TIMLIV ; (ВЗРЫВА ЧЕЛОВЕЧКА В ОКОШКЕ ОСТАВШИХСЯ ЖИЗНЕЙ).
LXI H,0C1CH ; УСТАНОВИТЬ АДРЕС МЕСТА ДЛЯ СТИРАНИЯ НАДПИСИ
SHLD PLACE ; "ИЩИ ВЫХОД".
LXI H,TEXT6
CALL TEXOUT ; СТЕРЕТЬ НАДПИСЬ.

```

```

    RET      ; ВЫЙТИ.
CALC6: LDA TIMLDR ; ВЗЯТЬ СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕЗАГРУЗКИ
    DCR A      ; УМЕНЬШИТЬ
    STA TIMLDR
    RNZ      ; ЕСЛИ НЕ РАВЕН НУЛЮ - ВЫЙТИ
    LDA LIVMEN ; ЖИЗНЕЙ БОЛЬШЕ НЕТ?
    ORA A
    JZ GMOVE2 ; ЕСЛИ ДА - КОНЕЦ ИГРЫ.
    MVI A,STONE ; КОД ЗАПОЛНЕНИЯ ЭКРАНА, ПЕРЕД ПЕРЕЗАГРУЗКОЙ УРОВНЯ
NEXLV3: STA FULKOD
    LDA ATRMEN ; УСТАНОВИТЬ В АТРИБУТ ЧЕЛОВЕЧКА ВЕКТОР "ВПРАВО".
    ANI 8FH
    ORI RIGHT
    STA ATRMEN
    CALL SCFULL ; ЗАПОЛНИТЬ ЭКРАН КОДОМ.
    POP PSW     ; ВЕРНУТЬ УКАЗАТЕЛЬ СТЕКА И
    LDA PREXIT  ; ПРОВЕРИТЬ: БЫЛ ПЕРЕХОД НА СЛ. УРОВЕНЬ, ИЛИ ЭТО
    CPI 02      ; ПЕРЕЗАГРУЗКА СТАРОГО?
    JNZ NSTART  ; ЕСЛИ ПЕРЕЗАГРУЗКА - ТО ВЫПОЛНЯТЬ!
    LDA NUMBL   ; ЕСЛИ ПЕРЕХОД - ПРОВЕРИТЬ НОМЕР СЛЕДУЮЩ.УРОВНЯ.
    CPI 03      ; ТРЕТИЙ УРОВЕНЬ?
    JZ GIRL     ; ЕСЛИ ДА - ВЫВЕСТИ РЕКЛАМУ С ДЕВУШКОЙ.
    JMP NSTART  ; ПЕРЕЗАГРУЗИТЬ УРОВЕНЬ.
CALC4: RET
OPENEX: LHLD INDEXT ; ВЗЯТЬ ИНДЕКСЫ ВЫХОДА,
    CALL INDEX ; ОПРЕДЕЛИТЬ ЕГО АДРЕС В МАССИВЕ.
    MVI M,EXIT ; УСТАНОВИТЬ ПО ЭТОМУ АДРЕСУ КОД ВЫХОДА.
    LXI H,0C1CH ; УСТАНОВИТЬ МЕСТО ВЫВОДА НА НАДПИСЬ "АЛМАЗЫ:".
    SHLD PLACE
    LXI H,TEXT4 ; СТЕРЕТЬ НАДПИСЬ "АЛМАЗЫ:00" ПРОБЕЛАМИ
    CALL TEXOUT
    LXI H,0C1CH ; ВМЕСТО НАДПИСИ "АЛМАЗЫ: 00" НАПИСАТЬ "ИЩИ ВЫХОД"
    SHLD PLACE ; ЦВЕТОМ, КОТОРЫМ ВЫВОДЯТСЯ РАМКИ.
    LXI H,TEXT5
    CALL TEXOUT
    JMP CALC1 ; ОТОБРАЗИТЬ НУЛЕВОЙ РЕЗУЛЬТАТ.
KILMEN: MVI A,01H ; УСТАНОВИТЬ ПРИЗНАК ГИБЕЛИ ЧЕЛОВЕЧКА.
    STA PRGMEN
    LHLD INDMEN ; ВЗЯТЬ ИНДЕКСЫ ЧЕЛОВЕЧКА И
    CALL INDEX
    MVI M,EXPLOS ; ЗАПИСАТЬ ПО НИМ КОД ВЗРЫВА.
    MVI A,TRACE ; ПОСЛЕ ВЗРЫВА БУДЕТ КОД СЛЕДА.
    STA BUFEPR
    MVI A,TONEXP ; ЗВУК ВЗРЫВА.
    STA TON
    MVI A,TIMEXP+8 ; ВРЕМЯ ВЗРЫВА.
    STA STEXP
    JMP CALC3 ; ПРОВЕРИТЬ КОЛИЧЕСТВО ЖИЗНЕЙ.
GMOVE2: POP PSW ; ВЕРНУТЬ УКАЗАТЕЛЬ СТЕКА.
    JMP GIRL ; ВЫВЕСТИ РЕКЛАМУ С ДЕВУШКОЙ.
;
NEXLEV: LDA PREXIT ; ПРОВЕРИТЬ: ПРИЗНАК ПЕРЕХОДА ПРОВЕРЯЕТСЯ 2-Й РАЗ?
    CPI 02
    JZ NEXLV1 ; ЕСЛИ ДА - ОТРАБОТКА ВРЕМЕНИ ПЕРЕД ПЕРЕХОДОМ
    MVI A,02
    STA PREXIT
    MVI A,10 ; УСТАНОВИТЬ ВРЕМЯ ПЕРЕД ПЕРЕХОДОМ НА СЛ. УРОВЕНЬ.
    STA TIMLDN
    MVI A,1 ; БЛОКИРОВАТЬ КЛАВИАТУРУ.
    STA STPKEY
    LHLD INDMEN ; НА МЕСТЕ ЧЕЛОВЕЧКА ОСТАВИТЬ КОД ВЫХОДА.
    CALL INDEX
    MVI M,EXIT
    LDA NUMBL ; УСТАНОВИТЬ НОМЕР СЛ.УРОВНЯ
    INR A
    STA NUMBL
    LDA LIVMEN ; УВЕЛИЧИТЬ НА 1 ЧИСЛО ПОПЫТОК ЗА ПРОХОД УРОВНЯ.
    INR A
    STA LIVMEN
    LXI H,0C1CH ; ВЫВЕСТИ ТЕКСТ " НАШЕЛ "
    SHLD PLACE

```

```

LXI H,TEXT7
CALL TEXOUT
LXI H,4000H ; ВЫСОТА ТОНА ПРИ ВЫХОДЕ.
SHLD VOCPAN
RET
NEXLV1: LDA TIMLDN ; ВРЕМЯ ОТРАБОТАНО?
DCR A
STA TIMLDN
JNZ NEXLV2 ; ЕСЛИ НЕТ - ЗВУК.
LDA DEMO ; ЭТО БЫЛА ДЕМОНСТРАЦИЯ ?
ORA A
JNZ START ; ЕСЛИ ДА - НА ВЫВОД ЗАСТАВКИ.
LXI H,0C1CH ; ЕСЛИ ДА - СТЕРЕТЬ НАДПИСЬ " НАШЕЛ "
SHLD PLACE
LXI H,TEXT6
CALL TEXOUT
MVI A,MEN ; УСТАНОВИТЬ КОД ЧЕЛОВЕЧКА ДЛЯ ЗАПОЛНЕНИЯ ЭКРАНА.
JMP NEXLV3
NEXLV2: LHLD VOCPAN ; ЗВУК ПРИ ВЫХОДЕ.
MOV A,H
SUI 4
MOV H,A
SHLD VOCPAN
MVI A,TONPAN
STA TON
RET
;
SCFULL: LXI H,BAZA ; НАЧАЛО РАБОЧЕГО МАССИВА 40*22.
LXI B,880 ; СЧЕТЧИК ЕГО ЭЛЕМЕНТОВ.
SCFUL1: LDA FULKOD ; ЗАПОЛНИТЬ ЭКРАН КОДОМ ИЗ "FULKOD"
MOV M,A
INX H
DCX B
MOV A,B
ORA C
JNZ SCFUL1
TIMSHO: MVI B,20H ; ВРЕМЯ ПОКАЗА ЗАПОЛНЕННОГО ЭКРАНА ПЕРЕД ПЕРЕЗАГР.
SCFUL2: PUSH B
CALL WINDOW ; ОТОБРАЗИТЬ ОКНО
POP B
DCR B ; ВРЕМЯ КОНЧИЛОСЬ?
JNZ SCFUL2
RET
;
FLASHT: LDA FZFLAS ; ВЗЯТЬ ФАЗУ МИГАНИЯ (ИНВЕРСНЫЙ/НОРМАЛЬНЫЙ) .
INR A
STA FZFLAS
ANI 01H ; ЕСЛИ ЧЕТНЫЙ - ИНВЕРСИЯ, ЕСЛИ НЕТ - НОРМАЛЬНЫЙ.
JNZ INFLAS
MVI A,11H ; УСТАНОВИТЬ НОРМАЛЬНЫЙ РЕЖИМ ВЫВОДА СИМВОЛА.
JMP FLUSH1 ; НА ВЫВОД.
INFLAS: MVI A,TONTIM ; УСТАНОВИТЬ ЗВУК ТРЕВОГИ ПРИ ИНВЕРСНОМ ВЫВОДЕ.
STA TON
MVI A,10H ; УСТАНОВИТЬ ИНВЕРСНЫЙ РЕЖИМ.
FLUSH1: CALL SYMOUT ; ВКЛЮЧИТЬ УСТАНОВЛЕННЫЙ РЕЖИМ.
LXI H,151CH ; УСТАНОВИТЬ МЕСТО ВЫВОДА СООБЩЕНИЯ "ВРЕМЯ:"
SHLD PLACE
LXI H,TEXT3 ; АДРЕС СООБЩЕНИЯ.
CALL TEXOUT
RET
;
OPLEVE: DB 00H ; ПРИЗНАК ОТОБРАЖЕНИЯ НОМЕРА УРОВНЯ.
TIME: DB 140 ; ВРЕМЯ ДЛЯ ПРОХОДА УРОВНЯ.
TIMEDL: DB 00H ; СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕКЛЮЧЕНИЯ ВРЕМЕНИ.
CONTDL: DB 08H ; КОНСТАНТА ЗАДЕРЖКИ ПЕРЕКЛЮЧЕНИЯ ВРЕМЕНИ.
TIMLDR: DB 00H ; СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕЗАГРУЗКИ УРОВНЯ.
FULKOD: DB 00H ; БУФЕР КОДА ЗАПОЛНЕНИЯ ЭКРАНА ПЕРЕД ПЕРЕЗАГРУЗКОЙ.
FAZTIM: DB 11H ; ШАБЛОН ЧАСТОТЫ МИГАНИЯ СООБЩЕНИЯ "ВРЕМЯ:"
FZFLAS: DB 00H ; ПЕРЕКЛЮЧАТЕЛЬ МИГАНИЯ.
TIMLDN: DB 00H ; СЧЕТЧИК ЗАДЕРЖКИ ЗАГРУЗКИ НОВОГО УРОВНЯ.
;

```

```

;*****
; ПРЕОБРАЗОВАНИЕ ДВОИЧНОГО ЧИСЛА (БАЙТА) В
; ДВОИЧНО-ДЕСЯТИЧНОЕ И ВЫВОД ЕГО НА ЭКРАН.
;*****
;
; ДЛЯ БОЛЬШЕЙ УНИВЕРСАЛЬНОСТИ ВОЗМОЖЕН ВЫВОД
; РАЗНОГО КОЛИЧЕСТВА РАЗРЯДОВ - ОТ 1 ДО 3.
;
; ВХОДНЫЕ ПАРАМЕТРЫ: "А" - ЧИСЛО В ДВОИЧНОМ КОДЕ.
;                   "Е" - ЧИСЛО ВЫВОДИМЫХ РАЗРЯДОВ.
;                   "D" - ЦВЕТ ВЫВОДА (1,2,4,8 - СМОТРИ "SYMOUT")
;
OUTST:  LXI H,0      ; ОБНУЛЕНИЕ ТЕКУЩЕЙ СУММЫ.
        MVI B,8      ; СЧЕТЧИК ЦИКЛОВ.
        MOV C,A      ; ПОМЕСТИТЬ ЧИСЛО В "С".
OUTST1: MOV A,C
        RLC          ; СДВИГ ДВОИЧНОГО ЧИСЛА ВЛЕВО.
        MOV C,A
        MOV A,L      ; ДВОИЧНО-ДЕСЯТИЧНОЕ УДВОЕНИЕ СУММЫ С УЧЕТОМ ПЕРЕНОСА.
        ADC L
        DAA
        MOV L,A
        MOV A,H
        ADC H
        DAA
        MOV H,A
        DCR B        ; ПРОВЕРИТЬ КОНЕЦ ЦИКЛОВ.
        JNZ OUTST1   ; ЕСЛИ КОНЕЦ - В "HL" 3-Х РАЗРЯДНОЕ ДВОИЧ.-ДЕС. ЧИСЛО.
        MOV A,D      ; ВЗЯТЬ ЦВЕТ ВЫВОДА И
        CALL SYMOUT  ; УСТАНОВИТЬ ЕГО.
        MOV A,E      ; ПРОВЕРИТЬ ЧИСЛО ВЫВОДИМЫХ РАЗРЯДОВ.
        CPI 1
        JZ ONEST
        CPI 2
        JZ TWOST
        MOV A,H      ; ВО ВСЕХ ОСТАЛЬНЫХ СЛУЧАЯХ ВЫВОДИТЬ 3 РАЗРЯДА.
        ANI 0FH      ; ВЫДЕЛИТЬ СТАРШИЙ РАЗРЯД ЧИСЛА.
        ADI 30H      ; ПРИВЕСТИ К СТАНДАРТНОМУ КОДУ ASCII.
        CALL SYMOUT  ; ВЫВЕСТИ ЧЕРЕЗ П/П ВЫВОДА СИМВОЛА.
TWOST:  MOV A,L
        RRC          ; ВЫДЕЛИТЬ ВТОРОЙ РАЗРЯД.
        RRC
        RRC
        RRC
        ANI 0FH
        ADI 30H
        CALL SYMOUT
ONEST:  MOV A,L
        ANI 0FH      ; ВЫДЕЛИТЬ ТРЕТИЙ (САМЫЙ МЛАДШИЙ) РАЗРЯД ЧИСЛА.
        ADI 30H
        CALL SYMOUT
        RET
;
;
;*****
; ОТОБРАЖЕНИЕ КОЛИЧЕСТВА ЖИЗНЕЙ ЧЕЛОВЕЧКА
;*****
;
; ПРИМЕНЕНО ОРИГИНАЛЬНОЕ ОТОБРАЖЕНИЕ КОЛ-ВА
; ПОПЫТОК ВВИДЕ ПОДВИЖНЫХ ЧЕЛОВЕЧКОВ.
; ОНИ ПОЛНОСТЬЮ ПОВТОРЯЮТ МУЛЬТИПЛИКАЦИЮ
; ОСНОВНОГО ЧЕЛОВЕЧКА, Т.К. ВЕКТОР МУЛЬТИПЛИ-
; КАЦИИ У НИХ ОБЩИЙ.
;
OUTLIV: LDA LIVMEN   ; ВЗЯТЬ КОЛ-ВО ЖИЗНЕЙ ЧЕЛОВЕЧКА
        MOV C,A      ; СЧЕТЧИК ОТОБРАЖАЕМЫХ ЖИЗНЕЙ.
        MVI B,5      ; СЧЕТЧИК ЧИСЛА ВЫВОДИМЫХ СПРАЙТОВ (ДЛИНА ОКНА ЖИЗНЕЙ).
        LXI H,0FD10H ; АДРЕС ВЫВОДА САМОГО ЛЕВОГО СПРАЙТА.
OTLIV1: SHLD ADSPR
        DCR C        ; ОТОБРАЖАТЬ ЖИЗНЬ?
        JM OTLIV2    ; ЕСЛИ ЖИЗНИ КОНЧИЛИСЬ - ОТОБРАЖАТЬ ПУСТОЕ МЕСТО.

```

```

JNZ OTLIV4 ; ЕСЛИ НЕ КОНЧИЛИСЬ И ЭТО НЕ КРАЙНЯЯ СЛЕВА - ЧЕЛОВЕЧКА.
LDA TIMLIV ; ЕСЛИ ЭТО КРАЙНЯЯ ЛЕВАЯ (БЫВАША ПОПЫТКА) - ПРОВЕРИТЬ:
ORA A ; МОЖЕТ ОНА ЕЩЕ ПОГИБАЕТ ?
JZ OTLIV2 ; ЕСЛИ НЕТ - ВЫВОДИТЬ ПУСТОЕ МЕСТО.
DCR A ; ЕСЛИ ДА - УМЕНЬШИТЬ ВРЕМЯ ВЫВОДА ПОПЫТКИ
STA TIMLIV
MVI A,EXPLOS ; И ВЫВОДИТЬ НА ЕЕ МЕСТЕ СПРАЙТ ВЗРЫВА.
JMP OTLIV3
OTLIV4: MVI A,MEN
JMP OTLIV3
OTLIV2: XRA A
OTLIV3: PUSH H
PUSH B
CALL OUTSPR ; ВЫВЕСТИ ЧЕРЕЗ ДРАЙВЕР ВЫВОДА СПРАЙТА.
POP B
POP H
DCR H ; ПЕРЕЙТИ НА СПРАЙТ ЛЕВЕЕ ПРЕДЫДУЩЕГО.
DCR H
DCR B ; ВСЕ СПРАЙТЫ ОКНА ЖИЗНЕЙ ВЫВЕДЕНА?
JNZ OTLIV1 ; ЕСЛИ НЕТ - ПРОДОЛЖАТЬ.
RET
;
LIVMEN: DB 06H ; СЧЕТЧИК ЖИЗНЕЙ ЧЕЛОВЕЧКА.
TIMLIV: DB 00H ; ВРЕМЯ ОТОБРАЖЕНИЯ ГИБЕЛИ ПОПЫТКИ.
;
;
; *****
; ПОДПРОГРАММА ПРОВЕРКИ ВКЛЮЧЕНИЯ ПАНОРАМИРОВАНИЯ
; ПРИ ДОСТИЖЕНИИ ЧЕЛОВЕЧКОМ ГРАНИЦ ОКНА (УЧАСТКА
; МАССИВА), ОТОБРАЖАЕМОГО НА ЭКРАНЕ.
; *****
;
; П/П ПО ИНДЕКСАМ ОКНА И ИНДЕКСАМ ЧЕЛОВЕЧКА
; ОПРЕДЕЛЯЕТ НЕОБХОДИМОСТЬ ВКЛЮЧЕНИЯ ПАНОРАМИРОВАНИЯ
; А ТАКЖЕ ЗАДАЕТ НАПРАВЛЕНИЕ ПАНОРАМИРОВАНИЯ.
;
TSTRAN: PUSH H ; СОХРАНИТЬ ИСПОЛЬЗУЕМЫЕ РЕГИСТРЫ.
PUSH D
PUSH PSW
LHLD INDVMS ; ВЗЯТЬ ИНДЕКСЫ ОКНА И
XCHG ; ПОМЕСТИТЬ В "DE".
LHLD INDMEN ; ВЗЯТЬ ИНДЕКСЫ ЧЕЛОВЕЧКА.
MOV A,E ; ВЗЯТЬ В "A" ИНД.СТРОКИ ОКНА (ЕГО НИЖНЮЮ ГРАНИЦУ) И
CMP L ; СРАВНИТЬ С ИНДЕКСОМ СТРОКИ ЧЕЛОВЕЧКА.
JP RANDOW ; ЕСЛИ РАВНЫ - ВКЛЮЧИТЬ ПАНОРАМИРОВАНИЕ ВНИЗ.
ADI HEIGHT-2 ; УСТАНОВИТЬ В "A" ВЕРХНЮЮ ГРАНИЦУ ОКНА И
CMP L ; СРАВНИТЬ С ИНДЕКСОМ СТРОКИ ЧЕЛОВЕЧКА.
JM PANUP ; ЕСЛИ РАВНЫ - ВКЛЮЧИТЬ ПАНОРАМИРОВАНИЕ ВВЕРХ.
TSTRHR: MOV A,D ; ВЗЯТЬ В "A" ИНД.СТОЛБЦА ОКНА (ЕГО ЛЕВУЮ ГРАНИЦУ) И
CMP H ; СРАВНИТЬ С ИНД.СТОЛБЦА ЧЕЛОВЕЧКА.
JP PANLEF ; ЕСЛИ РАВНЫ - ВКЛЮЧИТЬ ПАНОРАМИРОВАНИЕ ВЛЕВО.
ADI LENGTH-2 ; УСТАНОВИТЬ В "A" ПРАВУЮ ГРАНИЦУ ОКНА И
CMP H ; СРАВНИТЬ С ИНД.СТОЛБЦА ЧЕЛОВЕЧКА.
JP EXPANT
MVI A,01 ; ВКЛЮЧИТЬ ПАНОРАМИРОВАНИЕ ВПРАВО.
STA PANHOR
EXPANT: POP PSW ; ВЕРНУТЬ ИСПОЛЬЗУЕМЫЕ РЕГИСТРЫ И ВЫЙТИ ИЗ П/П.
POP D
POP H
RET
;
RANDOW: MVI A,02 ; УСТАНОВИТЬ ВЕКТОР ВЕРТИКАЛЬНОГО ПАНОРАМИРОВАНИЯ ВНИЗ.
STA PANVER
JMP TSTRHR ; ПЕРЕЙТИ НА ПРОВЕРКУ ЛЕВОЙ И ПРАВОЙ ГРАНИЦЫ ОКНА.
PANUP: MVI A,01 ; УСТАНОВИТЬ ВЕКТОР ВЕРТИКАЛЬНОГО ПАНОРАМИР. ВВЕРХ.
STA PANVER
JMP TSTRHR
PANLEF: MVI A,02 ; УСТАНОВИТЬ ВЕКТОР ГОРИЗОНТАЛЬНОГО ПАНОРАМИР. ВЛЕВО.
STA PANHOR
JMP EXPANT ; НЕ ПРОВЕРЯТЬ ПРАВУЮ ГРАНИЦУ И ВЫЙТИ ИЗ П/П.
;

```

```

INDVMS: DW 0000H      ; ИНДЕКСЫ ОТОБРАЖАЕМОГО ОКНА (ЕГО ЛЕВЫЙ НИЖНИЙ УГОЛ) .
;
; *****
; ПОДПРОГРАММА ПАНОРАМИРОВАНИЯ (ПЕРЕМЕЩЕНИЯ)
; ОТОБРАЖАЕМОГО УЧАСТКА МАССИВА В ОСНОВНОМ
; МАССИВЕ ЭЛЕМЕНТОВ.
; *****
;
; ПРИ ПАНОРАМИРОВАНИИ В ДАННОЙ ПРОГРАММЕ "BOULDER",
; ОКНО 16*10 ПЕРЕМЕЩАЕТСЯ ПО МАССИВУ 40*22 ПО ЗОНАМ.
; Т.Е. ЕСЛИ ОКНО "ТОЛКНУТЬ" В ТУ ИЛИ ДРУГУЮ
; СТОРОНУ, ТО ОНО БУДЕТ ПЕРЕМЕЩАТЬСЯ ДО СЛЕДУЮЩЕЙ ЗОНЫ И
; ЗАФИКСИРУЕТСЯ ТАМ. ОКНО МОЖНО "ТОЛКАТЬ" И В ОБРАТНУЮ
; СТОРОНУ ПРИ ЕГО ПЕРЕМЕЩЕНИИ (Т.Е КОГДА ОНО ЕЩЕ НЕ ДОШЛО
; ДО ФИКСАЦИИ В СЛЕДУЮЩЕЙ ЗОНЕ), ТОГДА ОНО БУДЕТ ДВИГАТЬСЯ
; В ОБРАТНОМ НАПРАВЛЕНИИ ДО ФИКСАЦИИ В СТАРОЙ ЗОНЕ.
; ТАКЖЕ, ЕСЛИ ОКНО ДВИЖУЩЕЕСЯ ПО ГОРИЗОНТАЛИ
; "ТОЛКНУТЬ" В ВЕРТИКАЛЬНОМ НАПРАВЛЕНИИ (И НАОБОРОТ), ОКНО
; НАЧНЕТ ДВИГАТЬСЯ ПО ДИАГОНАЛИ ДО ФИКСАЦИИ ПО ВЕРТИКАЛИ ИЛИ
; ПО ГОРИЗОНТАЛИ (КАКОЙ ГРАНИЦЫ ЗОНЫ БЫСТРЕЕ ДОСТИГНЕТ),
; ПОСЛЕ ЧЕГО БУДЕТ ДВИГАТЬСЯ В ОДНОМ НАПРАВЛЕНИИ ДО ОКОНЧА-
; ТЕЛЬНОЙ ФИКСАЦИИ В ОСТАВШЕЙСЯ ЗОНЕ.
; ДАННАЯ П/П ТАКЖЕ УСТАНАВЛИВАЕТ ТИП СПРАЙТОВОЙ
; РАЗВЕРТКИ ДЛЯ "VMSOUT" В ЗАВИСИМОСТИ ОТ НАПРАВЛЕНИЯ ПАНОРАМИРОВАНИЯ.
; НАПРАВЛЕНИЕ ПАНОРАМИРОВАНИЯ ОПРЕДЕЛЯЕТСЯ ДВУМЯ
; ВЕКТОРАМИ: "PANVER" - ВЕРТИКАЛЬНОЕ, "PANHOR" - ГОРИЗОНТАЛЬНОЕ.
; КАЖДЫЙ ИЗ ЭТИХ ВЕКТОРОВ МОЖЕТ ПРИНИМАТЬ ПО ТРИ ЗНАЧЕНИЯ:
;
; "00" - ПАНОРАМИРОВАНИЕ В ДАННОМ НАПРАВЛЕНИИ ОТСУТСТВУЕТ.
; "01" - ПАНОРАМИРОВАНИЕ В СТОРОНУ УВЕЛИЧЕНИЯ ИНДЕКСОВ (ОТ НАЧАЛА МАССИВА) .
; "02" - ПАНОРАМИРОВАНИЕ В СТОРОНУ УМЕНЬШЕНИЯ ИНДЕКСОВ.
;
; ЗОНЫ ФИКСАЦИИ ОКНА: ПО ГОРИЗОНТАЛИ ЧЕРЕЗ 8 ЭЛЕМЕНТОВ (СТОЛБЦОВ) МАССИВА.
; ПО ВЕРТИКАЛИ ЧЕРЕЗ 6 ЭЛЕМЕНТОВ (СТРОК) МАССИВА.
;
PANORM: PUSH H      ; СОХРАНИТЬ ИСПОЛЬЗУЕМЫЕ РЕГИСТРЫ.
        PUSH PSW
        LHL D INDVMS ; ВЗЯТЬ БАЗОВЫЕ ИНДЕКСЫ ОТОБРАЖАЕМОГО ОКНА.
        LDA PANVER   ; ВЗЯТЬ ВЕКТОР ВЕРТИКАЛЬНОГО ПАНОРАМИРОВАНИЯ.
        ORA A        ; ВЕРТИКАЛЬНОЕ ПАНОРАМИРОВАНИЕ ОТСУТСТВУЕТ?
        JZ TSTHOR    ; ЕСЛИ ДА - ПРОВЕРЯТЬ ГОРИЗОНТАЛЬНОЕ.
        CPI 01       ; УВЕЛИЧИТЬ ИНДЕКС СТРОКИ ? (ПАНОРАМИРОВАНИЕ ВВЕРХ?) .
        JNZ VERDCR   ; ЕСЛИ НЕТ - УМЕНЬШИТЬ (ПАНОРАМИРОВАТЬ ВНИЗ) .
        INR L        ; УВЕЛИЧИТЬ ИНДЕКС СТРОКИ.
        JMP TSTHOR    ; ПЕРЕЙТИ НА ПРОВЕРКУ ГОРИЗОНТАЛЬНОГО ПАНОРАМИРОВАНИЯ.
VERDCR: DCR L        ; УМЕНЬШИТЬ ИНДЕКС СТРОКИ.
TSTHOR: MVI A,01     ; УСТАНОВИТЬ СПРАЙТОВУЮ РАЗВЕРТКУ ПО СТОБЦАМ ДЛЯ
        STA TIRVMS   ; П/П "VMSOUT".
        LDA PANHOR   ; ВЗЯТЬ ВЕКТОР ГОРИЗОНТАЛЬНОГО ПАНОРАМИРОВАНИЯ.
        ORA A        ; ГОРИЗОНТАЛЬНОЕ ПАНОРАМИРОВАНИЕ ОТСУТСТВУЕТ?
        JZ OFTIR     ; ЕСЛИ ДА - ПЕРЕЙТИ НА ПРОВЕРКУ ЗОН И ПЕРЕКЛ.ТИП РАЗВ.
        CPI 01       ; ПАНОРАМИРОВАНИЕ ВПРАВО?
        JNZ HORDCR   ; ЕСЛИ НЕТ - ТО ВЛЕВО.
        INR H        ; УВЕЛИЧИТЬ ИНДЕКС СТОЛБЦА.
        JMP TSTZON   ; ПЕРЕЙТИ НА ПРОВЕРКУ ЗОН ФИКСАЦИИ ОКНА.
HORDCR: DCR H        ; УМЕНЬШИТЬ ИНДЕКС СТОЛБЦА.
        JMP TSTZON   ; ОБОЙТИ ПЕРЕКЛЮЧЕНИЕ ТИПА РАЗВЕРТКИ.
OFTIR: STA TIRVMS    ; УСТАНОВИТЬ РАЗВЕРТКУ ПО СТРОКАМ ДЛЯ П/П "VMSOUT".
TSTZON: MOV A,L      ; ВЗЯТЬ ИНДЕКС СТРОКИ И ПРОВЕРИТЬ НАЛИЧИЕ ФИКСАЦИИ ПО
        CPI 0        ; ВЕРТИКАЛЬНЫМ ЗОНАМ. (0, 6, 12) .
        JZ RESVER    ; ЕСЛИ ФИКСАЦИЯ ЕСТЬ - СБРОСИТЬ ВЕКТОР ВЕРТ.ПАНОРАМ.
        CPI 6
        JZ RESVER
        CPI 12
        JZ RESVER
ZONHOR: MOV A,H      ; ВЗЯТЬ ИНДЕКС СТОЛБЦА И ПРОВЕРИТЬ ФИКСАЦИЮ ПО ГОРИ-
        CPI 0        ; ЗОНТАЛЬНЫМ ЗОНАМ. (0, 8, 16, 24) .
        JZ RESHOR    ; ЕСЛИ ФИКСАЦИЯ ЕСТЬ - СБРОСИТЬ ВЕКТОР ГОР.ПАНОРАМ.
        CPI 8
        JZ RESHOR
        CPI 16

```

```

JZ RESHOR
CPI 24
JZ RESHOR
SAVEIM: SHLD INDVMS ; ЗАПИСАТЬ ИЗМЕНЕННЫЕ ИНДЕКСЫ ОКНА.
POP PSW ; ВЕРНУТЬ ИСПОЛЬЗУЕМЫЕ РЕГИСТРЫ.
POP H
RET ; ВЫЙТИ ИЗ П/П ПАНОРАМИРОВАНИЯ.
;
RESVER: XRA A ; СБРОСИТЬ ВЕКТОР ВЕРТИКАЛЬНОГО ПАНОРАМИРОВАНИЯ.
STA PANVER
JMP ZONHOR ; ПЕРЕЙТИ НА ПРОВЕРКУ ФИКСАЦИИ ЗОН ПО ГОРИЗОНТАЛИ.
RESHOR: XRA A ; СБРОСИТЬ ВЕКТОР ГОРИЗОНТАЛЬНОГО ПАНОРАМИРОВАНИЯ.
STA PANHOR
JMP SAVEIM ; ПЕРЕЙТИ НА ЗАПИСЬ ИНДЕКСОВ ОКНА.
;
PANVER: DB 00H ; ВЕКТОР ВЕРТИКАЛЬНОГО ПАНОРАМИРОВАНИЯ.
PANHOR: DB 00H ; ВЕКТОР ГОРИЗОНТАЛЬНОГО ПАНОРАМИРОВАНИЯ.
;
; *****
; ПОДПРОГРАММА ИНДЕКСИРОВАНИЯ 2-Х МЕРНОГО
; МАССИВА БАЙТОВ (КОДОВ ЭЛЕМЕНТОВ).
; РАЗМЕРНОСТЬ МАССИВА - 40*22 ЭЛЕМЕНТА.
; *****
;
; ПОДПРОГРАММА ОСУЩЕСТВЛЯЕТ ВЫЧИСЛЕНИЕ АДРЕСА
; ЭЛЕМЕНТА В МАССИВЕ 40*22 ПО ЕГО ИНДЕКСАМ И
; БАЗОВОМУ АДРЕСУ МАССИВА.
;
; ВХОДНЫЕ ПАРАМЕТРЫ: L - ИНДЕКС СТРОКИ,
; H - ИНДЕКС СТОЛБЦА,
;
; ВЫХОДНЫЕ ПАРАМЕТРЫ: HL - АДРЕС ЭЛЕМЕНТА.
;
; ИСПОЛЬЗУЕМЫЕ ДАННЫЕ И КОНСТАНТЫ:
;
; "BAZA" - БАЗОВЫЙ АДРЕС МАССИВА.
;
INDEX: PUSH D ; СОХРАНИТЬ ВСПОМОГАТЕЛЬНЫЕ РЕГИСТРЫ.
PUSH H ; СОХРАНИТЬ ИНДЕКСЫ.
MVI H,0 ; ОСТАВИТЬ В "HL" ТОЛЬКО ИНД.СТРОКИ.
DAD H ; УМНОЖИТЬ ИНДЕКС СТРОКИ НА 8 ПУТЕМ
DAD H ; СДВИГА "HL".
DAD H
MOV D,H ; ПРОДУБИЛИРОВАТЬ РЕЗУЛЬТАТ В "DE" И
MOV E,L
DAD D ; УМНОЖИТЬ ЕЩЕ НА 5 СЛОЖЕНИЕМ "HL" И "DE".
DAD D
DAD D
DAD D
XCHG ; ПОМЕСТИТЬ УМНОЖЕН. НА 40 ИНДЕКС СТРОКИ В "DE".
POP H ; ВЕРНУТЬ ИНДЕКСЫ И
MOV L,H ; ВЫДЕЛИТЬ В "HL" ТОЛЬКО
MVI H,0 ; ИНДЕКС СТОЛБЦА.
DAD D ; ПОЛУЧИТЬ В "HL" СМЕЩЕНИЕ В МАССИВЕ ОТ БАЗОВОГО АДР.
LXI D,BAZA ; ПОМЕСТИТЬ В "DE" БАЗОВЫЙ АДРЕС МАССИВА.
DAD D ; СЛОЖИТЬ СО СМЕЩЕНИЕМ И ПОЛУЧИТЬ АДРЕС ЭЛЕМЕНТА В "HL"
POP D ; ВЕРНУТЬ ЗНАЧЕНИЕ "DE" И
RET ; ВЫЙТИ ИЗ П/П "INDEX".
;
;
; *****
; ГРАФИЧЕСКИЙ ИНТЕРФЕЙС ВЫВОДА НА ЭКРАН ОТОБРАЖАЕМОГО УЧАСТКА
; МАССИВА (ОКНА), ИЗ ОСНОВНОГО МАССИВА ЭЛЕМЕНТОВ.
; *****
;
; ВХОДНЫЕ ПАРАМЕТРЫ: В ЯЧЕЙКЕ "MVIDEO" ДОЛЖЕН
; НАХОДИТСЯ БАЗОВЫЙ АДРЕС ОТОБРАЖАЕМОГО
; УЧАСТКА. ЭТОТ ЭЛЕМЕНТ БУДЕТ ОТОБРАЖЕН
; СПРАЙТОМ В ЛЕВОМ НИЖНЕМ УГЛУ ЭКРАНА И ОТНО-
; СТЕЛЬНО НЕГО БУДЕТ ВЫВЕДЕНО ОКНО. ОСНОВНОЙ ЧАСТЬЮ
; ИНТЕРФЕЙСА ВЫВОДА ОКНА ЯВЛЯЕТСЯ ДРАЙВЕР

```



```

; ВЫВОДА СПРАЙТА "OUTSPR".
;   ДЛЯ УЛУЧШЕНИЯ ЗРИТЕЛЬНОГО ВОСПРИЯТИЯ
;   ПРИ ПАНОРАМИРОВАНИИ ЭКРАНА, МЕНЯЕТСЯ ТИП
;   СПРАЙТОВОЙ РАЗВЕРТКИ В ЗАВИСИМОСТИ ОТ НАПРАВЛЕНИЯ
;   ПАНОРАМИРОВАНИЯ: ПРИ ГОРИЗОНТАЛЬНОМ ПАНОРАМИРОВАНИИ
;   ИСПОЛЬЗУЕТСЯ РАЗВЕРТКА ПО СТОЛБЦАМ, А ПРИ ВЕРТИКАЛЬ-
;   НОМ - ПО СТРОКАМ.
;   ИСПОЛЬЗУЕМЫЕ ДАННЫЕ И КОНСТАНТЫ:
;
;   "MVIDEO" - НАЧАЛЬНЫЙ АДРЕС (БАЗА) ОТОБРАЖАЕМОГО УЧАСТКА МАССИВА.
;   "SCREEN" - НАЧАЛЬНЫЙ АДРЕС ДЛЯ ВЫВОДА СПРАЙТОВ НА ЭКРАН.
;   "TIPVMS" - ПЕРЕКЛЮЧАТЕЛЬ ТИПА СПРАЙТОВОЙ РАЗВЕРТКИ.
;   "LINEMS" - КОЛИЧЕСТВО ЭЛЕМЕНТОВ В СТРОКЕ ОСНОВНОГО МАССИВА.
;   "LENGTH" - КОЛИЧЕСТВО ЭЛЕМЕНТОВ В СТРОКЕ ОКНА (ШИРИНА ОКНА).
;   "HEIGHT" - КОЛИЧЕСТВО ЭЛЕМЕНТОВ В СТОЛБЦЕ ОКНА (ВЫСОТА ОКНА).
;
VMSOUT: LHLD MVIDEO ; ВЗЯТЬ БАЗОВЫЙ АДРЕС ОТОБР. УЧАСТКА
        XCHG          ; МАССИВА И ПОМЕСТИТЬ ЕГО В DE.
STWIND: LXI H,SCREEN  ; АДРЕС ВЫВОДА ПЕРВОГО СПРАЙТА НА ЭКРАН.
        LDA TIPVMS    ; ВЗЯТЬ ТИП СПРАЙТОВОЙ РАЗВЕРТКИ И ПРОВЕРИТЬ:
        ORA A          ; "0"-РАЗВЕРТКА ПО СТРОКАМ, "1"-ПО СТОЛБЦАМ.
        JNZ STRVMS     ; ПЕРЕЙТИ НА РАЗВЕРТКУ ПО СТОЛБЦАМ ЕСЛИ НЕ "0".
        MVI B,HEIGHT   ; УСТАНОВИТЬ СЧЕТЧИК ВЫВОДИМЫХ СТРОК ОКНА.
VMS2:   MVI C,LENGTH   ; УСТАНОВИТЬ СЧЕТЧИК ВЫВОДИМЫЙ СТОЛБЦОВ ОКНА.
VMS1:   SHLD ADSPR      ; УСТ. АДРЕС ВЫВОДИМОГО СПРАЙТА (ДЛЯ OUTSPR).
        LDAX D          ; КОД ЭЛЕМЕНТА ИЗ МАССИВА В АКК.
        PUSH H
        PUSH B
        PUSH D
        CALL OUTSPR    ; ВЫВЕСТИ СПРАЙТ НА ЭКРАН.
        POP D
        POP B
        POP H
        INX D           ; НА АДРЕС СЛЕДУЮЩЕГО ЭЛЕМЕНТА В СТРОКЕ
        INR H           ; НА АДРЕС СЛЕДУЮЩЕГО
        INR H           ; СПРАЙТА НА ЭКРАНЕ.
        DCR C           ; ВСЕ ЭЛЕМЕНТЫ СТРОКИ ВЫВЕДЕНЫ?
        JNZ VMS1        ; ЕСЛИ НЕТ - ПРОДОЛЖАТЬ ВЫВОД СТРОКИ.
        MOV A,E         ; ПЕРЕЙТИ НА СЛЕДУЮЩУЮ СТРОКУ В
        ADI LINEMS-LENGTH ; ОТОБРАЖАЕМОМ УЧАСТКЕ МАССИВА, ТО ЕСТЬ НА
        MOV E,A         ; СТРОКУ ВЫШЕ В МАССИВЕ 40*22, НО В СТОЛБЦЕ УЧАСТКА.
        JNC STRSPR      ; ЕСЛИ НЕТ ПЕРЕНОСА - НА УВЕЛИЧ.СТРОКИ СПРАЙТОВ.
        INR D           ; КОРРЕКЦИЯ СТ.БАЙТА АДРЕСА ПРИ СЛОЖЕНИИ.
STRSPR: MOV A,H         ; ВЕРНУТЬ АДРЕС ВЫВОДИМОГО СПРАЙТА В НАЧАЛО
        SUI LENGTH*2    ; СТРОКИ СПРАЙТОВ ОКНА.
        MOV H,A         ;
        MOV A,L         ; ПЕРЕЙТИ НА СЛЕДУЮЩУЮ СТРОКУ СПРАЙТОВ.
        ADI 16          ; ВЫСОТА СПРАЙТА = 16 СТРОК РАЗВЕРТКИ ЭКРАНА.
        MOV L,A
        DCR B           ; ВСЕ СТРОКИ ВЫВЕДЕНЫ?
        JNZ VMS2        ; ЕСЛИ НЕТ - УСТАНОВИТЬ СЧ.ЭЛЕМ. В СТРОКЕ И ПРОДОЛЖАТЬ.
        JMP EXSVMS      ; ЕСЛИ ДА - ВЫЙТИ.
STRVMS: MVI B,LENGTH    ; УСТАНОВИТЬ СЧЕТЧИК СТОЛБЦОВ ОКНА
STVMS2: MVI C,HEIGHT     ; УСТАНОВИТЬ СЧЕТЧИК СТРОК ОКНА.
STVMS1: SHLD ADSPR      ; УСТАНОВИТЬ АДРЕС ДЛЯ ВЫВОДА СПРАЙТА НА ЭКРАН.
        LDAX D          ; ВЗЯТЬ В "A" КОД ЭЛЕМЕНТА ИЗ МАССИВА.
        PUSH B          ; СОХРАНИТЬ СЧЕТЧИКИ И ДАННЫЕ В СТЕКЕ.
        PUSH H
        PUSH D
        CALL OUTSPR    ; ВЫВЕСТИ СПРАЙТ НА ЭКРАН.
        POP D           ; ВЗЯТЬ АДРЕС ЭЛЕМЕНТА В МАССИВЕ.
        LXI H,LINEMS    ; РАЗМЕР ШАГА ДЛЯ ПЕРЕХОДА НА СЛ.СТРОКУ (КОЛ.ЭЛ.В СТР.).
        DAD D
        XCHG            ; В DE - АДРЕС СЛЕДУЮЩЕГО ЭЛЕМЕНТА В СТОЛБЦЕ.
        POP H           ; ВЗЯТЬ АДРЕС ДЛЯ ВЫВОДА СПРАЙТА.
        MOV A,L         ; ПЕРЕЙТИ НА АДРЕС ВЫВОДА СЛЕДУЮЩЕГО
        ADI 16          ; СПРАЙТА В СТОЛБЦЕ.
        MOV L,A
        POP B           ; ВЗЯТЬ СЧЕТЧИКИ И ПРОВЕРИТЬ:
        DCR C           ; ВСЕ СТРОКИ (ВСЕ ЭЛ.ОДНОГО СТОЛБЦА) ВЫВЕДЕНЫ?
        JNZ STVMS1      ; ЕСЛИ НЕТ - ПРОДОЛЖАТЬ ВЫВОДИТЬ СТОЛБЕЦ.

```

```

PUSH B      ; СОХРАНИТЬ СЧЕТЧИКИ.
LXI B,LINEMS*HEIGHT ; ВЫЧЕСТЬ ИЗ АДРЕСА ЭЛЕМЕНТА ПРОИЗВ. КОЛ.СТРОК ОКНА
MOV A,E      ; НА КОЛ. ЭЛ. СТРОКИ МАССИВА (ВЕРНУТЬ АДР.НАЧАЛА СТОЛБ)
SUB C
MOV E,A
MOV A,D
SBB B
MOV D,A
INX D      ; ПЕРЕЙТИ НА СЛЕДУЮЩИЙ СТОЛБЕЦ В МАССИВЕ.
LXI B,HEIGHT*16 ; ВЫЧЕСТЬ ИЗ АДРЕСА СПРАЙТА ВЫСОТУ ЭКРАНА В БАЙТАХ,
MOV A,L      ; ЧТОБЫ ВЕРНУТЬ АДРЕС НА НИЖНИЙ СПРАЙТ В СТОЛБЦЕ ЭКРАНА
SUB C
MOV L,A
MOV A,H
SBB B
MOV H,A
INR H
INR H      ; ПЕРЕЙТИ НА СЛЕДУЮЩИЙ СТОЛБЕЦ СПРАЙТОВ В ЭКРАНЕ.
POP B      ; ВЕРНУТЬ СЧЕТЧИКИ И
DCR B      ; ПРОВЕРИТЬ: ВСЕ СТОЛБЦЫ ВЫВЕДЕНЫ?
JNZ STVMS2  ; ЕСЛИ НЕТ - НАЧАТЬ ВЫВОД СЛЕДУЮЩЕГО СТОЛБЦА.
EXSVMS: JMP MULTIC
;
MVIDEO: DW 0000H      ; АДРЕС БАЗОВОГО ЭЛЕМЕНТА ОТОБРАЖАЕМОГО УЧАСТКА.
SCREEN: EQU 0E030H    ; АДРЕС НАЧАЛА ЭКРАНА ДЛЯ ВЫВОДА СПРАЙТОВ.
TIPVMS: DB 00H       ; ТИП СПРАЙТОВОЙ РАЗВЕРТКИ.
;
LINEMS: EQU 0040      ; КОЛИЧЕСТВО ЭЛЕМЕНТОВ В СТРОКЕ ОСНОВНОГО МАССИВА.
LENGTH: EQU 0016     ; ШИРИНА ОКНА.
HEIGHT: EQU 0010     ; ВЫСОТА ОКНА.
;
;*****
; ВЫВОД ЗАСТАВКИ С ДЕВУШКОЙ.
;*****
;
GIRL:  LXI H,FONPAL    ; УСТАНОВИТЬ НЕВИДИМУЮ ПАЛИТРУ ДЛЯ БЫСТРОГО
CALL MOVPAL ; ИСЧЕЗНОВЕНИЯ СТАРОГО ИЗОБРАЖЕНИЯ.
MVI A,1
STA WRPAL
HLT
CALL CLEAR ; ОЧИСТИТЬ ЭКРАН.
LXI H,GIRPAL ; АДРЕС ПАЛИТРЫ ЦВЕТОВ РЕКЛАМЫ С ДЕВУШКОЙ.
CALL MOVPAL ; ПЕРЕСЛАТЬ ПАЛИТРУ В РАБ.ОБЛАСТЬ.
MVI A,1 ; РАЗРЕШИТЬ ЗАПИСЬ ПАЛИТРЫ В ОЗУ ЦВЕТ.ГЕНЕР.
STA WRPAL
EI
HLT      ; ДОЖДАТЬСЯ ПРЕРЫВАНИЯ ДЛЯ УСТАНОВКИ ЦВЕТОВ.
DI
CALL SHIRMA ; СКРЫТЬ ШИРМОЙ БЫСТРЫЙ ВЫВОД РЕКЛАМЫ.
LXI H,GIRGRF ; НАЧАЛО АРХИВИРОВАННОЙ РЕКЛАМЫ С ДЕВУШКОЙ.
LXI D,0A000H ; АДРЕС ВЫВОДА.
MVI A,22H ; НАЧАЛЬНЫЙ КОД ШИФРА.
STA SHIFR
CALL REARCH ; РАЗОРХИВИРОВАТЬ ГРАФИКУ В ЭКРАН.
CALL RNDPIC ; ОТКРЫТЬ ШИРМУ ПО ПСЕВДО СЛУЧАЙНОМУ ЗАКОНУ.
LXI B,0FFFFH ; ЗАДЕРЖКА, ДЛЯ ПОКАЗА ГРАФИКИ
DELGRF: DCX B
XTHL
XTHL
XTHL
XTHL
XTHL
XTHL
MOV A,B
ORA C
JNZ DELGRF
EI
HLT
STPGIR: LDA STATUS ; ПРОВЕРИТЬ НАЖАТИЕ НА ЛЮБУЮ КЛАВИШУ.
CPI 0FFH
JZ STPGIR

```

```

CALL RNDPIC ; ЗАКРЫТЬ ШИРМУ ПО ПСЕВДОСЛУЧАЙНОМУ ЗАКОНУ.
LXI H,FONPAL
CALL MOVPAL
MVI A,1
STA WRPAL
CALL CLEAR ; СТЕРЕТЬ ДЕВУШКУ.
LXI H,GAMPAL ; АДРЕС ИГРОВОЙ ПАЛИТРЫ ЦВЕТОВ.
CALL MOVPAL
MVI A,1
STA WRPAL
EI
HLT
LDA LIVMEN ; ПРОВЕРИТЬ - ЭТО КОНЕЦ ИГРЫ?
ORA A
JNZ NSTART ; ЕСЛИ НЕТ - ГРУЗИТЬ СЛЕДУЮЩИЙ УРОВЕНЬ.
JMP START ; ЕСЛИ ДА - НАЧАТЬ ИГРУ СНАЧАЛА.
;
;
GIRGRF: EQU 2A00H ; НАЧАЛО ШИФРОВАННОГО АРХИВА РЕКЛ. С ДЕВУШКОЙ.
;
MOVPAL: LXI D,COLOR ; УСТАНОВКА ПАЛИТРЫ (В HL-НАЧ.АДРЕС ПАЛИТРЫ).
LXI B,10H
MOVBLK: MOV A,M ; ПЕРЕСЫЛКА БЛОКА (HL-ОТКУДА, DE-КУДА, BC-ДЛИНА).
STAX D
INX H
INX D
DCX B
MOV A,B
ORA C
JNZ MOVBLK
RET
;
SHIRMA: DI ; ЗАПОЛНИТЬ ПЛОСКОСТЬ 8000-9FFF КОДОМ FF, Т.Е.
LXI H,0 ; СДЕЛАТЬ ШИРМУ ПЕРЕД ПОКАЗОМ РЕКЛАМЫ.
DAD SP
LXI SP,0A000H
LXI B,200H
LXI D,0FFFFH
SHIRM1: PUSH D
PUSH D
PUSH D
PUSH D
PUSH D
PUSH D
PUSH D
PUSH D
DCX B
MOV A,B
ORA C
JNZ SHIRM1
SPHL
EI
RET
;
; *****
; СТИРАНИЕ/ЗАПОЛНЕНИЕ 8-МИ КИЛОБАЙТНОЙ ОБЛАСТИ
; (ПЛОСКОСТИ ЭКРАНА) ПО ПСЕВДОСЛУЧАЙНОМУ
; АЛГОРИТМУ.
; *****
;
; СТИРАНИЕ/ЗАПОЛНЕНИЕ ПРОИЗВОДИТСЯ ПО ТОЧКАМ (ПИКСЕЛЯМ).
; ПОЛОЖЕНИЕ ТОЧКИ ОПРЕДЕЛЯЕТСЯ ПО ИНДЕКСАМ СТРОКИ И
; СТОЛБЦА (ДИАПАЗОН ИНДЕКСОВ 0-255). ФОРМИРОВАНИЕ ИНДЕКСОВ
; ПРОИЗВОДИТСЯ ПРОГРАММОЙ ПСК НА БАЗЕ 16-ТИ РАЗРЯДНОГО
; СДВИГОВОГО РЕГИСТРА С ВЫНЕСЕННЫМИ СУММАТОРАМИ:
; ВХОД РЕГИСТРА = ИНВЕРСИЯ ОБЩЕЙ СУММЫ #(6+8)+(11+15)#.
; МЛАДШИЙ РАЗРЯД СЧИТАТЬ НУЛЕВЫМ. ИСХОДНАЯ КОНСТАНТА ДЛЯ
; ФОРМИРОВАНИЯ ПОЛНОЙ ПОСЛЕДОВАТЕЛЬНОСТИ - 0000.
;
; ПРОГРАММА ФОРМИРУЕТ НЕПОВТОРЯЮЩИЕСЯ АДРЕСА 65535 ТОЧЕК
; ЗА 65535 ШАГОВ. ЗАПИСЬ ТОЧКИ ПРОИЗВОДИТСЯ ЧЕРЕЗ ИСКЛ. ИЛИ,

```

```

; ЧТО ПОЗВОЛЯЕТ ИСПОЛЬЗОВАТЬ ОДИН И ТОТ ЖЕ АЛГОРИТМ ДЛЯ
; СТИРАНИЯ И ЗАПОЛНЕНИЯ ПЛОСКОСТИ-ШИРМЫ.
;
RNDPIC: LXI H,0      ; ИСХОДНАЯ КОНСТАНТА ДЛЯ ФОРМИРОВАНИЯ ПСК.
RNDPC1: XCHG        ; СОХРАНИТЬ В DE. (ПРИ ЦИКЛЕ).
        MOV H,D      ; ВЗЯТЬ В HL, НЕ ТЕРЯЯ DE.
        MOV L,E
RNDTN1: MOV A,H      ; ВЫЧИСЛЕНИЕ НОМЕРА СТОЛБЦА ЭКРАНА ПО ИНДЕКСУ
RRC      ; СТОЛБЦА ПУТЕМ ДЕЛЕНИЯ НА 8.
RRC
RRC
ANI 1FH      ; СТОЛБЕЦ ЭКРАНА В 5-ТИ МЛ.БИТАХ (Т.К НА ЭКРАНЕ
MOV B,A      ; 32 СТОЛБЦА. СОХРАНИТЬ СТОЛБЕЦ В "В".
MOV A,H      ; ВЫДЕЛИТЬ СДВИГ ПИКСЕЛЯ В СТОЛБЦЕ ЭКРАНА (В БАЙТЕ) И
ANI 07      ; ПОКА ДЕРЖАТЬ В "А".
MOV H,B      ; ВЗЯТЬ СТОЛБЕЦ ЭКРАНА В "Н" (СТРОКА ЭКРАНА В "L").
LXI B,8000H  ; БАЗОВЫЙ АДРЕС (НАЧАЛО) ЗАПОЛНЯЕМОЙ ОБЛАСТИ.
DAD B      ; ПОЛУЧИТЬ В "HL" АДРЕС ЭКРАННОГО БАЙТА.
MOV C,A      ; ПОМЕСТИТЬ В "С" СЧЕТЧИК СДВИГОВ В БАЙТЕ.
ORA A      ; ЕСЛИ СЧЕТЧИК РАВЕН НУЛЮ, ТО
MVI A,80H   ; ЭТО САМАЯ ЛЕВАЯ ТОЧКА В ЭТОМ БАЙТЕ И
JZ PICS1    ; СРАЗУ ПЕРЕЙТИ НА ЕЕ ВЫВОД.
PICS2: RRC   ; ЕСЛИ НЕТ - СДВИГАТЬ ДО НУЖНОГО ПОЛОЖЕНИЯ.
DCR C
JNZ PICS2
PICS1: XRA M      ; ЗАПИСАТЬ ТОЧКУ В ЭКРАН ЧЕРЕЗ ИСКЛ. ИЛИ.
MOV M,A
XCHG      ; ВЗЯТЬ ТЕКУЩЕЕ СОСТОЯНИЕ СДВИГОВОГО РЕГИСТРА ПСК.
MOV A,L    ; ПОМЕСТИТЬ 6-Й РАЗРЯД В 0-Й РАЗРЯД АККУМ. СДВИНУВ
RLC      ; МЛ.БАЙТ ДВА РАЗА ВЛЕВО.
RLC
XRA H      ; СУММИРОВАТЬ ПО МОДУЛЮ 2 С 8-М РАЗРЯДОМ.
MOV C,A    ; В МЛ.РАЗРЯДЕ "С" - СУММА 6 И 8 РАЗРЯДОВ.
MOV A,H    ; ПОМЕСТИТЬ 15-Й РАЗРЯД В 0-Й РАЗРЯД АККУМ.
RLC
MOV B,A    ; СОХРАНИТЬ В "В".
RRC      ; ПОМЕСТИТЬ 11-Й РАЗРЯД В 0-Й РАЗРЯД АККУМ. СДВИНУВ
RRC      ; 4 РАЗА.
RRC
RRC
XRA B      ; В МЛ. РАЗРЯДЕ АКК. - СУМА 11 И 15 РАЗРЯДОВ.
XRA C      ; ПОЛУЧИТЬ СУММУ #(6+8)+(11+15)# РАЗР. РЕГИСТРА ПСК.
CMA      ; ИНВЕРТИРОВАТЬ И
ANI 01     ; ВЫДЕЛИТЬ МЛ.РАЗРЯД (ИМЕННО В НЕМ БИТ ОБР.СВЯЗИ ПСК).
DAD H      ; СДВИНУТЬ РЕГИСТР ПСК И
ORA L      ; ПОДАТЬ НА ЕГО ВХОД БИТ ОБР.СВЯЗИ.
MOV L,A
MOV A,H    ; ПРОВЕРИТЬ КОНЕЦ ВСЕЙ ПОСЛЕДОВАТЕЛЬНОСТИ ПО НУЛЕВОМУ
ORA L      ; ЗНАЧЕНИЮ РЕГИСТРА ПСК.
JNZ RNDPC1
RET      ; ЕСЛИ 0, ТО ПРОШЛО 65535 ЦИКЛОВ. (ВСЯ ДЛИНА ПСП)-ВЫЙТИ
;
; ПАЛИТРА ЦВЕТОВ РЕКЛАМЫ С ДЕВУШКОЙ.
;
GIRPAL: DB 00,38H,2FH,3FH,0C9H,0FH,0FFH,0CFH
        DB 00,00,00,00,00,00,00,00
;
; ПАЛИТРА ЦВЕТОВ ИГРЫ.
;
GAMPAL: DB 00,00,00,00,00,7BH,00,00,00,0FFH
        DB 00,00,00,0DEH,00,00,00,00
;
; НЕВИДИМАЯ ПАЛИТРА.
;
FONPAL: DB 00,00,00,00,00,00,00,00,00,00
        DB 00,00,00,00,00,00,00,00,00,00
;
; *****
; ДРАЙВЕР ВЫВОДА СПРАЙТА
; *****
;

```

```

; ВХОДНЫЕ ПАРАМЕТРЫ ДЛЯ ДРАЙВЕРА:
;
;   В "А" - КОД ЭЛЕМЕНТА, В ЯЧЕЙКАХ "ADRSР" - АДРЕС ДЛЯ
; ВЫВОДА СПРАЙТА В ЭКРАН (ДЛЯ НИЖНЕГО БАЙТА ЛЕВОГО СТОЛБИКА) .
; АДРЕС В ЯЧЕЙКЕ ПОСЛЕ РАБОТЫ ДРАЙВЕРА НЕ СОХРАНЯЕТСЯ.
;
; ИСПОЛЬЗУЕМЫЕ ДАННЫЕ И КОНСТАНТЫ:
;
; "BAZATR" - НАЧАЛЬНЫЙ АДРЕС МАССИВА АТТРИБУТОВ ЭЛЕМЕНТОВ.
; "STPLOS" - ЧИСЛО ИСПОЛЬЗУЕМЫХ ПЛОСКОСТЕЙ.
; "BAZGRF" - НАЧАЛО ЗНАКОГЕНЕРАТОРА СПРАЙТОВ.
; "MULT" - КОД ПЕРВОГО ЭЛЕМЕНТА, С КОТОРОГО НАЧИНАЕТСЯ МУЛЬТИПЛИКАЦИЯ.
; "BUFSP" - БУФЕР ДЛЯ ВРЕМЕННОГО ХРАНЕНИЯ ЗНАЧЕНИЯ УКАЗАТЕЛЯ СТЕКА.
; "CHMUL" - ЧАСТОТА МУЛЬТИПЛИКАЦИИ.
; "STMUL" - СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕКЛЮЧЕНИЯ ФАЗЫ МУЛЬТИПЛИКАЦИИ.
; "FAZMUL" - ШАБЛОН НОМЕРОВ ФАЗ МУЛЬТИПЛИКАЦИИ.
;
;
OUTSPR: ORA A          ; НУЛЕВОЙ КОД ?
        JZ VACSPR      ; ЕСЛИ ДА - НА ВЫВОД ПУСТОГО СПРАЙТА.
        ANI 0CFH       ; ИСКЛ. БИТЫ ОПОРЫ БАБОЧКИ.
        CPI WALL       ; СТЕНКА?
        JNZ F1
        LXI D,BAZGRF+20H ; АДРЕС ГРАФИКИ СТЕНКИ.
        MVI B,3         ; ЦВЕТ СТЕНКИ.
        JMP ZERSPR      ; НА ВЫВОД СПРАЙТА.
F1: CPI SOIL           ; ГРУНТ ?
        JNZ F2
        LXI D,BAZGRF+40H
COLSOI: MVI B,2        ; ЦВЕТ ГРУНТА МЕНЯЕТСЯ КОМАНДОЙ "STA COLSOI+1".
        JMP ZERSPR
F2: MVI B,1           ; ЦВЕТ ВСЕХ ОСТАЛЬНЫХ ЭЛЕМЕНТОВ - 1.
        CPI STONE      ; КАМЕНЬ?
        JNZ F3
        LXI D,BAZGRF+60H
        JMP ZERSPR
F3: MOV L,A           ; СОХРАНИТЬ ВРЕМЕННО КОД В "L"
        LDA FAZMUL     ; ВЗЯТЬ ШАБЛОН ФАЗ МУЛЬТИПЛИКАЦИИ,
        ANI 3          ; ВЫДЕЛИТЬ ФАЗУ И
        MOV C,A        ; ПОМЕСТИТЬ В РЕГ. "C".
        MOV A,L
        CPI CRYSTL     ; АЛМАЗ ?
        JNZ F4
        MOV A,C
        ORA A          ; ФАЗА НУЛЕВАЯ?
        JNZ F31        ; ЕСЛИ НЕТ - ПРОВЕРИТЬ 1-Ю.
        LXI D,BAZGRF+80H
        JMP ZERSPR
F31: DCR A
        JNZ F32
        LXI D,BAZGRF+0A0H
        JMP ZERSPR
F32: LXI D,BAZGRF+0C0H
        JMP ZERSPR
F4: CPI BUTTER       ; БАБОЧКА?
        JNZ F5
        MOV A,C
        ORA A
        JNZ F41
        LXI D,BAZGRF+1A0H
        JMP ZERSPR
F41: DCR A
        JNZ F42
        LXI D,BAZGRF+1C0H
        JMP ZERSPR
F42: LXI D,BAZGRF+1E0H
        JMP ZERSPR
F5: CPI EXPLOS       ; ВЗРЫВ?
        JNZ F6
        MOV A,C
        ORA A

```

```

JNZ F51
LXI D,BAZGRF+140H
JMP ZERSPR
F51: DCR A
JNZ F52
LXI D,BAZGRF+160H
JMP ZERSPR
F52: LXI D,BAZGRF+180H
JMP ZERSPR
F6: CPI EXIT ; ВЫХОД?
JNZ F7
MOV A,C
ORA A
JNZ F61
LXI D,BAZGRF+0E0H
JMP ZERSPR
F61: DCR A
JNZ F62
LXI D,BAZGRF+100H
JMP ZERSPR
F62: LXI D,BAZGRF+120H
JMP ZERSPR
;
; ОБРАБОТКА КОДА ЧЕЛОВЕЧКА ПО ЕГО АТРИБУТУ.
;
F7: MOV C,A
LDA ATRMEN
RRC ; СДВИНУТЬ БАЙТ АТРИБУТОВ 4 РАЗА, ЧТОБЫ
RRC ; ПОЛУЧИТЬ ТРИ БИТА ЗНАЧЕНИЯ ВЕКТОРА В
RRC ; МЛАДШИХ РАЗРЯДАХ.
RRC
ANI 07H ; ВЫДЕЛИТЬ КОД ВЕКТОРА.
MOV E,A ; УМНОЖИТЬ КОД ВЕКТОРА НА ТРИ
ADD A ; И СОХРАНИТЬ УТРОЕННОЕ
ADD E ; ЗНАЧЕНИЕ ВЕКТОРА В
MOV E,A ; В РЕГИСТРЕ "Е".
MOV A,C ; ВЗЯТЬ В "А" КОД ЭЛЕМЕНТА И
SUI MULT ; ВЫЧЕСТЬ КОД ПЕРВОГО ЭЛЕМЕНТА С МУЛЬТИПЛИКАЦИЕЙ.
MOV C,A ; УТРОИТЬ ПОЛУЧЕННОЕ ЗНАЧЕНИЕ И
ADD A
ADD C
ADI MULT ; СЛОЖИТЬ С КОДОМ ПЕРВОГО ЭЛ. С МУЛЬТИПЛИКАЦИЕЙ.
ADD E ; ПРИБАВИТЬ К ЭТОМУ РЕЗУЛЬТАТУ УТРОЕННОЕ ЗНАЧЕ-
MOV C,A ; НИЕ ВЕКТОРА И СОХР. В "С" (ЭТО 1-Й СПРАЙТ ИЗ 3).
LDA FAZMUL ; ВЗЯТЬ ШАБЛОН НОМЕРОВ ФАЗ МУЛЬТИПЛИКАЦИЙ И
ANI 03H ; ВЫДЕЛИТЬ НОМЕР ФАЗЫ ИЗ ДВУХ МЛАДШИХ РАЗРЯДОВ.
ADD C ; СЛОЖИТЬ С "С" И ТЕМ САМЫМ ПОЛУЧИТЬ НОМЕР СПРАЙТА
MOV C,A ; ДЛЯ ОТОБРАЖЕНИЯ ЭЛЕМЕНТА. СОХРАНИТЬ В "С".
;
; ВЫЧИСЛЕНИЕ АДРЕСА НАЧАЛА ГРАФИКИ СПРАЙТА ПО ЕГО НОМЕРУ.
; (В ДАННОМ ДРАЙВЕРЕ РАБОТАЕТ ТОЛЬКО ДЛЯ ЧЕЛОВЕЧКА, ДЛЯ ОСТАЛЬНЫХ ФИГУР
; ОПРЕДЕЛЕНЫ АБСОЛЮТНЫЕ АДРЕСА ГРАФИКИ ДЛЯ УВЕЛИЧЕНИЯ БЫСТРОДЕЙСТВИЯ)
;
ADRGRF: MOV L,C ; ПОДГОТОВИТЬ "HL" ДЛЯ УМНОЖЕНИЯ НОМЕРА СПРАЙТА НА
MVI H,0 ; 32 (ЧИСЛО БАЙТ ГРАФИКИ В СПРАЙТЕ), ЧТОБЫ ВЫЧИСЛИТЬ
DAD H ; СМЕЩЕНИЕ СПРАЙТА ОТ НАЧАЛА ЗНАКОГЕНЕРАТОРА.
DAD H
DAD H
DAD H
DAD H ; УМНОЖИЛИ НА 32.
LXI D,BAZGRF ; В "DE" УСТАНОВИТЬ НАЧАЛЬНЫЙ АДРЕС ЗГ. И СЛОЖИТЬ
DAD D ; СО СМЕЩЕНИЕМ В "HL".
XCHG ; ПОЛУЧЕННЫЙ АДРЕС НАЧАЛА ГРАФИКИ СПРАЙТА СОХР. В "DE".
ZERSPR: LDA STPLOS ; ВЗЯТЬ ЧИСЛО ИСПОЛЪЗУЕМЫХ ПЛОСКОСТЕЙ.
;
; НАЧАЛО ЦИКЛА ВЫВОДА СПРАЙТА НА ЭКРАН ПЭВМ "ВЕКТОР-06Ц" ПО ПЛОСКОСТЯМ.
;
PUSH PSW ; СОХРАНИТЬ СЧЕТЧИК ЧИСЛА ПЛОСКОСТЕЙ.
PLOS: PUSH D ; СОХРАНИТЬ АДРЕС НАЧАЛА ГРАФИКИ ВЫВОДИМОГО СПРАЙТА.
LXI H,0 ; ПОДГОТОВИТЬСЯ К ВЗЯТИЮ ЗНАЧЕНИЯ "SP" В "HL".
DAD SP ; ВЗЯТЬ ЗНАЧЕНИЕ "SP" ПУТЕМ СЛОЖЕНИЯ С "HL" И

```

```

SHLD BUFSP ; СОХРАНИТЬ ЕГО В БУФЕРЕ.
XCHG      ; ВЗЯТЬ В "HL" АДРЕС НАЧАЛА ГРАФИКИ СПРАЙТА И
DI
SPHL      ; НАСТРОИТЬ НА НЕГО УКАЗАТЕЛЬ СТЕКА.
LHLD ADSPR ; ВЗЯТЬ В "HL" АДРЕС ДЛЯ ВЫВОДА СПРАЙТА В ЭКРАН.
MVI C,2    ; УСТАНОВИТЬ СЧЕТЧИК СТОЛБЦОВ В СПРАЙТЕ.
MOV A,B    ; ВЗЯТЬ В "A" БАЙТ АТРИБУТОВ И
RRC        ; ПРОВЕРИТЬ БИТ ВЫВОДА СПРАЙТА В ДАННУЮ ПЛОСКОСТЬ.
MOV B,A    ; СОХРАНИТЬ "B" ДЛЯ ПРОВЕРКИ СЛЕДУЮЩИХ БИТ.
JC SAVSPR  ; ЕСЛИ ПРОВЕРЯЕМЫЙ БИТ "1", ТО ВЫВОДИТЬ СПРАЙТ.
LXI SP,BAZGRF ; ЕСЛИ "0", ТО ВЫВОДИТЬ "НУЛЕВОЙ СПРАЙТ" ИЗ НАЧАЛА ЗГ.
;
; ВЫВОД СПРАЙТА В ЭКРАН ИДЕТ ПО СТОЛБИКАМ, ПРИЧЕМ ДЛЯ ПОВЫШЕНИЯ
; БЫСТРОДЕЙСТВИЯ ВЫВОД ОДНОГО СТОЛБИКА СДЕЛАН "В ЛОВ".
;
SAVSPR: POP D      ; ВЗЯТЬ ДВА БАЙТА ГРАФИКА ИЗ ЗНАКОГЕНЕРАТОРА.
MOV M,E           ; ЗАПИСАТЬ ПЕРВЫЙ БАЙТ,
INR L             ; ПОДНЯТЬСЯ НА СЛЕДУЮЩУЮ СТРОКУ ЭКРАНА,
MOV M,D           ; ЗАПИСАТЬ ВТОРОЙ БАЙТ,
INR L             ; ПОДНЯТЬСЯ НА СЛЕДУЮЩУЮ СТРОКУ ЭКРАНА.
;
; ДАЛЕЕ ПОВТОРИТЬ ЭТОТ КУСОК ЕЩЕ 7 РАЗ, ДЛЯ ВЫВОДА 16 БАЙТ ОДНОГО СТОЛБИКА.
;
POP D
MOV M,E
INR L
MOV M,D
INR L
POP D
MOV M,E
INR L
MOV M,D
INR L
POP D
MOV M,E
INR L
MOV M,D
INR L
POP D
MOV M,E
INR L
MOV M,D
INR L
POP D
MOV M,E
INR L
MOV M,D
INR L
POP D
MOV M,E
INR L
MOV M,D
INR L
POP D
MOV M,E
INR L
MOV M,D
INR L
POP D
MOV M,E
INR L
MOV M,D
INR L
;
DCR C           ; ВСЕ СТОЛБИКИ СПРАЙТА ВЫВЕДЕНЫ?
JZ NEXPLS      ; ЕСЛИ ДА - ПЕРЕЙТИ НА АДРЕС СЛЕДУЮЩЕЙ ПЛОСКОСТИ.
MOV A,L        ; ВЗЯТЬ ЗНАЧЕНИЕ СТРОКИ ЭКРАНА И
SUI 16         ; ВЕРНУТЬ ЕЕ НА НИЖНИЙ (ДЛЯ ЭКРАНА) КОНЕЦ СТОЛБИКА.
MOV L,A
INR H          ; ЗНАЧЕНИЕ СТОЛБЦА ЭКРАНА НА ВЫВОД СЛЕДУЮЩЕГО СТОЛБИКА.
JMP SAVSPR     ; ПРОДОЛЖАТЬ ЗАПИСЫВАТЬ СПРАЙТ В ЭКРАН.
;
; ПЕРЕХОД НА СЛЕДУЮЩУЮ ПЛОСКОСТЬ.
;
NEXPLS: LHLD BUFSP ; ВЗЯТЬ СТАРОЕ ЗНАЧЕНИЕ "SP" ИЗ БУФЕРА.
SPHL

```

```

EI
POP D      ; ВЕРНУТЬ В "DE" АДРЕС НАЧАЛА ГРАФИКИ ДАННОГО СПРАЙТА.
POP PSW    ; ВЕРНУТЬ В "A" ЗНАЧЕНИЕ СЧЕТЧИКА ПЛОСКОСТЕЙ.
DCR A      ; ВСЕ ИСПОЛЬЗУЕМЫЕ ПЛОСКОСТИ ВЫВЕДЕНЫ?
;-----
RZ          ; ВЫЙТИ ИЗ ДРАЙВЕРА ВЫВОДА СПРАЙТА "OUTSPR".
;-----
PUSH PSW    ; СОХРАНИТЬ СЧЕТЧИК ПЛОСКОСТЕЙ В СТЕКЕ.
LHLD ADSPR  ; ВЗЯТЬ АДРЕС ВЫВОДИМОГО СПРАЙТА В "HL".
MOV A,H     ; СТАРШИЙ БАЙТ АДРЕСА
SUI 20H     ; УСТАНОВИТЬ НА СЛЕДУЮЩУЮ ПЛОСКОСТЬ.
MOV H,A
SHLD ADSPR  ; ЗАПИСАТЬ НОВОЕ ЗНАЧЕНИЕ АДРЕСА ВЫВОДА СПРАЙТА.
JMP PLOSK   ; ВЫВОДИТЬ СПРАЙТ В СЛЕДУЮЩУЮ ПЛОСКОСТЬ.
;
VACSPR: LXI H,0      ; ВЫВОД "ПУСТОГО" СПРАЙТА. (СДЕЛАН ОТДЕЛЬНО ПО
DAD SP          ; БЫСТРОМУ АЛГОРИТМУ ДЛЯ ПОВЫШЕНИЯ БЫСТРОДЕЙСТВИЯ).
SHLD BUFSP
LHLD ADSPR
MOV A,L
ADI 16
MOV L,A
MOV D,H
MOV E,L
LXI B,0
LDA STPLOS
DI
VACSP1: SPHL
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
INR H
SPHL
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
PUSH B
DCR A
JZ RVACSP
MOV A,H
SUI 20H
MOV H,A
DCR H
MVI A,1
JMP VACSP1
RVACSP: LHLD BUFSP
SPHL
EI
RET
;
; ПОДПРОГРАММА ПЕРЕКЛЮЧЕНИЯ ФАЗ МУЛЬТИПЛИКАЦИИ.
;
MULTIC: LDA STMUL    ; ВЗЯТЬ СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕКЛЮЧЕНИЯ ФАЗ.
DCR A              ; УМЕНЬШИТЬ И ЗАПИСАТЬ
STA STMUL          ; ИЗМЕНЕННЫЙ РЕЗУЛЬТАТ.
RNZ                ; ВЫЙТИ ИЗ "MULTIC" ЕСЛИ ЗАДЕРЖКА НЕ ОТРАБ.
LDA CHMUL          ; ВЗЯТЬ КОНСТАНТУ ЧАСТОТЫ МУЛЬТИПЛИКАЦИИ И
STA STMUL          ; ЗАГРУЗИТЬ ЕЮ СЧЕТЧИК ЗАДЕРЖКИ.
LDA FAZMUL         ; ВЗЯТЬ ШАБЛОН ФАЗ МУЛЬТИПЛИКАЦИИ И
RRC                ; СДВИНУТЬ ЕГО НА ДВА БИТА Т.К. НОМЕР
RRC                ; ФАЗЫ НАХОДИТСЯ В ДВУХ РАЗРЯДАХ.
STA FAZMUL         ; ЗАПИСАТЬ ИЗМЕНЕННЫЙ ШАБЛОН НА МЕСТО.

```



```

RET      ; ВЫЙТИ ИЗ П/П "MULTIC"
;
;-----
; ОПРЕДЕЛЕНИЕ КОНСТАНТ, ДАННЫХ И МЕТОК, ИСПОЛЬЗУЕМЫХ В "OUTSPR" И
; "MULTIC".
;-----
;
BAZGRF: EQU 1D00H+800H ; АДРЕС НАЧАЛА ЗНАКОГЕНЕРАТОРА СПРАЙТОВ.
MULT:   EQU 04H       ; КОД ЭЛЕМЕНТА, С КОТОРОГО НАЧИНАЕТСЯ МУЛЬТИПЛ.
;
ADRSPR: DW 0000H      ; АДРЕС ДЛЯ ВЫВОДА СПРАЙТА НА ЭКРАН.
STPLOS: DB 01H        ; СЧЕТЧИК ИСПОЛЬЗУЕМЫХ ПЛОСКОСТЕЙ.
CHMUL:  DB 01H        ; ЧАСТОТА МУЛЬТИПЛИКАЦИИ.
FAZMUL: DB 64H        ; ШАБЛОН ФАЗ МУЛЬТИПЛИКАЦИИ.
STMUL:  DB 01H        ; СЧЕТЧИК ЗАДЕРЖКИ ПЕРЕКЛЮЧЕНИЯ ФАЗ.
BUFSP:  DW 0000H      ; БУФЕР ДЛЯ ВРЕМЕННОГО ХРАНЕНИЯ ЗНАЧЕНИЯ "SP".
;
BAZATR: DB 00H        ; НАЧАЛО МАССИВА АТРИБУТОВ ЭЛЕМЕНТОВ. (0-Й НЕ ИСПОЛЬЗ.)
        DB 03H        ; АТРИБУТ СТЕНКИ (ЦВЕТ 3).
        DB 02H        ; АТРИБУТ СЪЕДАЕМОГО УЧАСТКА (ЦВЕТ 2).
        DB 01H        ; АТРИБУТ КАМНЯ (ЦВЕТ 1).
        DB 81H        ; АТРИБУТ КРИСТАЛЛА (ЦВЕТ 1, ПРОСТАЯ МУЛЬТИПЛИКАЦИЯ).
        DB 83H        ; АТРИБУТ ВЫХОДА ИЗ ЛАБ. (ЦВЕТ 3, ПРОСТ.МУЛЬТ.).
        DB 81H        ; АТРИБУТ ВЗРЫВА (ЦВЕТ1, ПРОСТ.МУЛЬТ.).
        DB 81H        ; АТРИБУТ БАБОЧКИ (ЦВЕТ1, ПРОСТ.МУЛЬТ.).
ATRMEN: DB 81H        ; АТРИБУТ ЧЕЛОВЕЧКА (ЦВЕТ1, ВЕКТ.МУЛЬТ. - 6 ВЕКТОРОВ).
        NOP
        NOP
        NOP
        NOP
;
; РАЗАРХИВАТОР-ДЕШИФРАТОР ГРАФИКИ ДЛЯ ВЫВОДА РЕКЛАМЫ.
;
; ВХОДНЫЕ ДАННЫЕ:
; HL-НАЧАЛО АРХИВА
; DE-НАЧАЛО ТЕКСТА
; SHIFR - НАЧАЛЬНАЯ УСТАНОВКА КОДА ДЕШИФРАЦИИ
;
; ВЫХОДНЫЕ ДАННЫЕ:
; DE-КОНЕЦ ТЕКСТА
;
REARCH: MOV A,M
        INX H
        ORA A
        PUSH PSW
        JM  RARC0
        ANI 3FH
        MOV B,A
        ORA M
        JZ  DONE
        MOV C,M
        INX H
        JMP RARC1
RARC0:  MVI B,0
        ANI 3FH
        MOV C,A
RARC1:  POP PSW
        ANI 40H
        JZ  MOV5
        MOV A,M
        INX H
        PUSH PSW
RARC2:  POP PSW
        STAX D
        INX D
        PUSH PSW
        DCX B
        MOV A,B
        ORA C
        JNZ RARC2
        POP PSW

```

```

    JMP REARCH
;
MOVS:  LDA SHIFR    ; ВЗЯТЬ ШАБЛОН ДЕШИФРАЦИИ HEARCH.КУСКА
      RRC          ; СДВИНУТЬ И
      STA SHIFR    ; ЗАПОМНИТЬ.
      STA GETKOD+1  ; УСТАНОВИТЬ КОД ДЛЯ ДЕШИФРАЦИИ.
MOVS1: MOV A,M
GETKOD: XRI 00H     ; ДЕШИФРАЦИЯ HEАРХИВИРОВАННОГО КУСКА.
      STAX D
      INX H
      INX D
      DCX B
      MOV A,B
      ORA C
      JNZ MOVS1
      JMP REARCH
DONE:  POP PSW
      RET
;
SHIFR: DB 0CCH     ; НАЧАЛЬНАЯ УСТАНОВКА КОДА ДЕШИФРАЦИИ HEARCH.КУСКОВ.
;
;
;*****
; ПОДПРОГРАММА ВЫВОДА СООБЩЕНИЯ
;*****
;
; ВХОДНЫЕ ПАРАМЕТРЫ: "H1" - АДРЕС НАЧАЛА СООБЩЕНИЯ, СООБЩЕНИЕ ДОЛЖНО
; ЗАКАНЧИВАТЬСЯ КОДОМ 00H. КОД 0EH РАСШИРЯЕТ 7-МИ БИТНЫЙ КОД ДО 8-МИ
; БИТОВ, КОД 0FH СНИМАЕТ ЭТО РАСШИРЕНИЕ, НО НЕ ОГРАНИЧИВАЕТ 8-МИ
; БИТНЫЕ КОДЫ.
;
TECHOUT: MOV A,M    ; ВЗЯТЬ КОД СИМВОЛА
      ORA A        ; ПРОВЕРИТЬ КОНЕЦ СООБЩЕНИЯ.
      RZ          ; ЕСЛИ КОНЕЦ, ТО ВЫЙТИ.
      CPI 0EH      ; ЭТО КОД РАСШИРЕНИЯ?
      JZ SET8      ; ЕСЛИ ДА - РАСШИРИТЬ.
      CPI 0FH      ; ЗАПРЕТ РАСШИРЕНИЯ?
      JZ RESET8    ; ЕСЛИ ДА - СНЯТЬ РАСШИРЕНИЕ.
      PUSH B
      MOV C,A      ; СОХРАНИТЬ КОД В "C".
      LDA REGSYM   ; ВЗЯТЬ МАСКУ РАСШИРЕНИЯ КОДА.
      ORA C        ; НАЛОЖИТЬ МАСКУ НА КОД.
      POP B
      CALL SYMOUT  ; ВЫВЕСТИ СИМВОЛ.
TXOUT1: INX H      ; АДРЕС СЛЕДУЮЩЕГО СИМВОЛА.
      JMP TECHOUT  ; ПРОДОЛЖАТЬ ВЫВОД.
;
SET8:  MVI A,80H   ; УСТАНОВИТЬ "1" В 7-МОМ БИТЕ МАСКИ РАСШИРЕНИЯ.
      STA REGSYM
      JMP TXOUT1
;
RESET8: XRA A      ; СБРОСИТЬ МАСКУ РАСШИРЕНИЯ.
      STA REGSYM
      JMP TXOUT1
;
REGSYM: DB 00H     ; МАСКА РАСШИРЕНИЯ КОДА.
;
;
;*****
; ДРАЙВЕР ВЫВОДА СИМВОЛА
;*****
;
; ПОДПРОГРАММА ВЫВОДИТ СИМВОЛ МАТРИЦЕЙ 8*8.
; ДЛЯ СКОРОСТИ (НАПРИМЕР ДЛЯ ОТОБРАЖЕНИЯ СЧЕТЧИКОВ ВО ВРЕМЯ ИГРЫ),
; СИМВОЛ ВЫВОДИТСЯ ТОЛЬКО В ОДНУ ИЗ 4-Х ПЛОСКОСТЕЙ.
;
; ВХОДНЫЕ ПАРАМЕТРЫ: "A" - КОД СИМВОЛА, ЯЧЕЙКА "COLSYM" - ЦВЕТ СИМВОЛА,
; (ПРИНИМАЕТ ЗНАЧЕНИЯ СТАРШЕГО БАЙТА АДР.ПЛОСКОСТИ - 80H, A0H, C0H, E0H).
; ЯЧЕЙКИ "PLACE" - МЕСТО ВЫВОДА СИМВОЛА НА ЭКРАН, ТОЧНЕЕ ИНДЕКСЫ СТРОКИ
; И СТОЛБЦА. НУЛЕВЫМ ИНДЕКСАМ СООТВЕТСТВУЕТ ЛЕВЫЙ НИЖНИЙ УГОЛ ЭКРАНА.
; МАКСИМАЛЬНЫЕ ИНДЕКСЫ - 31 (1FH).

```

MOV M, A

```

LHLD PLACE
MOV A,H      ; ВЗЯТЬ ИНДЕКС СТОЛБЦА И
CPI 1FH      ; ПРОВЕРИТЬ: ЭТО КОНЕЦ СТРОКИ?
JNC LINEFR   ; ЕСЛИ ДА - ТО ПЕРЕВЕСТИ СТРОКУ.
INR H        ; ЕСЛИ НЕТ - ТО НА СЛЕДУЮЩЕЕ МЕСТО В СТРОКЕ.
JMP WRPLAC   ; НА ЗАПИСЬ НОВОГО ЗНАКОМЕСТА.
TSWORK: CPI 9      ; ЭТО СЛУЖЕБНЫЕ КОДЫ УСТАНОВКИ ЦВЕТА СИМВОЛА?
JNC TSWRK1   ; ЕСЛИ НЕТ - НА ПРОВЕРКУ "CR" И "LF".
POP D        ; ВЕРНУТЬ РЕГИСТРЫ ДЛЯ УМЕНЬШЕНИЯ ОБЪЕМА ПРОГРАММЫ
POP H        ; ПРИ УСТАНОВКЕ ЦВЕТА (ПОСЛЕ УСТАНОВКИ СРАЗУ RET ).
CPI 1        ; УСТАНОВИТЬ ЦВЕТ 1?
JZ SCOLS1
CPI 2        ; ЦВЕТ 2?
JZ SCOLS2
CPI 4        ; ЦВЕТ 4?
JZ SCOLS4
CPI 8        ; ЦВЕТ 8?
JZ SCOLS8
RET          ; ВЫЙТИ ИЗ П/П ВЫВОДА СИМВОЛА, ЕСЛИ ДРУГОЙ УПР.КОД.
SCOLS1: MVI A,0E0H ; УСТАНОВИТЬ СТАРШИЙ БАЙТ АДРЕСА ПЛОСКОСТИ, КОТОРОЙ
STA COLSYM   ; ПРИСВОЕН 1-Й МАТЕМАТИЧЕСКИЙ ЦВЕТ.
RET
SCOLS2: MVI A,0C0H ; АДРЕС ПЛОСКОСТИ 2-ГО МАТ. ЦВЕТА.
STA COLSYM
RET
SCOLS4: MVI A,0A0H ; АДРЕС ПЛОСКОСТИ 4-ГО МАТ. ЦВЕТА.
STA COLSYM
RET
SCOLS8: MVI A,80H  ; АДРЕС ПЛОСКОСТИ 8-ГО МАТ. ЦВЕТА.
STA COLSYM
RET
TSWRK1: LHLD PLACE ; ВЗЯТЬ ИНДЕКСЫ.
CPI 0DH      ; ЭТО ВОЗВРАТ КАРЕТКИ?
JNZ TSWNEX   ; ЕСЛИ НЕТ - ПРОДОЛЖАТЬ ПРОВЕРКУ ДРУГИХ СЛ.СИМВОЛОВ.
MVI H,0      ; ЕСЛИ ДА - ЗНАКОМЕСТО В НАЧАЛО ДАННОЙ СТРОКИ.
JMP WRPLAC   ; НА ЗАПИСЬ НОВОГО ЗНАКОМЕСТА.
TSWNEX: CPI 0AH  ; ЭТО ПЕРЕВОД СТРОКИ?
JNZ TSWRK2   ; ЕСЛИ НЕТ - ПРОВЕРЯТЬ ВКЛ/ВЫКЛ ИНВЕРСИИ.
LINEFR: MVI H,0 ; ПРИ ПЕРЕВОДЕ СТРОКИ ВЕРНУТЬ КАРЕТКУ.
MOV A,L      ; ВЗЯТЬ ИНДЕКС СТРОКИ И ПРОВЕРИТЬ:
ORA A        ; ЭТО ПОСЛЕДНЯЯ СТРОКА (САМАЯ НИЖНЯЯ В ЭКРАНЕ?).
JZ WRPLAC    ; ЕСЛИ ДА - ЗАПИСАТЬ ИНДЕКСЫ НЕ ВЫПОЛНЯЯ ПЕРЕВОДА СТР.
DCR L        ; ЕСЛИ НЕТ - УМЕНЬШИТЬ ИНДЕКС СТРОКИ.
WRPLAC: SHLD PLACE ; ЗАПИСАТЬ ИНДЕКСЫ НОВОГО ЗНАКОМЕСТА.
EXTSYM: POP D ; ВЫЙТИ ИЗ П/П ВЫВОДА СИМВОЛА.
POP H
RET
TSWRK2: CPI 10H   ; ЭТО КОД ВКЛЮЧЕНИЯ ИНВЕРСИИ?
JZ ONINV       ; ЕСЛИ ДА - ВКЛ.
CPI 11H        ; КОД ВЫКЛЮЧЕНИЯ ИНВЕРСИИ?
JZ OFINV
JMP EXTSYM     ; ВЫЙТИ ИЗ П/П ВЫВОДА СИМВОЛА ЕСЛИ ДРУГИЕ КОДЫ.
ONINV: MVI A,2FH  ; КОД КОМАНДЫ "СМА" ЗАНЕСТИ В ЦИКЛ ВЫВОДА СИМВОЛА
JMP SETINV
OFINV: XRA A     ; КОД КОМАНДЫ "NOR" ЗАНЕСТИ В ЦИКЛ ВЫВОДА СИМВОЛА.
SETINV: PUSH B
LXI H,INVERT   ; В "HL" АДРЕС ПЕРВОГО БАЙТА-КОМАНДЫ ПЕРЕКЛ. ИНВЕРС.
LXI D,5        ; ШАГ МЕЖДУ ЭТИМИ БАЙТАМИ
MVI B,8        ; СЧЕТЧИК ЗАПИСИ БАЙТ-КОМАНД ПЕРЕКЛ.ИНВЕРСИИ
SINVR1: MOV M,A  ; ЗАПИСАТЬ БАЙТ-КОМАНДУ.
DAD D          ; АДРЕС СЛЕДУЮЩЕЙ.
DCR B
JNZ SINVR1
POP B
JMP EXTSYM
;
COLSYM: DB 00H  ; БУФЕР ЦВЕТА СИМВОЛА.
;
PLACE: DW 0000H ; ИНДЕКСЫ ЗНАКОМЕСТА.
;
;

```

```

;*****
; ДРАЙВЕР БЫСТРОЙ ОЧИСТКИ 4-Х ЭКРАННЫХ
; ПЛОСКОСТЕЙ ПЭВМ "ВЕКТОР-06Ц".
;*****
;
;
CLEAR:  DI
        PUSH H          ; СОХРАНИТЬ ИСПОЛЬЗУЕМЫЕ РЕГИСТРЫ.
        PUSH B
        PUSH D
        PUSH PSW
        LXI H,0000H ; ПОДГОТОВИТЬ HL ДЛЯ ВЗЯТИЯ ЗНАЧЕНИЯ SP.
        DAD SP       ; ВЗЯТЬ ЗНАЧЕНИЕ SP ПУТЕМ СЛОЖЕНИЯ С HL.
        LXI SP,0000H ; НАЧАТЬ СТИРАНИЕ С АДРЕСА 0FFFFH.
        LXI D,0000H ; В DE - ЗНАЧЕНИЕ, КОТОРЫМ БУДЕТ СТЕРТ ЭКРАН.
        LXI B,0800H ; В BC - СЧЕТЧИК.
CLEAR1: PUSH D          ; НАЧАТЬ СТИРАТЬ ЭКРАН ПУТЕМ "ОПУСКАНИЯ" СТЕКА.
        PUSH D          ; ВЫПОЛНИТЬ 8 ОПЕРАЦИЙ "В ЛОБ" ДЛЯ СКОРОСТИ.
        PUSH D
        PUSH D
        PUSH D
        PUSH D
        PUSH D
        DCX B           ; УМЕНЬШИТЬ СЧЕТЧИК И
        MOV A,B         ; ПРОВЕРИТЬ ЕГО.
        ORA C           ; КОНЕЦ СЧЕТА?
        JNZ CLEAR1      ; ЕСЛИ НЕТ - ПРОДОЛЖАТЬ СТИРАНИЕ.
        SPHL            ; ВЕРНУТЬ ПРЕЖНЕЕ ЗНАЧЕНИЕ SP И
        POP PSW         ; ВСЕХ РЕГИСТРОВ.
        POP D
        POP B
        POP H
        EI
        RET
;
; ВЫВОД ГРАФИЧЕСКОГО ИЗОБРАЖЕНИЯ "ВОЛГОГРАД ПРОГРАММА И ГРАФИКА ЛЕБЕДЕВ А.З"
;
AVTOR:  LXI H,AVTSCR     ; НАЧАЛО ГРАФИКИ В ЭКРАНЕ.
AVT0:   LXI D,AVTGRF     ; НАЧАЛО ГРАФИКИ В ПАМЯТИ.
        MVI B,24        ; КОЛИЧЕСТВО СТРОК ГРАФИКИ В ЭКРАНЕ (ПОСТРОЧН. ВЫВОД).
AVT1:   MVI C,18         ; КОЛИЧЕСТВО СТОЛБЦОВ.
AVT2:   LDAX D
        MOV M,A
        MOV A,H
        ADI 20H
        MOV H,A
        LDAX D
        MOV M,A
        MOV A,H
        SUI 20H
        MOV H,A
        INX D
        INR H
        DCR C
        JNZ AVT2
        MOV A,H
        SUI 18
        MOV H,A
        INR L
        DCR B
        JNZ AVT1
        RET
;
AVTGRF: EQU 1B00H+800H ; НАЧАЛО ГРАФИКИ АВТОРА В ПАМЯТИ ПРОГРАММЫ.
AVTSCR: EQU 0810BH    ; НАЧАЛО ГРАФИКИ АВТОРА В ЭКРАНЕ.
;
; ТЕКСТЫ СООБЩЕНИЙ В ИГРОВОМ ЭКРАНЕ:
;
TEXT1:  DB 08H         ; ЦВЕТ РАМКИ ДЛЯ НАДПИСЕЙ.
        DB 99H,9DH,9DH,9DH,9DH,9DH,9DH,9DH
        DB 9DH,9DH,9DH,9DH,9DH,9DH,9DH,9DH

```

```

    DB 9DH,9DH,9DH,9DH,9DH,9DH,9DH,9DH
    DB 9DH,9DH,9DH,9DH,9DH,9DH,9DH,8BH
    DB 8AH
    DB 04H      ; ЦВЕТ СООБЩЕНИЯ.
    DB 'УРОВЕНЬ:  АЛМАЗЫ:  ВРЕМЯ:  '
    DB 08H,8AH
    DB 98H,9DH,9DH,9DH,9DH,9DH,9DH,9DH
    DB 9DH,9DH,9DH,9DH,9DH,9DH,9DH,9DH
    DB 9DH,9DH,9DH,9DH,9DH,9DH,9DH,9DH
    DB 9DH,9DH,9DH,9DH,9DH,9DH,9DH,8CH
    DB 00H      ; КОНЕЦ СООБЩЕНИЯ.
;
TEXT2:  DB 08H
        DB 20H,20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H
        DB 99H,9DH,9DH,9DH,9DH,9DH
        DB 9DH,9DH,9DH,9DH,9DH,8BH
        DB 20H,20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H
        DB 8AH,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,8AH
        DB 20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H
        DB 8AH,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,8AH
        DB 20H,20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H,20H
        DB 20H,20H,20H,20H,20H,20H
        DB 98H,9DH,9DH,9DH,9DH,9DH
        DB 9DH,9DH,9DH,9DH,9DH,8CH
        DB 00H
;
TEXT3:  DB 04H
        DB ' ВРЕМЯ:',11H      ; ТЕКСТ ДЛЯ МИГАНИЯ "ВРЕМЯ:"
        DB 00H
;
TEXT4:  DB 04H
        DB '      ',01H,'      ',00H      ; ТЕКСТ ДЛЯ СТИРАНИЯ "АЛМАЗЫ:  "
;
TEXT5:  DB 08H
        DB 'ИЩИ ВЫХОД',00H      ; ТЕКСТ ВМЕСТО "АЛМАЗЫ:00"
;
TEXT6:  DB 08H
        DB '      ',00H      ; ТЕКСТ ДЛЯ СТИРАНИЯ "ИЩИ ВЫХОД"
;
TEXT7:  DB 08H
        DB ' НАШЕЛ ! ',00H      ; ТЕКСТ, КОГДА ВЫХОД НАЙДЕН.
;
;*****
; НАЧАЛО ПРОГРАММЫ. ВЫВОД ЗАСТАВКИ.
;      ИЗМЕНЕНИЕ РЕЖИМОВ.
;*****
;
START:  DI      ; СТАРТ ПРОГРАММЫ.
        LXI SP,100H
        LXI H,FONPAL      ; УСТ. НЕВИДИМОЙ ПАЛИТРЫ.
        CALL MOVPAL
        MVI A,1      ; РАЗРЕШЕНИЕ ЗАПИСИ ПАЛИТРЫ.
        STA WRPAL
        EI
        HLT
        CALL CLEAR
        MVI A,0FFH
        STA SCROLL
        LDA TURBO      ; СОХРАНИТЬ УСТАНОВЛЕННЫЙ РЕЖИМ ЭКРАНА.
        STA BUFTUR
        XRA A      ; НОМЕР НУЛЕВОГО УРОВНЯ (МАССИВА ЗАСТАВКИ) .
        STA NUMBL

```

```

STA TURBO      ; ЗАСТАВКА ВСЕГДА В РЕЖИМЕ "ТУРБО".
STA DEMO       ; СБРОСИТЬ ПРИЗНАК РЕЖИМА ДЕМОНСТРАЦИИ.
MVI A,1
STA CHLEV      ; РАЗРЕШИТЬ ЗАГРУЗКУ И
CALL LDLEV     ; ЗАГРУЗИТЬ УРОВЕНЬ.
LXI H,0E05FH   ; УСТАНОВИТЬ НАЧАЛО ОКНА ВЫШЕ, ЧЕМ В ИГРЕ.
SHLD STWIND+1
LXI H,0C700H   ; АДРЕС НАЧАЛА НАДПИСИ АВТОРА В ЗАСТАВКЕ.
CALL AVT0
CALL OUTTXZ    ; ВЫВЕСТИ ТЕКСТОВОЕ СООБЩЕНИЕ.
LXI H,ZASPAL   ; УСТАНОВИТЬ ПАЛИТРУ ЦВЕТОВ ЗАСТАВКИ
CALL MOVPAL
MVI A,1
STA WRPAL
MVI A,20       ; УСТАНОВИТЬ ЗАДЕРЖКУ ПРОВЕРКИ КЛАВИШ ПОСЛЕ
STA TIMSTR     ; ПОЯВЛЕНИЯ ЗАСТАВКИ.
MVI A,200
STA DEMTIM
ZASTAV: IN 01   ; ПРОВЕРИТЬ КЛАВИШИ ВЫБОРА РЕЖИМА
CMA
ANI 0E0H
JZ ZAST01      ; ЕСЛИ НИ ОДНА НЕ НАЖАТА - НЕ ПРОВЕРЯТЬ ИХ.
RLC
RLC
RLC
ANI 3          ; ПОЛУЧИТЬ КОД СОСТОЯНИЯ КЛАВИШ (И СРАЗУ РЕЖИМ).
STA BUFTUR     ; ЗАПИСАТЬ В БУФЕР, ЧТОБЫ ПОТОМ ВЗЯТЬ ДЛЯ ИГРЫ.
CALL OUTTXZ    ; ОТОБРАЗИТЬ ИЗМЕНЕННОЕ СОСТОЯНИЕ РЕЖИМА.
ZAST01: HLT
CALL WINDOW    ; ОТОБРАЗИТЬ ОКНО.
CALL PANORM    ; ПАНОРАМИРОВАТЬ.
CALL TSTPAN    ; УСТАНОВИТЬ ВЕКТОРА ПАНОРАМИРОВАНИЯ.
LDA PANHOR     ; ПРОВЕРИТЬ КОНЕЦ ПАНОРАМИРОВАНИЯ ПО ГОРИЗ. ВЕКТОРУ.
ORA A
JNZ ZAST02     ; ЕСЛИ НЕ НОЛЬ, НЕ ДЕЛАТЬ РАССОГЛАСОВАНИЕ.
LDA INDMEN+1    ; ЕСЛИ НОЛЬ - РАССОГЛАСОВАНИЕ НА ВСЮ СТРОКУ МАССИВА.
XRI 27H        ; (Т.Е ИНДЕКС ЧЕЛОВЕЧКА ЛИБО 01H, ЛИБО 27H), ЧТОБЫ
STA INDMEN+1    ; ОКНО ДВИГАЛОСЬ ЗА НИМ И ПОКАЗЫВАЛО ВСЮ НАДПИСЬ.
ZAST02: LXI H,0C050H ; ВЫВОД ОБРАМЛЕНИЯ ТЕКСТОВОГО СООБЩЕНИЯ АЛМАЗАМИ.
LXI B,0510H
ZAST03: CALL SPRAJT ; ВЫВОД ВЕРХНЕЙ СТРОЧКИ АЛМАЗОВ.
INR H
INR H
DCR C
JNZ ZAST03
LXI H,0C040H
ZAST04: CALL SPRAJT ; ВЫВОД БОКОВЫХ СТРОЧЕК АЛМАЗОВ.
MOV A,H
ADI 30
MOV H,A
CALL SPRAJT
MOV A,H
SUI 30
MOV H,A
MOV A,L
SUI 10H
MOV L,A
DCR B
JNZ ZAST04
LXI H,0C200H
LXI B,0202H
ZAST05: CALL SPRAJT ; ВЫВОД В НИЖНЕЙ, НЕПОЛНОЙ СТРОЧКЕ.
INR H
INR H
DCR C
JNZ ZAST05
MOV A,H
ADI 20
MOV H,A
ZAST06: CALL SPRAJT
INR H

```

```

    INR H
    DCR B
    JNZ ZAST06
    HLT
    LDA TIMSTR ; ПРОВЕРИТЬ ОТРАБОТКУ ВРЕМЕНИ С НАЧАЛА ПОЯВЛЕНИЯ
    ORA A      ; ЗАСТАВКИ. ЕСЛИ ВРЕМЯ ОТРАБОТАНО - РАЗРЕШИТЬ
    JZ ZAST07  ; ПРОВЕРКУ КЛАВИШ.
    DCR A
    STA TIMSTR
    JMP ZASTAV
ZAST07: LDA STATUS
    CMA
    MOV B,A
    IN 01
    CMA
    ANI 0E0H
    ORA B
    JNZ ZAST13
    LDA DEMTIM
    DCR A
    STA DEMTIM
    JZ ZAST14
    JMP ZASTAV
ZAST13: MVI A,200
    STA DEMTIM
    LDA KEYKOD
    CPI CR      ; ЭТО "БК" ?
    JZ ZAST12   ; ЕСЛИ ДА - ПЕРЕЙТИ К НАЧАЛУ ИГРЫ.
    CPI LF      ; ЭТО "ПС" ?
    JNZ ZASTAV  ; ЕСЛИ НЕТ - ПРОДОЛЖАТЬ ОТОБРАЖЕНИЕ ЗАСТАВКИ.
ZAST14: MVI A,1 ; ЕСЛИ ДА - УСТАНОВИТЬ
    STA DEMO    ; ПРИЗНАК РЕЖИМА ДЕМОНСТРАЦИИ.
ZAST12: MVI A,20 ; УСТАНОВИТЬ ВРЕМЯ ДЛЯ ПОКАЗА РАССЫПАНИЯ НАДПИСИ.
    STA TIMSTR
ZAST08: HLT
    HLT
    CALL WINDOW ; ОТОБРАЖЕНИЕ РАССЫПАНИЯ НАДПИСИ, ПЕРЕД ПЕРЕХОДОМ
    CALL MOVELM ; К ИГРЕ.
    LDA TIMSTR
    ORA A
    JZ ZAST00
    DCR A
    STA TIMSTR
    JMP ZAST08
ZAST00: LDA BUFTUR ; ВЗЯТЬ ИЗ БУФЕРА РЕЖИМ ЭКРАНА И ПЕРЕЙТИ К ИГРЕ.
    STA TURBO
    JMP GSTART
;
SPRAJT: DI      ; ВЫВОД ОДНОГО СПРАЙТА ПРИ ВЫВОДЕ СТРОЧЕК АЛМАЗОВ.
    PUSH B
    SHLD ADSPR
    MVI A,CRYSTL
    CALL OUTSPR
    LHLD ADSPR
    POP B
    EI
    RET
;
OUTTXZ: LDA BUFTUR ; ОТОБРАЖЕНИЕ ТЕКСТА С УЧЕТОМ ИЗМЕНЕНИЯ РЕЖИМА.
    RRC
    JNC ZAST10
    MVI A,10H ; ЕСЛИ ЭТО ЦВЕТНОЙ РЕЖИМ, ТО "СС" В ИНВЕРСИИ (КОД 10H).
    STA INVCOL+1 ; А В ДРУГИХ ВЫКЛЮЧИТЬ ИНВЕРСИЮ (КОДОМ 11H).
    MVI A,11H
    STA INVCTL+1
    STA INVRUS+1
    JMP ZAST9
ZAST10: RRC
    JNC ZAST11
    MVI A,10H ; РЕЖИМ МОНО.
    STA INVCTL+1

```



```
MVI A,11H
STA INVCOL+1
STA INVRUS+1
JMP ZAST9
ZAST11: MVI A,10H      ; РЕЖИМ ТУРБО.
STA INVRUS+1
MVI A,11H
STA INVCOL+1
STA INVCTL+1
ZAST9:  MVI A,2
STA LINEFR+1
LXI H,0208H
SHLD PLACE
LXI H,TEXZAS
CALL TEXOUT
XRA A
STA LINEFR+1
RET

;
BUFTUR: DB 00          ; БУФЕР РЕЖИМА НА ВРЕМЯ ЗАСТАВКИ.
TIMSTR: DB 20          ; БУФЕР ВРЕМЕНИ ЗАДЕРЖКИ ПРОВЕРКИ КЛАВИШ И РАССЫП.
DEMO:   DB 00          ; ПРИЗНАК ДЕМОНИСТРАЦИИ.
DEMTIM: DB 200         ; ЗАДЕРЖКА ПЕРЕХОДА В РЕЖИМ ДЕМО ЕСЛИ НЕ НАЖ. КЛАВ.

;
ZASPAL: DB 00,3FH,0FH,0DEH,0FFH,00,00,00 ; ПАЛИТРА ЗАСТАВКИ.
        DB 38H,00,00,00,00,00,00,00

;
TEXZAS: DB 04H          ;          ТЕКСТ ЗАСТАВКИ.
        DB '  РЕЖИМ          УПРАВЛЕНИЕ',0AH,0AH
INVCTL: DB 1,11H,' UC',11H,8,'-МОНО      ',1,'ВК',8,'-СТАРТ/НАЧАЛО',0AH
INVCOL: DB 1,11H,' CC',11H,8,'-ЦВЕТ      ',1,'ПС',8,'-ДЕМОНИСТРАЦИЯ',0AH
INVRUS: DB 1,11H,' PUC',11H,8,'-ТУРБО    ',1,'ТАВ',8,'-САМОУБИЙСТВО',0AH
        DB 00

;
BAZA:   DS 880          ; НАЧАЛО РАБОЧЕГО МАССИВА ИГРЫ 880 БАЙТ (40*22).
;
```